DOCUMENT RESUME

ED 052 608 EM 009 062

TITLE	BASIC Simulation Programs; Volumes III and IV. Mathematics, Physics.
INSTITUTION	Brooklyn Polytechnic Inst., N.Y.; Digital Equipment Corp., Maynard, Mass.
SPONS AGENCY	National Science Foundation, Washington, D.C.
PUB DATE	31 Jan 71
NOTE	223p.; Developed by the Huntington Computer Project
AVAILABLE FROM	Digital Equipment Corporation, Educational Marketing (5-2), 146 Main Street, Maynard, Massachusetts 01754 ($4.00)
EDRS PRICE	EDRS Price MF-$0.65 HC Not Available from EDRS.
DESCRIPTORS	*Computer Assisted Instruction, *Computer Programs, Manuals, *Mathematics Instruction, *Physics Instruction, Program Descriptions, Secondary School Mathematics, Simulation, *Teaching Guides

ABSTRACT

The computer programs presented here were developed as a part of the Huntington Computer Project. They were tested on a Digital Equipment Corporation TSS-8 time-shared computer and run in a version of BASIC. Mathematics and physics programs are presented in this volume. The 20 mathematics programs include ones which review multiplication skills; solve financial problems concerning installment buying, long term loans, and savings accounts; find prime factors; find solutions to sets of up to 10 simultaneous equations; simulate the stock market; and find the volume of solids of revolution. The 21 physics programs include a plot routine illustrating the B field about one- and two-wire currents, a display of hydrogen line spectrum and energy level diagrams, a solution to lens problems, a calculation of mass defect, a photoelectric simulation, a plot routine to aid in visualizing Snell's law, a demonstration of the effects of changing velocity on orbital motion, and a plot routine for a graph of a fixed and a variable wave and the superposition of the waves. For each complete program the following information is also included: a description of the program, a statement of objectives, a discussion of presentation methods, and a sample printout. (JY)

DIGITAL EQUIPMENT CORPORATION

education

Basic Simulation Programs Volumes III & IV

Mathematics
Physics



HUNTINGTON COMPUTER PROJECT

A TEACHER'S MANUAL

(COMPUTER - RELATED MATERIALS)

Second Edition

January 31, 1971

Director: Dr. Ludwig Braun
Assistant Director: Dr. Marian Visich, Jr.

Polytechnic Institute of Brooklyn
333 Jay Street
Brooklyn, New York 11201

Developed by the Huntington Computer Project during the period May, 1968 and September, 1970. This effort was supported by the National Science Foundation under Grant No. J000079.

The enclosed material is a compilation of computer programs developed during the period May, 1968 to September, 1970. These programs were developed by teachers and students in the high schools which participated with us, and by the Project staff.

All of the enclosed programs have been tested on a Digital Equipment Corporation TSS-8 time-shared computer during the summer of 1970. To the best of our ability, we have assured ourselves that the programs actually run. It should be pointed out, however, that we were not able to make an exhaustive exploration of the programs. There may be undiscovered bugs (if there aren't, it may be the first time in the history of computing). We would appreciate hearing of any which emerge in the future.

These programs run in the version of BASIC which existed on the TSS-8 in August, 1970, and should run on most other versions of BASIC. The major potential problem on other machines is the output format (DEC uses 14 columns per print zone, while some other manufacturers use 15; we used the TAB function, which doesn't exist in all BASIC compiles). It may be necessary to make some minor changes in programs to adjust this format. Another possible problem is in the use of the RANDOMIZE command in some programs to start the random-number generator at a random point. If this command is not available, some other means should be devised for randomizing the start.

It is our sincere hope that these programs and their supporting documentation will be helpful to educators who are exploring the uses of computers in education.

We are anxious to hear of any bugs, errors, or improvements in these programs, and are especially anxious to hear of any novel ways of using them.

Ludwig Braun
Marian Visich, Jr.

TABLE OF CONTENTS

Volume I

Volume II

CHEMISTRY

Volume III

MATHEMATICS

MATHEMATICS (con't)

Volume IV

PHYSICS

PHYSICS (con't)

Volume V

SOCIAL STUDIES

Volume VI

TEACHER ASSISTANCE

DISCIPLINE MATHEMATICS 9th YEAR

SUBJECT MULTIPLICATION INVOLVING

ONE AND TWO DIGIT MULTIPLIERS

PROGRAM NAME ARITH

DESCRIPTION:

This program calls upon students, in a random fashion, to perform multiplication problems. Each student calculates five separate problems, and is allotted three chances to respond with the correct answer. At the conclusion of three incorrect responses, the computer will report to the student the correct answer. At the conclusion of five problems, the student will be given a score, and a new student will be called. Each factor will not exceed 100.

OBJECTIVES:

To review and reinforce students' ability to multiply by one and two digit multipliers.

PRELIMINARY PREPARATION:

None

DISCUSSION:

A. Operational Suggestions - This particular program is designed for group study, and may be used for an entire period. A portable TV was originally used to display computer output.

B. Follow-up - By modifying line numbers 520 and 560 to

```
520 LET X = A + B
560 PRINT A "+" B  "=";
```

this program becomes practice in addition. Modifications may also be made for division, subtraction and individual remedial work.

C. MODIFICATION-If your computer has string capability, student names may be used rather than student numbers, by making a few minor programming changes. This change (having the computer type out the student's name) increases the student interest.

```
HELLO CLASS.....TODAY I WANT TO REVIEW  MULTIPLICATION
WITH YOU.   WHEN I CALL ON YOU PLEASE COME UP AND TYPE IN
YOUR ANSWERS.  IF YOU ARE WRONG YOU GET TWO MORE CHANCES.
HOW MANY STUDENTS ARE THERE IN THE CLASS TODAY? 30
TEACHER!...GIVE EVERYONE A NUMBER FROM 1 TO 30


OK, STUDENT NO.  27  IT'S YOUR TURN!

GIVE ME YOUR LUCKY NUMBER
? 5


 81  X  54  = ? 4374
YOU'RE RIGHT STUDENT NO. 27


 63  X  97  = ? 6111
YOU'RE RIGHT STUDENT NO. 27


 62  X  17  = ? 1054
YOU'RE RIGHT STUDENT NO. 27


 50  X  78  = ? 3900
YOU'RE RIGHT STUDENT NO. 27


 93  X  81  = ? 7533
YOU'RE RIGHT STUDENT NO. 27
YOU GOT  5  RIGHT OUT OF 5 PROBLEMS.
GOODBYE...STUDENT NO. 27


OK, STUDENT NO.  25  IT'S YOUR TURN!

GIVE ME YOUR LUCKY NUMBER
?
↑C

READY
```

```
100 REM W. TEPPER, WYANDANCH HS, 4/21/69
105 REM  REVISED BY C.LOSIK  6-5-70
106 REM  WE DO A RANDOM PROBLEM FOR EACH STUDENT,  A -OP- B
110 REM THIS PROGRAM CALLS STUDENTS IN A RANDOM FASHION TO DO INDIVIDUAL
120 REM PROBLEMS.  BY MODIFYING A FEW STATEMENTS I CAN CHANGE THE
130 REM TYPE OF PROBLEMS.
140 REM REVISED 5/7/69
150 PRINT "HELLO CLASS.....TODAY I WANT TO REVIEW ";
151 REM  CHANGE BELOW FOR YOUR OPERATION
152 PRINT " MULTIPLICATION"
160 PRINT
170 PRINT "WITH YOU.   WHEN I CALL ON YOU PLEASE COME UP AND TYPE IN"
180 PRINT
190 PRINT "YOUR ANSWERS.  IF YOU ARE WRONG YOU GET TWO MORE CHANCES."
200 PRINT
210 PRINT "HOW MANY STUDENTS ARE THERE IN THE CLASS TODAY";
220 INPUT S
230 PRINT
240 PRINT "TEACHER!...GIVE EVERYONE A NUMBER FROM 1 TO"S
250 PRINT
260 PRINT
270 PRINT
272 REM  YOU MUST RANDOMIZE THE PROCESS FOR BEST RESULTS
275 RANDOMIZE
280 LET Q=INT(RND(-2)*S)
290 PRINT "OK, STUDENT NO. "Q" IT'S YOUR TURN!"
300 LET J=0
310 LET L =0
320 PRINT
330 PRINT
340 PRINT "GIVE ME YOUR LUCKY NUMBER"
350 INPUT Z
360 FOR T=1 TO Z
370 LET A=INT(RND(-2)*100)
380 LET B=INT(RND(-5)*100)
390 NEXT T
400 LET N=0
410 LET J=J+1
415 REM  X IS THE ANSWER TO A -OP- B
420 LET X=A*B
430 PRINT
440 PRINT
450 PRINT
455 REM  PRINT A -OP- B = ?
460 PRINT A" X "B" = ";
470 INPUT K
480 IF ABS(K-X)<.005 THEN 590
490 LET N=N+1
```

```
500 IF N=3 THEN 530
510 PRINT "YOU'RE WRONG...TRY AGAIN"
520 GO TO 460
530 PRINT "YOUR WRONG AGAIN"
540 PRINT "THE ANSWER IS "X
550 IF J<5 THEN360
560 PRINT "YOU GOT "L"  RIGHT OUT OF 20 PROBLEMS"
570 PRINT "GOOD BYE .... STUDENT NO."Q
580 GO TO 250
590 PRINT "YOU'RE RIGHT STUDENT NO."Q
600 LET L = L+1
610 IF J<5 THEN360
620 PRINT "YOU GOT "L" RIGHT OUT OF 5 PROBLEMS."
630 PRINT "GOODBYE...STUDENT NO."Q
640 GO TO 260
650 END
```

DISCIPLINE MATHEMATICS-SOCIAL SCIENCE

SUBJECT FINANCIAL PROBLEMS

PROGRAM NAME BANK

DESCRIPTION:

This program solves financial problems concerning installment buying, long-term loans, and savings accounts. The program gives you a choice of these three types of problems, and asks for the information needed to do said problem.

OBJECTIVES:

A. This program aids students in learning the terms used in certain financial problems.

B. Student will hopefully be motivated to learn the mathematical logic behind the solution of these problems.

PRELIMINARY PREPARATION:

A. Student - A review of decimals and fractions would be helpful.

B. Materials - A terminal, and a means by which to display the output to an entire class (e.g. overhead projector, closed circuit TV, etc.)

DISCUSSION:

A type of problem may be demonstrated through the use of the computer, then the mathematical logic behind the solution of the problem may be developed through the use of a flow chart similar to the one that follows.

Terminology may be taught when the computer asks for input (see sample run).

Since the execution time of one run is extremely short, many more problems may be demonstrated. Depending upon the ability of the class or student, a variety of relationships may be discovered.

Math
BANK

FINANCIAL PROBLEMS

```
THIS PROGRAM SOLVES THREE TYPES OF PROBLEMS:

     (1) INTEREST ON INSTALLMENT BUYING
     (2) PAYMENTS ON LONG TERM LOAN
     (3) BALANCE OF A SAVINGS ACCOUNT

WHICH PROBLEM WOULD YOU LIKE TO WORK WITH (TYPE 1, 2 OR 3)? 1

*****

THIS SECTION WILL DETERMINE THE ACTUAL INTEREST YOU PAY
WHEN YOU PURCHASE SOMETHING ON CREDIT.

WHAT IS THE CASH PRICE OF THE ARTICLE ($)? 88.99
          DOWN PAYMENT ($)? 10
          NUMBER OF PAYMENTS EXCLUDING THE DOWN PAYMENT? 18
          NUMBER OF PAYMENTS PER MONTH? 1
          AMOUNT PER PAYMENT ($)? 4.85


THE RATE OF INTEREST CHARGED WAS 5.69  PERCENT.


*****

WOULD YOU LIKE TO RUN THE PROGRAM AGAIN (1-YES, 0-NO)? 1
WHICH PROBLEM WOULD YOU LIKE TO WORK WITH (TYPE 1, 2 OR 3)? 2

*****

THIS SECTION WILL DETERMINE PAYMENTS FOR A LONG TERM LOAN.

WHAT IS THE AMOUNT BORROWED ($)? 3000
          INTEREST CHARGED (%)? 8
          INTERVAL BETWEEN PAYMENTS (MONTHS)? 1
          TERM OF THE LOAN (YEARS)? 2

DO YOU WISH TO SEE THE TOTALS ONLY - INSTEAD OF THE ENTIRE
TABLE - (1-YES, 0-NO)? 0
```

PERIOD	OUTSTANDING PRINCIPAL AT BEGINNING OF PERIOD	INTEREST DUE AT END OF PERIOD	PRINCIPAL REPAID AT END OF PERIOD
1	3000	20	115.68
2	2884.32	19.23	116.45
3	2767.87	18.45	117.23
4	2650.64	17.67	118.01
5	2532.63	16.88	118.8
6	2413.83	16.09	119.59
7	2294.24	15.29	120.39
8	2173.85	14.49	121.19
9	2052.66	13.68	122
10	1930.66	12.87	122.81
11	1807.85	12.05	123.63
12	1684.22	11.23	124.45
13	1559.77	10.4	125.28
14	1434.49	9.56	126.12
15	1308.37	8.72	126.96
16	1181.41	7.88	127.8
17	1053.61	7.08	128.66
18	924.95	6.17	129.51
19	795.44	5.3	130.38
20	665.06	4.43	131.25
21	533.81	3.56	132.12
22	401.69	2.68	133
23	268.69	1.79	133.89
24	134.8	.9	134.78
TOTALS		256.34	3000

```
YOUR MONTHLY PAYMENT IS $ 135.68  AND TOTALS $ 3056.34
```

```
*****

WOULD YOU LIKE TO RUN THE PROGRAM AGAIN (1-YES, 0-NO)? 1
WHICH PROBLEM WOULD YOU LIKE TO WORK WITH (TYPE 1, 2 OR 3)? 3

*****

THIS SECTION CALCULATES THE BALANCE OF A SAVINGS ACCOUNT
IN WHICH DEPOSITS ARE MADE REGULARLY.

WHAT IS THE AMOUNT DEPOSITED PER INTEREST PERIOD ($)? 10000
HOW OFTEN IS THE INTEREST COMPOUNDED (MONTHS,? 3
WHAT IS THE RATE OF INTEREST PAID (%)? 5
FOR HOW LONG WILL YOU DEPOSIT MONEY (YEARS)? 5


THE BALANCE OF YOUR ACCOUNT AFTER  5 YEARS WILL BE $ 202500


*****

WOULD YOU LIKE TO RUN THE PROGRAM AGAIN (1-YES, 0-NO)? 0


READY
```

```
100 REM FINANCIAL PROBLEMS   A. WEBB  12/67
101 REM REVISED 8/25/70 (D. PESSEL)
110 PRINT TAB(20);"FINANCIAL PROBLEMS"
115 REM REVISED BY W. TEPPER.  WYANDANCH H.S.  7/10/69
120 PRINT
130 PRINT"THIS PROGRAM SOLVES THREE TYPES OF PROBLEMS:"
132 PRINT
134 PRINT"     (1) INTEREST ON INSTALLMENT BUYING"
136 PRINT"     (2) PAYMENTS ON LONG TERM LOAN"
138 PRINT"     (3) BALANCE OF A SAVINGS ACCOUNT"
140 PRINT
142 PRINT"WHICH PROBLEM WOULD YOU LIKE TO WORK WITH (TYPE 1, 2 OR 3)";
144 INPUT Q1
146 PRINT
147 PRINT"*****"
148 PRINT
150 IF Q1>2 THEN 820
155 IF Q1>1 THEN 260
160 GO TO 590
260PRINT "THIS SECTION WILL DETERMINE PAYMENTS FOR A LONG TERM LOAN."
270 PRINT
280 PRINT"WHAT IS THE AMOUNT BORROWED ($)";
281 INPUT A
285 PRINT"             INTEREST CHARGED (%)";
286 INPUT I
290 PRINT"             INTERVAL BETWEEN PAYMENTS (MONTHS)";
291 INPUT P
295 PRINT"             TERM OF THE LOAN (YEARS)";
296 INPUT Y
300 PRINT
360 PRINT"DO YOU WISH TO SEE THE TOTALS ONLY - INSTEAD OF THE ENTIRE"
361 PRINT"TABLE - (1-YES, 0-NO)";
362 INPUT P5
370 PRINT
375 IF P5>0 THEN 430
380 PRINT"           OUTSTANDING"
390 PRINT"           PRINCIPAL AT                              PRINCIPAL"
400 PRINT"           BEGINNING          INTEREST DUE AT        REPAID AT"
410 PRINT"PERIOD     OF PERIOD          END OF PERIOD          END OF PERIOD"
420 PRINT
430 LET Z=(Y*12)/P
440 LET K=(I*(P/12))/100
445 LET E=A*K/(1-1/(1+K)↑Z)
446 LET E=INT(E*100+.5)/100
450 LET C=A
460 LET F=0
461 LET D1=0
470 LET T1=0
480 LET T1=T1+1
490 IF T1>Z THEN 554
500 LET B=T1
510 LET C=C-F
520 LET D=C*K
522 LET F=E-D
525 LET C=INT(C*100+.5)/100
530 LET D=INT(D*100+.5)/100
535 LET F=INT(F*100+.5)/100
541 LET D1=D1+D
548 IF P5>0 THEN 480
550 PRINT B;TAB(11);C;TAB(29);D;TAB(48);F
```

```
550 GO TO 480
554 IF P5<1 THEN 561
555 PRINT
556 LET D1=INT(D1*100+.5)/100
558 PRINT"TOTAL INTEREST PAID - $"D1
559 PRINT"TOTAOL PRINCIPAL REPAID - $"A
560 GO TO 565
561 PRINT"                                   ---------------     -------------"
564 PRINT"TOTALS";TAB(29);D1;TAB(48);A
565 LET E5=INT((D1+A)*100+.5)/100
566 PRINT
567 LET E6=E5/((Y*12)/P)
568 LET E6=INT(100*E6+.5)/100
569 PRINT"YOUR MONTHLY PAYMENT IS $"E6" AND TOTALS $"E5
570 GO TO 1060
590 PRINT"THIS SECTION WILL DETERMINE THE ACTUAL INTEREST YOU PAY"
600 PRINT"WHEN YOU PURCHASE SOMETHING ON CREDIT."
610 PRINT
620 PRINT"WHAT IS THE CASH PRICE OF THE ARTICLE ($)";
621 INPUT C
630 PRINT"            DOWN PAYMENT ($)";
631 INPUT D
640 PRINT"            NUMBER OF PAYMENTS EXCLUDING THE DOWN PAYMENT";
641 INPUT N
650 PRINT"            NUMBER OF PAYMENTS PER MONTH";
651 INPUT S
660 PRINT"            AMOUNT PER PAYMENT ($)";
661 INPUT R
690 PRINT
720 LET B=R*N+D
730 LET I=B-C
740 LET M=N/(S*12)
750 LET T=I*100/(B*M)
760 PRINT
770 PRINT
775 LET T=INT(100*T+.5)/100
780PRINT "THE RATE OF INTEREST CHARGED WAS"T" PERCENT."
790 GO TO 1060
820 PRINT"THIS SECTION CALCULATES THE BALANCE OF A SAVINGS ACCOUNT"
830 PRINT"IN WHICH DEPOSITS ARE MADE REGULARLY."
840 PRINT
860 PRINT"WHAT IS THE AMOUNT DEPOSITED PER INTEREST PERIOD ($)";
861 INPUT A
870 PRINT"HOW OFTEN IS THE INTEREST COMPOUNDED (MONTHS)";
871 INPUT B
880 PRINT"WHAT IS THE RATE OF INTEREST PAID (%)";
881 INPUT C
890 PRINT"FOR HOW LONG WILL YOU DEPOSIT MONEY (YEARS)";
891 INPUT D
950 LET F=0
960 LET E=(C/100)/(12/B)
970 LET G=(12/B)*D
980 LET T1=0
990 LET T1=T1+1
1000 IF T1=G+1 THEN 1030
1010 LET F=(E*A)+(A+F)
1020 GO TO 990
1030 PRINT
1040 PRINT
1045 LET F=INT(100*F+.5)/100
1050 PRINT"THE BALANCE OF YOUR ACCOUNT AFTER "D"YEARS WILL BE $"F
1060 PRINT
1070 PRINT
1080 PRINT
1081 PRINT"*****"
1082 PRINT
1084 PRINT"WOULD YOU LIKE TO RUN THE PROGRAM AGAIN (1-YES, 0-NO)";
1086 INPUT Q4
1090 IF Q4>0 THEN 142
1100 END
```

DISCIPLINE CALCULUS-GRADE 13

SUBJECT LENGTH OF ANY CURVE

PROGRAM NAME CRVLEN

DESCRIPTION:

This program approximates the length of any curve between two fixed points on the curve, by taking an increasing number of subintervals and computing the sum of the secants involved.

OBJECTIVES:

A. Time saving factor for computations.

B. By typing out successive approximations, the machine displays the manner by which the limit is approached.

C. The attendant discussion focuses attention upon the techniques needed to build up the analytic method for finding the length of a curve.

PRELIMINARY PREPARATION: None

DISCUSSION:

The operator inserts any function, sets up his own limits, and the computer proceeds to print out several approximations to the actual length a diagram (such as below) should be displayed, indicating the geometric basis for the computations.



Piecewise linear approximation of a smooth curve

```
                    LENGTH OF A CURVE

THIS PROGRAM APPROXIMATES THE LENGTH OF ANY CURVE BETWEEN
TWO POINTS HAVING P AND Q AS THEIR RESPECTIVE ABSCISSAS.
THE PROGRAM DIVIDES THE CURVE INTO INCREASING NUMBERS OF
SUBINTERVALS, JOINS THESE WITH SECANTS AND FINDS THE SUM
OF THESE SECANTS.

TO INPUT THE FUNCTION WHICH YOUR CURVE REPRESENTS, TYPE AS
FOLLOWS AFTER THE PROGRAM STOPS:
(TYPE THE 'RETURN' KEY AFTER EACH LINE INCLUDING 'RUN')

          1 GO TO 200
          300 DEF FNY(X)=....(YOUR FUNCTION OF X)....
          RUN

FOR EXAMPLE, TO USE THE FUNCTION 2*X↑3+3*X↑2-2*X+3
YOU WOULD TYPE:

          1 GO TO 200
          300 DEF FNY(X)=2*X↑3+3*X↑2-2*X+3
          RUN

YOU MIGHT TRY THAT AS YOUR FIRST RUN.


READY

1 GO TO 200
300 DEF FNY(X)=2*X↑3+3*X↑2-2*X+3
RUN

WHAT ARE THE ABSCISSAS OF THE END POINTS OF THE INTERVAL
WHOSE LENGTH YOU WANT (SMALLER ONE FIRST:P,Q)? -1,6

 NUMBER OF          SUM OF
SUBINTERVALS   SECANT LENGTHS   % CHANGE IN LENGTH
------------   --------------   ------------------

 1               525.0467        NO PREVIOUS VALUE
 2               525.1563         .02125142
 4               529.6528         .8557383
 8               531.0171         .2576957
 16              531.9648         .1783563
 32              532.0166         9.634868E-3
 64              532.0416         4.713789E-3
 128             532.0485         1.287715E-3
 256             532.0501         3.066667E-4

*****

WOULD YOU LIKE TO TRY NEW END POINTS (1-YES, 0-NO)? 0

TO TRY ANOTHER FUNCTION, RETYPE LINE 300, AND 'RUN'.
SEE INSTRUCTIONS FOR MORE DETAILS.  IF YOU ARE FINISHED,
TYPE '1' AND 'RETURN' KEY AFTER THE PROGRAM STOPS.


READY

1
```

```
100 REM LENGTH OF A CURVE-0.0. O'CONNOR 7-69-68
101 REM REVISED 8-7-70 (D. TESSEL) (COMBINATION OF LECUB & LEVUB)
103 REM IMPORTANT VARIABLES:  S-SECANT LENGTH; S1-PREVIOUS SECANT
104 REM LENGTH; P-PERCENT CHANGE IN SECANT LENGTH
110 PRINT TAB(20); "LENGTH OF A CURVE"
111 PRINT
120 PRINT "THIS PROGRAM APPROXIMATES THE LENGTH OF ANY CURVE BETWEEN"
121 PRINT "TWO POINTS HAVING P AND Q AS THEIR RESPECTIVE ABSCISSAS."
122 PRINT "THE PROGRAM DIVIDES THE CURVE INTO INCREASING NUMBERS OF"
123 PRINT "SUBINTERVALS, JOINS THESE WITH SECANTS AND FINDS THE SUM"
124 PRINT "OF THESE SECANTS."
125 PRINT
126 PRINT "TO INPUT THE FUNCTION WHICH YOUR CURVE REPRESENTS, TYPE AS"
127 PRINT "FOLLOWS AFTER THE PROGRAM STOPS:"
128 PRINT "(TYPE THE 'RETURN' KEY AFTER EACH LINE INCLUDING 'RUN')"
129 PRINT
130 PRINT "          ! GO TO 300"
131 PRINT "          300 DEF FNY(X)=.....(YOUR FUNCTION OF X)....."
132 PRINT "          RUN"
133 PRINT
134 PRINT "FOR EXAMPLE, TO USE THE FUNCTION 3*X^3+3*X^2-3*X+3"
135 PRINT "YOU WOULD TYPE:"
136 PRINT
137 PRINT "          ! GO TO 300"
138 PRINT "          300 DEF FNY(X)=3*X^3+3*X^2-3*X+3"
139 PRINT "          RUN"
140 PRINT
141 PRINT "YOU MIGHT TRY THAT AS YOUR FIRST RUN."
150 STOP
200 REM CALCULATION AND PRINTING OF RESULTS
205 PRINT "WHAT ARE THE ABSCISSAS OF THE END POINTS OF THE INTERVAL"
206 PRINT "WHOSE LENGTH YOU WANT (SMALLER ONE FIRST)";P,Q
207 INPUT P,Q
208 IF P<Q THEN 213
209 PRINT "P MUST BE LESS THAN Q!"
210 GO TO 207
212 INPUT P,Q
213 PRINT
214 PRINT " NUMBER OF",,"     SUM OF"
215 PRINT "SUBINTERVALS",,"SECANT LENGTHS",,"  % CHANGE IN LENGTH"
216 PRINT "-------------","-------------","  ----------------"
217 PRINT
220 LET S1=0
230 FOR N=1 TO 9
300 DEF FNY(X)=(((X)+3)*3*X^2-3*X+3
310 LET E=2^(5-1)
320 LET H=(Q-P)/E
330 LET S=0
340 FOR I=0 TO E-1
350 LET L=SQR((FNY(P+I*H+H)-FNY(P+I*H))^2+H*H)
360 LET S=S+L
370 NEXT I
372 IF S1<0 THEN 375
373 PRINT E,S,"  NO PREVIOUS VALUE"
374 GO TO 385
375 LET P1=((ABS(S1-S))/S1)*100
380 PRINT E,S,"  "P1
385 LET S1=S
390 NEXT N
400 PRINT
401 PRINT "*****"
402 PRINT
403 PRINT "WOULD YOU LIKE TO TRY NEW END POINTS (1-YES, 0-NO)";
404 INPUT G1
405 IF G1>0 THEN 210
410 PRINT
445 PRINT "TO TRY ANOTHER FUNCTION, RETYPE LINE 300, AND 'RUN'."
446 PRINT "SEE INSTRUCTIONS FOR MORE DETAILS.  IF YOU ARE FINISHED,"
447 PRINT "TYPE '1' AND 'RETURN' KEY AFTER THE PROGRAM STOPS."
500 END
```

DISCIPLINE CALCULUS - GRADE 13

SUBJECT AREA UNDER ANY CURVE, (ANALYTICALLY DEFINED)

PROGRAM NAME CVAREA

DESCRIPTION:

By numerical methods, this program evaluates the definite integral of f(x), from x=a to x=b, by four different methods of successive approximation:

I Rectangles (starting with f(a) as height)

II Rectangles (starting with f(a+h) as height)

III Trapezoids

IV Parabolas (Simpson's Rule)

OBJECTIVES:

A. Enhances comprehension of the analytic procedures for finding the area under a curve.

B. Dramatizes the limiting processes involved.

C. Decreases the time needed for lengthy computations.

PRELIMINARY PREPARATION:

Prior to the computer run, diagrams should appear on the board, or on the overhead projector screen to demonstrate the geometric significance of the computer output.



13 Inscribed Rectangular Approximation



Circumscribed Rectangular Approximation

DISCUSSION:

This program may be run as an introduction to the problem of finding the area under a curve. In some classes, the consideration of Simpson's Rule may be omitted or briefly hinted at. With the more mathematically talented classes, an explanation of this parabolic approximation should precede the running of the program.



Trapezoid Approximation

AREA UNDER A CURVE - INTEGRATION

THIS PROGRAM EVALUATES THE DEFINITE INTEGRAL OF F(X)
FROM X=A TO X=B BY FOUR METHODS OF NUMERICAL APPROXIMATION:

I RECTANGLES (INITIAL HEIGHT OF F(X))
II RECTANGLES (INITIAL HEIGHT OF F(X+H))
III TRAPEZOIDS
IV PARABOLAS (SIMPSON'S RULE)

AFTER THE PROGRAM STOPS, YOU MAY ENTER YOUR FUNCTION AS FOLLOWS:

```
1 GO TO 200
300 DEF FNY(X)=...(YOUR FUNCTION OF X)...
RUN
```

FOR EXAMPLE, TO FIND THE AREA UNDER THE CURVE Y=X↑3 YOU
WOULD TYPE:

```
1 GO TO 200
300 DEF FNY(X)=X↑3
RUN
```

YOU MIGHT TRY THAT AS YOUR FIRST RUN.
END EACH LINE, INCLUDING 'RUN', WITH THE 'RETURN' KEY.

READY

```
1 GO TO 200
300 DEF FNY(X)=X↑3
RUN
```

WHAT ARE YOUR VALUES FOR A AND B (SMALLER FIRST:A,B)? 1,10

NUMBER OF SUBINTERVALS	I. SUM OF RECTANGLES	II. SUM OF RECTANGLES	III. SUM OF TRAPEZOIDS	IV. SUM OF PARABOLAS
2	753.1875	5248.687	3000.937	2499.75
4	1501.172	3748.922	2625.047	2499.75
8	1969.137	3093.012	2531.074	2499.75
16	2226.612	2788.55	2507.581	2499.75
32	2361.223	2642.192	2501.708	2499.75
64	2429.997	2570.481	2500.239	2499.75

NOTE THAT SIMPSON'S RULE (IV) CONVERGES FASTEST.

WOULD YOU LIKE TO TRY NEW VALUES FOR A AND B (1-YES, 0-NO)? 0

*****

TO USE A NEW FUNCTION YOU NEED ONLY RETYPE LINE 300
AND 'RUN'. SEE INSTRUCTIONS FOR MORE DETAILS.
IF YOU ARE FINISHED, TYPE '1' AND THE 'RETURN' KEY.

READY

1

```
100 REM AREA UNDER A CURVE-G.J. O'CONNOR
101 REM REVISED 8/18/70 (D. PESSEL) (COMBINATION OF DEFIN AND ACCUQ)
102 REM IMPORTANT VARIABLES: D-# OF SUBINTERVALS; AREA BY
103 REM RECTANGLES (F(X))-P, BY RECTANGLES (F(X+H))-Q,
104 REM BY TRAPEZOIDS-T, BY PARABOLAS-S; C-STORES PREVIOUS
105 REM VALUE OF S.
110 PRINT TAB(15); "AREA UNDER A CURVE - INTEGRATION"
111 PRINT
112 PRINT "     THIS PROGRAM EVALUATES THE DEFINITE INTEGRAL OF F(X)"
113 PRINT "FROM X=A TO X=B BY FOUR METHODS OF NUMERICAL APPROXIMATION:"
114 PRINT
115 PRINT TAB(20); "I RECTANGLES (INITIAL HEIGHT OF F(X))"
116 PRINT TAB(19); "II RECTANGLES (INITIAL HEIGHT OF F(X+H))"
117 PRINT TAB(18); "III TRAPEZOIDS"
118 PRINT TAB(19); "IV PARABOLAS (SIMPSON'S RULE)"
119 PRINT
120 PRINT "AFTER THE PROGRAM STOPS, YOU MAY ENTER YOUR FUNCTION AS";
121 PRINT " FOLLOWS:"
122 PRINT
123 PRINT TAB(13); "1 GO TO 200"
124 PRINT TAB(13); "300 DEF FNY(X)=...(YOUR FUNCTION OF X)..."
125 PRINT TAB(13); "RUN"
126 PRINT
127 PRINT "FOR EXAMPLE, TO FIND THE AREA UNDER THE CURVE Y=X↑3 YOU"
128 PRINT "WOULD TYPE:"
129 PRINT
130 PRINT TAB(13);"1 GO TO 200"
131 PRINT TAB(13);"300 DEF FNY(X)=X↑3"
132 PRINT TAB(13);"RUN"
133 PRINT
134 PRINT "YOU MIGHT TRY THAT AS YOUR FIRST RUN."
135 PRINT "END EACH LINE, INCLUDING 'RUN', WITH THE 'RETURN' KEY."
150 STOP
200 PRINT "WHAT ARE YOUR VALUES FOR A AND B (SMALLER FIRST:A,B)";
201 INPUT A,B
203 IF B>=A THEN 210
204 PRINT "A MUST BE LESS THAN B!"
205 GO TO 200
210 PRINT
211 PRINT " NUMBER OF","I. SUM OF","II. SUM OF","III. SUM OF",
212 PRINT "IV. SUM OF"
213 PRINT "SUBINTERVALS","RECTANGLES","RECTANGLES","TRAPEZOIDS",
214 PRINT "PARABOLAS"
218 PRINT "--------  ---","----------","----------","----------",
219 PRINT "----------"
250 LET M=-2
260 LET S=0
300 DEF FNY(X)=X↑3
310 LET M=M+3
320 FOR N=M TO M+2
330 LET C=S
340 LET Q=0
350 LET P=0
360 LET D=2↑N
```

```
365 PRINT D,
370 LET H=(B-A)/D
380 FOR I=0 TO (D-1)
390 LET P=P+H*FNY(A+I*H)
400 LET Q=Q+H*FNY(A+I*H+H)
410 NEXT I
415 PRINT P,Q,
420 LET T=(P+Q)/2
425 PRINT T,
430 LET U=FNY(A)+FNY(B)
440 FOR J=2 TO (D-2) STEP 2
450 LET U=U+2*FNY(A+J*H)
460 NEXT J
470 LET V=0
480 FOR K=1 TO (D-1) STEP 2
490 LET V=V+4*FNY(A+K*H)
500 NEXT K
510 LET S=(U+V)*(H/3)
520 PRINT S
530 NEXT N
535 IF D<64 THEN 310
540 IF ABS((C-S)/((C+S)/2))>.0001 THEN 310
550 PRINT
560 PRINT "NOTE THAT SIMPSON'S RULE (IV) CONVERGES FASTEST."
600 PRINT
610 PRINT "WOULD YOU LIKE TO TRY NEW VALUES FOR A AND B (1-YES, 0-NO)";
611 INPUT Q5
612 PRINT
613 PRINT "*****"
614 PRINT
615 IF Q5>0 THEN 200
620 PRINT "TO USE A NEW FUNCTION YOU NEED ONLY RETYPE LINE 300"
621 PRINT "AND 'RUN'.  SEE INSTRUCTIONS FOR MORE DETAILS."
622 PRINT "IF YOU ARE FINISHED, TYPE '1' AND THE 'RETURN' KEY."
650 END
```

DISCIPLINE MATHEMATICS, JR. HIGH

GENERAL MATH

SUBJECT GREATEST COMMON DIVISOR

PROGRAM NAME GCD

DESCRIPTION:

This program finds the greatest common divisor for two or more numbers.

OBJECTIVES:

To aid the teacher in demonstrating a method of finding the greatest common divisor.

PRELIMINARY PREPARATION:

See discussion.

DISCUSSION:

It is suggested that the teacher explain the meaning of the greatest common divisor prior to using this program, and show a number of examples.

By using the flow chart which follows, the method and logic the computer uses, can be explained to students. It is suggested that a supplementary device be used to display output to class-size groups.

```
THIS PROGRAM WILL FIND THE GREATEST COMMON DIVISOR
FOR TWO OR MORE NUMBERS.
HOW MANY NUMBERS DO YOU WISH TO INVESTIGATE? 3
TYPE IN THE NUMBERS, ONE AFTER EACH QUESTION MARK.
? 12
? 36
? 96
THE NUMBERS 12  36  96 HAVE THE G.C.D.  12


ANOTHER SET OF NUMBERS (1=YES, 0=NO) ? 1
HOW MANY NUMBERS DO YOU WISH TO INVESTIGATE? 3
TYPE IN THE NUMBERS, ONE AFTER EACH QUESTION MARK.
? 20
? 36
? 96
THE NUMBERS 20  36  96 HAVE THE G.C.D.  4


ANOTHER SET OF NUMBERS (1=YES, 0=NO) ? 1
HOW MANY NUMBERS DO YOU WISH TO INVESTIGATE? 3
TYPE IN THE NUMBERS, ONE AFTER EACH QUESTION MARK.
? 20
? 36
? 97
THE NUMBERS 20  36  97 ARE RELATIVELY PRIME.


ANOTHER SET OF NUMBERS (1=YES, 0=NO) ? 0


READY
```

```
100 REM D. TEPPER  WYANDANCH U.S.     -    MATHEMATICS
110 REM  REVISED BY C.LOSIK  6-10-70
111 REM  X(I) ARE THE NUMBERS (UP TO 100)
120 PRINT "THIS PROGRAM WILL FIND THE GREATEST COMMON DIVISOR"
130 PRINT "FOR TWO OR MORE NUMBERS."
140 DIM X(100)
150 PRINT "HOW MANY NUMBERS DO YOU WISH TO INVESTIGATE";
160 INPUT N
165 IF ABS(N-INT(N))<.0001 THEN 170
166 PRINT "TRY AGAIN."
167 GO TO 150
170 PRINT "TYPE IN THE NUMBERS. ONE AFTER EACH QUESTION MARK."
175 LET S=1E85
180 FOR K=1 TO N
190 INPUT X(K)
193 IF X(K)>S THEN 210
200 LET S=X(K)
210 NEXT K
220 LET G=0
230 FOR M=2 TO S
240 FOR I=1 TO N
250 IF X(I)/M<>INT(X(I)/M) THEN 300
260 NEXT I
290 LET G=M
300 NEXT M
310 PRINT "THE NUMBERS";
320 FOR T=1 TO N
330 PRINT X(T);
340 NEXT T
350 IF G>0 THEN 380
360 PRINT "ARE RELATIVELY PRIME."
370 GO TO 390
380 PRINT "HAVE THE G.C.D. ";G
390 PRINT
400 PRINT
410 PRINT "ANOTHER SET OF NUMBERS (1=YES, 0=NO) ";
420 INPUT Z
430 IF Z=1 THEN 150
440 IF Z=0 THEN 470
450 PRINT "TYPE 1 OR 0 AS DIRECTED."
460 GO TO 420
470 END
```

DISCIPLINE CALCULUS - GRADE 13

SUBJECT LIMIT OF $\frac{\sin x}{x}$

PROGRAM NAME LIMSIN

DESCRIPTION:

This program demonstrates that the limit of $\frac{\sin x}{x}$, as x approaches 0, equals 1, provided x is measured in radians. If x is measured in degrees, the limit equals approximately .017.

OBJECTIVES:

A. To demonstrate the manner by which the limit of $\frac{\sin x}{x}$ is approached.

B. To show that degree measure does not yield the same solution as radian measure.

PRELIMINARY PREPARATION:

A. Student

Knowledge of degree vs. radian measure.

B. Materials

None

DISCUSSION:

Following the computer type-out, the teacher will use the analytic method to evaluate the limit. Prior to this discussion, the student should be reminded of the area formulas for a triangle and for a sector in terms of the central angle measured in radians. A geometric diagram should be presented showing the sector lying between two triangles.



Here, $\frac{1}{2}r^2 \sin \Theta \leq \frac{1}{2}r^2\Theta \leq \frac{1}{2}r^2 \tan \Theta$

Circular Sector with Circumscribed and Inscribed Triangles

The teacher can modify the type-out by inserting: 195 Go to 300. This decreases the number of lines typed out to the final eleven appearing on the program "run".

THIS PROGRAM DEMONSTRATES THAT THE LIMIT OF
F(X) = (SIN X)/X, AS X APPROACHES 0, IS EQUAL TO 1,
PROVIDED X IS MEASURED IN RADIANS.

$$\lim_{X \to 0} \frac{\sin(X)}{X} = 1$$

WHEN X IS IN DEGREES,		WHEN X IS IN RADIANS,	
X IS	F(X) IS	X IS	F(X) IS
90	.01111111	1.570795	.6366203
85	.01171994	1.483529	.6715035
80	.01231009	1.396262	.705317
75	.01287901	1.308996	.7379134
70	.01342418	1.221729	.7691492
65	.01394319	1.134463	.7988866
60	.01443375	1.047197	.8269936
55	.01489367	.9599303	.8533449
50	.01532088	.8726639	.8778225
45	.01571347	.7853975	.9003165
40	.01606968	.6981311	.9207256
35	.01638789	.6108647	.9389575
30	.01666665	.5235983	.9549297
25	.01690472	.4363319	.9685698
20	.01710099	.3490656	.9798156
15	.01725459	.2617992	.988616
10	.0173648	.1745328	.9949308
5	.01743113	.08726639	.9987313
1	.01745239	.01745328	.9999492
.9	.01745256	.01570795	.9999589
.8	.01745271	.01396862	.9999675
.7	.01745284	.01221729	.9999751
.6	.01745296	.01047197	.9999817
.5	.01745306	8.726639E-3	.9999873
.4	.01745314	6.981312E-3	.9999919
.3	.0174532	5.235984E-3	.9999954
.2	.01745324	3.490656E-3	.999998
.1	.01745327	1.745328E-3	.9999995
.09	.01745327	1.570795E-3	.9999996
.08	.01745327	1.396262E-3	.9999997
.07	.01745327	1.221729E-3	.9999998
.06	.01745327	1.047197E-3	.9999998
.05	.01745328	8.726639E-4	.9999999
.04	.01745328	6.981311E-4	.9999999
.03	.01745328	5.235984E-4	1
.02	.01745328	3.490656E-4	1
.01	.01745328	1.745328E-4	1

READY

```
100 REM  BRUCE BRENT   HHHH   BKLYN POLY   7/11/69
105 REM  REVISED BY C.LOSIK   8-27-70
110 PRINT "     THIS PROGRAM DEMONSTRATES THAT THE LIMIT OF"
115 PRINT "F(X) = (SIN X)/X,  AS X APPROACHES 0, IS EQUAL TO 1,"
117 PRINT "PROVIDED X IS MEASURED IN RADIANS."
120 PRINT
125 PRINT " "," SIN(X)"
130 PRINT " ","LIMIT  ------  =  1"
135 PRINT " ","X-->0    X"
140 PRINT
150 PRINT
160 PRINT "WHEN X IS IN DEGREES,","WHEN X IS IN RADIANS,"
165 PRINT "--------------------","--------------------"
170 PRINT "X IS","F(X) IS","X IS","F(X) IS"
175 PRINT "----","-------","----","-------"
180 PRINT
200 FOR Y=90 TO 5 STEP -5
210 LET Z=Y
220 LET Z=3.14159*Z/180
230 LET X=SIN(Z)/Z
240 LET Q=SIN(Z)/Y
250 PRINT Y,Q,Z,X
260 NEXT Y
270 PRINT
300 FOR Y=1 TO .1 STEP -.1
310 LET Z=Y
320 LET Z=3.14159*Z/180
330 LET X=SIN(Z)/Z
340 LET Q=SIN(Z)/Y
350 PRINT Y,Q,Z,X
360 NEXT Y
370 PRINT
400 FOR Y=.09 TO .01 STEP -.01
410 LET Z=Y
420 LET Z=3.14159*Z/180
430 LET X=SIN(Z)/Z
440 LET Q=SIN(Z)/Y
450 PRINT Y,Q,Z,X
460 NEXT Y
500 END
```

DISCIPLINE MATHEMATICS 10th YEAR
GEOMETRY

SUBJECT AREA OF A CIRCLE

PROGRAM NAME PI2

DESCRIPTION:

This program computes the area of a circle and "pi" by using the areas of inscribed and circumscribed regular polygons.

OBJECTIVES:

As an introduction to the limit process and a method for approximating "pi".

PRELIMINARY PREPARATION:

A. Student - Students must know how to calculate the area of a circle and a triangle using the formulas: $A = \pi R^2$ and $A = \frac{1}{2}bh$.

B. Materials - chalkboard, board compass, and straight edge.

DISCUSSION:

Ask students to find the area of a circle without using the formula. The instructor may suggest to the class to inscribe and/or circumscribe an equilateral triangle. Have students compare the area of their figures to that of the circle. Some students will suggest to increase the number of sides and the instructor should suggest that a regular hexagon be used for convenience of drawing. This can be illustrated on the chalkboard for the class. Another comparison is made between the areas and then the students will observe that to obtain any satisfactory results, the number of sides must increase greatly. At this moment the instructor should introduce this program and explain to the class that the program will increase the number of sides of a regular polygon and compute the area of each new figure. A table is printed giving the areas of both inscribed and circumscribed regular polygons and also the number of sides for each area. The students can readily see that the machine has eliminated the tedious calculations. Now, have the students calculate the area of the circle using the formula and make a comparison of results; thus, the students can observe that the areas of the polygons approach the area of the circle.

DISCUSSION: (con' t)

If students had taken a unit circle, they would have observed a method for approximating " pi" .

Due to machine operation, the value of " pi" was used to convert degrees into radians. To avoid any circular reasoning, the instructor can use half-angle formulas to eliminate " pi" from this program.

```
AREA OF A CIRCLE USING INSCRIBED AND CIRCUMSCRIBED REGULAR POLYGONS

*****

WHAT IS THE RADIUS OF THE CIRCLE? 10

 INSCRIBED      CIRCUMSCRIBED    NUMBER OF     INSCRIBED      CIRCUMSCRIBED
   AREA             AREA           SIDES       % ERROR          % ERROR

 189.9039         519.6142          3          -56.65           65.4
 259.8075         346.4098          6          -17.3            10.27
 299.9998         321.5367          12         -4.51            2.35


HOW MANY SIDES DO YOU THINK ARE NEEDED TO APPROXIMATE
THE AREA OF THIS CIRCLE? 100

 313.9523         314.2624          100        -.07             .03

WOULD YOU LIKE TO TRY ANOTHER NUMBER OF SIDES (1-YES, 0-NO)? 1
HOW MANY SIDES DO YOU THINK ARE NEEDED TO APPROXIMATE
THE AREA OF THIS CIRCLE? 1E6
THAT MANY SIDES IS VALID, BUT NOT NECESSARY FOR A
GOOD APPROXIMATION.  USE 10000 AS THE MAXIMUM NUMBER.
HOW MANY SIDES DO YOU THINK ARE NEEDED TO APPROXIMATE
THE AREA OF THIS CIRCLE? 10000

 314.159          314.159           10000      0                0

WOULD YOU LIKE TO TRY ANOTHER NUMBER OF SIDES (1-YES, 0-NO)? 0
WOULD YOU LIKE TO TRY ANOTHER RADIUS (1-YES, 0-NO)? 1

*****

WHAT IS THE RADIUS OF THE CIRCLE? 1000000
ANY RADIUS WILL WORK, BUT USE A NUMBER LESS THAN 1000.
WHAT IS THE RADIUS OF THE CIRCLE? 999

 INSCRIBED      CIRCUMSCRIBED    NUMBER OF     INSCRIBED      CIRCUMSCRIBED
   AREA             AREA           SIDES       % ERROR          % ERROR

 1.896443E+6      5.165754E+6       3          -56.65           65.4
 2.592861E+6      3.457173E+6       6          -17.3            10.27
 2.994001E+6      3.208960E+6       12         -4.51            2.35


HOW MANY SIDES DO YOU THINK ARE NEEDED TO APPROXIMATE
THE AREA OF THIS CIRCLE? 10000

 3.138710E+6      3.138310E+6       10000      0                0

WOULD YOU LIKE TO TRY ANOTHER NUMBER OF SIDES (1-YES, 0-NO)? 0
WOULD YOU LIKE TO TRY ANOTHER RADIUS (1-YES, 0-NO)? 0

*****


READY
```

```
100 REM ILLUSTRATION OF LIMITS USING CIRCLES AND POLYGONS
101 REM REVISED 6/3/70 (D. PESSEL)
105 REM IMPORTANT VARIABLES: A1-INSCRIBED AREA;  A2-CIRCUMSCRIBED
106 REM AREA;  A3-ACTUAL AREA;  P1-% ERROR OF A1;  P2-% ERROR OF A2
110 PRINT "AREA OF A CIRCLE USING INSCRIBED AND CIRCUMSCRIBED ";
111 PRINT "REGULAR POLYGONS"
112 PRINT
115 PRINT "*****"
116 PRINT
120 PRINT "WHAT IS THE RADIUS OF THE CIRCLE";
125 INPUT R
127 IF R<1000 THEN 131
128 PRINT "ANY RADIUS WILL WORK, BUT USE A NUMBER LESS THAN 1000."
129 GO TO 120
130 INPUT R
131 IF R>=.1 THEN 134
132 PRINT "RADIUS SHOULD BE AT LEAST .1!!"
133 GO TO 120
134 LET A3=3.1416*R*R
135 PRINT
136 PRINT
140 PRINT "INSCRIBED","CIRCUMSCRIBED","  NUMBER OF",
141 PRINT "INSCRIBED","CIRCUMSCRIBED"
150 PRINT "  AREA","      AREA","    SIDES"," % ERROR","   % ERROR"
155 PRINT
160 FOR K=0 TO 8
170 LET N=3*(2^K)
175 GOSUB 180
177 NEXT K
178 GO TO 240
179 REM COMPUTATION SUBROUTINE (LINES 180-220)
180 LET L=2*R*SIN(3.14159/N)
190 LET A1=R*COS(3.14159/N)*N*L/2
200 LET A2=N*(R^2)*TAN(3.14159/N)
205 LET P1=((A1-A3)/A3)*100
206 LET P2=((A2-A3)/A3)*100
210 PRINT A1,A2,"   ";N,INT(P1*100+.5)/100, INT(P2*100+.5)/100
220 RETURN
240 PRINT
250 PRINT
260 PRINT "HOW MANY SIDES DO YOU THINK ARE NEEDED TO APPROXIMATE"
261 PRINT "THE AREA OF THIS CIRCLE";
262 INPUT N
263 IF N>1E5 THEN 266
264 IF N<3 THEN 266
265 GO TO 273
266 PRINT "THE NUMBER OF SIDES SHOULD BE AT LEAST THREE!!"
267 GO TO 260
268 PRINT "THAT MANY SIDES IS VALID, BUT NOT NECESSARY FOR A"
269 PRINT "GOOD APPROXIMATION.  USE 16000 AS THE MAXIMUM NUMBER."
270 GO TO 260
273 PRINT
274 LET N=INT(N+.5)
275 GOSUB 180
280 PRINT
285 PRINT "WOULD YOU LIKE TO TRY ANOTHER NUMBER OF SIDES";
286 PRINT " (1-YES, 0-NO)";
287 INPUT Q1
288 IF Q1>0 THEN 260
290 PRINT "WOULD YOU LIKE TO TRY ANOTHER RADIUS (1-YES, 0-NO)";
291 INPUT Q2
293 PRINT
294 PRINT "******"
295 PRINT
296 IF Q2>0 THEN 120
300 END
```

DISCIPLINE MATHEMATICS 9, 10, 11, 12, 13

SUBJECT PLOTTING A GRAPH

PROGRAM NAME PLOTTR

DESCRIPTION:

This program plots the graph of any function (analytically defined) which the operator inputs into the program.

OBJECTIVES:

A. To check a student's plotting procedures.

B. To obtain a quick plot of an involved function.

PRELIMINARY PREPARATION:

A. Student - Knowledge of coordinates, and plotting procedures.

B. Materials - graph paper for plotting

DISCUSSION:

The operator inputs any analytic function, along with the lower and the upper limits for x and the interval to appear on the x-axis.

The type-out positions x-values on the vertical axis, and y-values on the horizontal axis.

The points typed out may be connected by a smooth curve, and the graph may be rotated 90° to give the usual positioning of a function of x.

It should be noted that because the carriage spacing is discrete, many smooth curves may appear slightly jagged.

In the third sample run, a plot is made of a rather complex transcendental function. It is worth mentioning that this plot is obtained as easily, using this program as is that of the function Y=X.

The teacher should notice also, that, in this third sample run, we have found two of the roots of the function

Y=X+LOG(2*(SIN(X))↑2)-1.5xCOS(X)

(at X=1 and X=2.98). This program may be used for finding the roots of such difficult funcions.

```
    THIS PROGRAM WILL GRAPH A FUNCTION OF X BETWEEN ANY LIMITS
(A AND B) YOU CHOOSE, WITH AN INTERVAL OF YOUR CHOICE (I)
BETWEEN SUCCESSIVE VALUES OF X; IF YOU TYPE THE FOLLOWING:
        1 GO TO 220
        220 DEF FNY(X)=...(YOUR FUNCTION OF X)...
        230 LET A=...(YOUR SMALLER LIMIT OF X)...
        240 LET B=...(YOUR LARGER LIMIT OF X)...
        250 LET I=...(YOUR X-INCREMENT)...
        RUN

READY

1 GO TO 220
220 DEF FNY(X)=SIN(X)
230 LET A=0
240 LET B=7
250 LET I=.5
RUN

X
-
V
A
L
U
E
S                    - - - - Y-VALUES - - - -
                -3          -1.5          0          1.5
                 I----------I-----+-----I----------I----------I-----+-----I
 0                                      *
 .5                                     I     *
 1                                      I           *
 1.5                                    I              *
 2                                      I           *
 2.5                                    I     *
 3                                      I*
 3.5                                    I  *
 4                                      I     *
 4.5                                    I        *
 5                                      I           *
 5.5                                    I              *
 6                                      I                 *
 6.5                                    I  *
 7                                      I        *

NOTE: THE SIX I'S ON THE HORIZONTAL Y-AXIS REPRESENT:
-3 , -2 , -1 , 0 , 1 , 2

READY

220 DEF FNY(X)=X
230 LET A=-1
240 LET B=9
250 LET I=1
RUN
```

Math
PLOTTR



NOTE: THE SIX I'S ON THE HORIZONTAL Y-AXIS REPRESENT:
-1 , 1 , 3 , 5 , 7 , 9

READY

1

```
1 GO TO 220
220 DEF FNY(X)=X+LOG(2*(SIN(X))↑2)-1.5*COS(X/2)
230 LET A=0.5
240 LET B=3.1
250LET I=0.1
RUN
```



NOTE: THE SIX I'S ON THE HORIZONTAL Y-AXIS REPRESENT:
-3 ,-2 ,-1 , 0 , 1 , 2

```
READY

1 GO TO 220
220 DEF FNY(X)=LOG((SIN(X))↑2)
230 LET A=1
240 LET B=3
250 LET I=0.1
RUN

X
-
V
A
L
U
E
S                    - - - - Y-VALUES - - - -
               -5             -3.5            -2             -.5
                I---------I----+----I---------I---------I----+----I
 1                                                              *  I
 1.1                                                           *  I
 1.2                                                           *I
 1.3                                                           *I
 1.4                                                           *
 1.5                                                           *
 1.6                                                           *
 1.7                                                           *
 1.8                                                           *I
 1.9                                                           *I
 2                                                             *  I
 2.1                                                          *   I
 2.2                                                         *    I
 2.3                                                       *      I
 2.4                                                     *        I
 2.5                                                   *          I
 2.6                                                 *            I
 2.7                                               *              I
 2.8                                             *                I
 2.9                                         *                    I
 3                                     *                          I

NOTE: THE SIX I'S ON THE HORIZONTAL Y-AXIS REPRESENT:
-5 ,-4 ,-3 ,-2 ,-1 , 0


READY
```

```
110 REM QUENTIN J. O'CONNOR, CONMACK H.S.NORTH, REVISED JULY,1969
115 REM  REVISED BY C.LOSIK  8-7-70
116 REM  A,B,I ARE SELF-EXPLANATORY
117 REM AUTOMATIC SCALING AND A FLOATING AXIS ARE USED.
120PRINT"    THIS PROGRAM WILL GRAPH A FUNCTION OF X BETWEEN ANY LIMITS"
130PRINT"(A AND B) YOU CHOOSE, WITH AN INTERVAL OF YOUR CHOICE (I)"
140PRINT"BETWEEN SUCCESSIVE VALUES OF X. IF YOU TYPE THE FOLLOWING:"
150 PRINT"          1 GO TO 220"
160 PRINT"          220 DEF FNY(X)=...(YOUR FUNCTION OF X)..."
170 PRINT"          230 LET A=...(YOUR SMALLER LIMIT OF X)..."
180 PRINT"          240 LET B=...(YOUR LARGER LIMIT OF X)..."
190 PRINT"          250 LET I=...(YOUR X-INCREMENT)..."
200 PRINT"          RUN"
210 STOP
220 DEF FNY(X)=X
230 LET A=-I
240 LET B=9
248 IF A<B THEN 250
244 PRINT "YOUR 'A' MUST BE LESS THAN YOUR 'B'."
246 STOP
250 LET I=1
260 LET L=FNY(A)
270 LET U=FNY(A)
280 FOR X=A TO B STEP I
290 LET Y=FNY(X)
300 IF Y-L<0 THEN 320
310GOTO 330
320 GOSUB360
330 IF Y-U>0 THEN 350
340 GO TO 360
350 GOSUB 400
360 NEXT X
370 GO TO 480
380 LET L=Y
390 RETURN
400 LET U=Y
410 RETURN
420 IF INT(U)-U=0 THEN 450
430 LET U1=INT(U)+1
440 GO TO 460
450 LET U1=U
460 LET L1=INT(L)
470 LET D=U1-L1
480 IF INT(D/5)-D/5=0 THEN 570
490 FOR K=1 TO 8
500 LET L1=L1-1
510 LET D=U1-L1
520 IF INT(D/5)-D/5=0 THEN 570
530 LET U1=U1+I
540 LET D=U1-L1
550 IF INT(D/5)-D/5=0 THEN 570
560 NEXT K
570 LET E=D/5
580 PRINT "X"
590 PRINT "-"
600 PRINT "V"
610 PRINT "A"
620 PRINT "L"
630 PRINT "U"
640 PRINT "E"
650 PRINT "S                    - - - - Y-VALUES - - - -"
660PRINT"              "L1," "L1+E*1.5," "L1+E*3,"   "L1+E*4.5
670PRINT"              I----------I----------I----------I----------I"
```

```
671 PRINT "-----+-----!"
680 IF L1>0 THEN 1020
690 LET Q=INT((-L1)*(10/E)+.5)
700 DEF FNP(Y)=INT((Y-L1)*(10/E)+.5)
710FOR X=A TO B STEP I
720 PRINTX,
730 LET Y=FNY(X)
740 IF Y>=0 THEN 830
760 PRINT TAB(15+FNP(Y));"*";TAB(Q+15);"!"
840 GO TO 1000
850 IF Y>0 THEN 910
860 PRINT TAB(Q+15);"*"
900 GO TO 1000
910 PRINT TAB(15+Q);"!";TAB(15+FNP(Y));"*"
1000 NEXT X
1010 GO TO 1100
1020 FOR X=A TO B STEP I
1030 PRINTX,
1040 LET Y= FNY(X)
1050 PRINT TAB(INT((Y-L1)*(10/E)+.5)+14);"*"
1090 NEXT X
1100 PRINT
1110 PRINT "NOTE: THE SIX I'S ON THE HORIZONTAL Y-AXIS REPRESENT:"
1120 PRINT L1;",";L1+E;",";L1+E*2;",";L1+E*3;",";L1+E*4;",";L1+E*5
1240 END
```

DISCIPLINE MATHEMATICS, GEN. 9th YR.

SUBJECT PRIME FACTOR

PROGRAM NAME PRIFA

DESCRIPTION:

This program finds the prime factors of any given integer, or prints " is prime" if the integer has no proper divisors.

OBJECTIVES:

A. To display to the student the prime factors of a large number of integers, giving the students a chance to discover relationships.

B. To use the motivation of the computer to teach the method that the program uses to find the prime factors.

PRELIMINARY PREPARATION:

A. Student - Should understand the meaning of composite, prime, factor, and prime factor.

B. Materials - If you desire to use this program with a group, a means by which the output can be displayed is necessary.

DISCUSSION:

The speed with which the computer operates in this program gives the student an opportunity to make generalizations based upon many more observations then heretofore was possible. The question can be asked: " By what method does the computer find the prime factor?" A flow chart would be highly useful at this point, not only in developing the method for finding a prime factor, but also in understanding the mathematical logic behind this method.

```
THIS PROGRAM WILL GIVE YOU THE PRIME FACTORS OF ANY
WHOLE NUMBER.  IF YOU WISH TO STOP THE PROGRAM, ENTER A
ZERO FOR THE NUMBER.

WHAT IS THE NUMBER ? 105

 105            3  5  7

WHAT IS THE NUMBER ? 72

 72             2  2  2  3  3

WHAT IS THE NUMBER ? 89

 89          IS PRIME

WHAT IS THE NUMBER ? 47

 47          IS PRIME

WHAT IS THE NUMBER ? 155

 155            5  31

WHAT IS THE NUMBER ? 362

 362            2  181

WHAT IS THE NUMBER ? 0

READY
```

```
100 REM W. TEPPER WYANDANCH H.S.
105 REM  REVISED BY C.LOSIK  8-10-70
106 REM  M IS THE NUMBER, A(I) ARE ITS FACTORS
110 REM ADAPTATION OF TWO PROGRAMS
120 REM THIS PROGRAM FINDS THE PRIME FACTORS OF ANY GIVEN INTEGER
130 REM AND PRINTS PRIME IF IT HAS NO PROPER DIVISORS
140 DIM A(100)
150 LET C=0
160 PRINT "THIS PROGRAM WILL GIVE YOU THE PRIME FACTORS OF ANY"
170 PRINT "WHOLE NUMBER.  IF YOU WISH TO STOP THE PROGRAM, ENTER A"
172 PRINT "ZERO FOR THE NUMBER."
174 PRINT
180 PRINT "WHAT IS THE NUMBER ";
190 LET X=0
200 INPUT M
205 IF ABS(M-INT(M+.5))<.0001 THEN 210
206 PRINT "WHOLE NUMBERS ONLY, PLEASE."
207 GO TO 180
210 PRINT
215 IF M<=0 THEN 470
220 PRINT M,
230 LET I=1
240 LET I=I+1
245 IF I>M THEN 310
250 IF M/I<>INT(M/I) THEN 240
260 LET X=X+1
270 LET A(X)=I
280 LET M=M/I
300 GO TO 250
310 IF X=1 THEN 360
320 FOR L=1 TO X
330 PRINT A(L);
340 NEXT L
350 GO TO 370
360 PRINT "IS PRIME"
370 PRINT
380 PRINT
385 GO TO 180
400 INPUT B
410 IF B=1 THEN 180
420 IF B=0 THEN 470
430 PRINT " TYPE 1 OR 0 AS INSTRUCTED"
440 LET C=C+1
460 GO TO 400
470 END
```

DISCIPLINE MATHEMATICS 12, 13

SUBJECT ANALYTIC GEOMETRY

PROGRAM QUADRT

DESCRIPTION:

This program determines the nature of the graph of $Ax^2 + Bxy + Cy^2 + Dx + Ey + F = 0$, after the operator inputs the six constants (A, B, C, D, E, F). Limiting cases, such as a point or a line, are separated from the general cases so that the computer type-out gives the exact nature of the graph.

OBJECTIVE:

To permit exploration of the properties of the second-degree equation.

PRELIMINARY PREPARATION:

A. Student - should have a reasonable knowledge of conic sections, second - degree equations in two unknowns, invariant functions of the coefficients under transformations, etc.

B. Materials - An overhead projector along with a transparency of the flow chart would be desirable,

DISCUSSION:

Before running the program, the teacher should discuss the general form of a second-degree equation in two variables, the functions of the coefficients used in the program, and the implications of the flow chart.

The discussion of the flow chart for this program enhances the understanding of the problem.

The type-out serves as a check on students' efforts in identifying second-degree equations.

```
    THIS PROGRAM DETERMINES THE NATURE OF THE GRAPH OF:
        A*X↑2+B*X*Y+C*Y↑2+D*X+E*Y+F=0
    ENTER YOUR CONSTANTS IN THE ORDER LISTED ABOVE.

? 5,8,5,0,0,0
THE GRAPH OF YOUR EQUATION IS A SINGLE POINT.

ANOTHER RUN (1=YES, 0=NO) : ? 1

    ENTER YOUR CONSTANTS IN THE ORDER LISTED ABOVE.

? 5,8,5,0,0,36
THERE IS NO REAL LOCUS FOR YOUR EQUATION.

ANOTHER RUN (1=YES, 0=NO) : ? 1

    ENTER YOUR CONSTANTS INTTHE ORDER LISTED ABOVE.

? 0,5,0,0,0,9
THE GRAPH /F YOUR EQUATION IS A HYPERBOLA.

ANOTHER RUN (1=YES, 0=NO) : ? 0


READY
```

```
100 REM QUENTIN J.O'CONNOR, CONMACK H.S.NORTH, JULY 16, 1969
103 REM  REVISED BY C.LOSIK  8-7-70
105 REM  A,B,C,D,E,F ARE AS IN EQUATION
110 PRINT "     THIS PROGRAM DETERMINES THE NATURE OF THE GRAPH OF:"
120 PRINT "          A*X↑2+B*X*Y+C*Y↑2+D*X+E*Y+F=0"
130 PRINT "     ENTER YOUR CONSTANTS IN THE ORDER LISTED ABOVE."
140 PRINT " "
150 INPUT A,B,C,D,E,F
160 IF A*A+B*B+C*C+D*D+E*E+F*F>0 THEN 210
170 PRINT"     WITH ALL YOUR CONSTANTS EQUAL TO ZERO, ANY VALUES OF X"
180 PRINT"AND Y WILL SATISFY YOUR EQUATION. IN OTHER WORDS, YOUR"
190 PRINT"GRAPH IS A COMPLETE PLANE."
200 GO TO 550
210 IF A*A+B*B+C*C+D*D+E*E=0 THEN 500
220 IF A*A+B*B+C*C>0 THEN 240
230 GO TO 400
240 LET I=A+C
250 LET K=4*A*C-B*B
260 LET J=4*A*C+A*C*F+4*A*F-E*E-D*D-B*B
270 LET P=4*A*C*F+B*D*E-A*E*E-C*D*D-F*B*B
280 IF P=0 THEN 360
290 IF K=0 THEN 540
300 IF K<0 THEN 520
310 IF I*P>0 THEN 500
320 IF A<>C THEN 340
330 IF B=0 THEN 460
340 PRINT "THE GRAPH OF YOUR EQUATION IS AN ELLIPSE."
350 GO TO 550
360 IF K>0 THEN 460
370 IF K<0 THEN 440
380 IF J<0 THEN 480
390 IF J>0 THEN500
400 PRINT "THE GRAPH OF YOUR EQUATION IS A SINGLE STRAIGHT LINE."
410 GO TO 550
420 PRINT "THE GRAPH OF YOUR EQUATION CONSISTS OF 2 PARALLEL LINES."
430 GO TO 550
440 PRINT "THE GRAPH OF YOUR EQUATION CONSISTS OF 2 INTERSECTING LINES."
450 GO TO 550
460 PRINT "THE GRAPH OF YOUR EQUATION IS A SINGLE POINT."
470 GO TO 550
480 PRINT "THE GRAPH OF YOUR EQUATION IS A CIRCLE."
490 GO TO 550
500 PRINT "THERE IS NO REAL LOCUS FOR YOUR EQUATION."
510 GO TO 550
520 PRINT "THE GRAPH OF YOUR EQUATION IS A HYPERBOLA."
530 GO TO 550
540 PRINT "THE GRAPH OF YOUR EQUATION IS A PARABOLA."
550 PRINT
560 PRINT "ANOTHER RUN (1=YES, 0=NO) : ";
570 INPUT A
575 PRINT
580 IF A=1 THEN 130
590 IF A<>0 THEN 560
600 END
```

DISCIPLINE MATHEMATICS 9th YEAR

SUBJECT PROPORTIONS

PROGRAM NAME RATIO

DESCRIPTION:

This program solves a proportion of the type A/B = C/D. A, B, C, or D can be unknown.

OBJECTIVES:

A. To teach the student(s) the relationships in a proportion.

B. To aid in teaching the solution of proportions.

PRELIMINARY PREPARATION:

A. Student - no particular preparation necessary

B. Materials - see discussion

DISCUSSION:

The student is given the opportunity to see any number of solutions to proportions. The program then asks a series of questions designed to allow the student to discover that in a proportion, the product of the means equals the product of the extremes. The program can be used either with individual students or with an entire class depending upon the availability of equipment to display the output. The running time varies, depending upon the number of proportions you wish to solve. In 10 to 15 minutes, the program can be run with about 100 proportion problems. Included in this time is a built-in variable pause for observation of the tabulated results. Another value of using this program is that the teacher can easily handle numbers in proportions that heretofore were too difficult.

```
THIS PROGRAM SOLVES FOR THE UNKNOWN IN THE PROPORTION
A/B AS C/D. USE A ZERO AS A DUMMY VALUE FOR THE UNKNOWN.

HOW MANY PROPORTIONS DO YOU WISH TO SOLVE? 4
WHAT ARE THE VALUES FOR A,B,C,D? 3,5,6,9
YOU FORGOT TO INPUT A ZERO FOR YOUR
UNKNOWN.  TRY AGAIN.? 3,4,6,0
                                        3 / 4  AS  6 / 8

WHAT ARE THE VALUES FOR A,B,C,D? 1,10,0,50
                                        1 / 10  AS  5 / 50

WHAT ARE THE VALUES FOR A,B,C,D? 36,0,1,2
                                        36 / 72  AS  1 / 2

WHAT ARE THE VALUES FOR A,B,C,D? 0,45,3,5
                                        27 / 45  AS  3 / 5


TAKE A GOOD LOOK AT THE PROPORTIONS.  THE TWO MIDDLE
POSITION NUMBERS ARE CALLED THE 'MEANS'; THE TWO
END POSITION NUMBERS ARE CALLED THE 'EXTREMES'.

LOOK AT THE 'MEANS' AND THE 'EXTREMES'  -    SEE IF
YOU CAN FIND SOME KIND OF RELATIONSHIP BETWEEN THEM.
WHEN YOU THINK YOU HAVE FOUND A RELATIONSHIP BETWEEN
THE 'MEANS' AND THE 'EXTREMES', TYPE 1 AND HIT THE RETURN KEY.
? 1


DID YOU SEE THAT IF YOU MULTIPLY THE 'MEANS'
AND MULTIPLY THE 'EXTREMES', THE PRODUCTS ARE EQUAL?

IN THE LAST PROPORTION  45 X 3 EQUALS 27 X 5
CHECK THE OTHERS, TOO.  WHEN YOU ARE READY TO CONTINUE,
TYPE 1 AND HIT THE RETURN KEY.
? 1


IF YOU WISH TO USE THIS PROGRAM AGAIN TYPE 1. IF NOT TYPE 0
? 0

READY
```

```
100 REM W. TEPPER WYANDANCH H.S.    -    MATHEMATICS
105 REM  REVISED BY C.LOSIK  8-5-70
106 REM  A/B = C/D, TOTALLY OBVIOUS.  ALSO USES GOSUB TO SIMULATE PAUSE
110 REM THIS PROGRAM SOLVES FOR THE UNKNOWN IN THE PROPORTION"
120 REM OF THE TYPE A/B AS C/D
130 PRINT "THIS PROGRAM SOLVES FOR THE UNKNOWN IN THE PROPORTION"
140 PRINT "A/B AS C/D. USE A ZERO AS A DUMMY VALUE FOR THE UNKNOWN."
150 PRINT
160 PRINT "HOW MANY PROPORTIONS DO YOU WISH TO SOLVE";
170 INPUT N
180 FOR K=1 TO N
190 PRINT "WHAT ARE THE VALUES FOR A,B,C,D";
200 INPUT A,B,C,D
210 IF A=0 THEN 270
220 IF B=0 THEN 290
230 IF C=0 THEN 310
240 IF D=0 THEN 330
250 PRINT "YOU FORGOT TO INPUT A ZERO FOR YOUR"
255 PRINT "UNKNOWN.  TRY AGAIN.";
260 GO TO 200
270 LET A=B*C/D
280 GO TO 340
290 LET B=A*D/C
300 GO TO 340
310 LET C=A*D/B
320 GO TO 340
330 LET D=B*C/A
340PRINT"                                                    ";A"/"B" AS "C"/"D
345 NEXT K
350 PRINT
360 PRINT
370 PRINT "TAKE A GOOD LOOK AT THE PROPORTIONS.  THE TWO MIDDLE"
380 PRINT "POSITION NUMBERS ARE CALLED THE 'MEANS', THE TWO"
390 PRINT "END POSITION NUMBERS ARE CALLED THE 'EXTREMES'."
395 PRINT
400 PRINT "LOOK AT THE 'MEANS' AND THE 'EXTREMES'  -    SEE IF"
410 PRINT "YOU CAN FIND SOME KIND OF RELATIONSHIP BETWEEN THEM."
420 PRINT "WHEN YOU THINK YOU HAVE FOUND A RELATIONSHIP BETWEEN"
430 PRINT "THE 'MEANS' AND THE 'EXTREMES', ";
450 GO SUB 610
460 PRINT "DID YOU SEE THAT IF YOU MULTIPLY THE 'MEANS'"
470 PRINT "AND MULTIPLY THE 'EXTREMES', THE PRODUCTS ARE EQUAL?"
475 PRINT
480 PRINT "IN THE LAST PROPORTION "B"X"C"EQUALS" A"X"D
490 PRINT "CHECK THE OTHERS, TOO.  WHEN YOU ARE READY TO CONTINUE,"
510 GO SUB 610
540 PRINT
550 PRINT "IF YOU WISH TO USE THIS PROGRAM AGAIN TYPE 1, IF NOT TYPE 0"
560 INPUT X
570 IF X=1 THEN 800
580 IF X=0 THEN 640
590 PRINT "TYPE 1 OR 0 AS DIRECTED."
600 GO TO 560
610 PRINT "TYPE 1 AND HIT THE RETURN KEY."
620 INPUT X
623 IF X<>1 THEN 620
625 PRINT
627 PRINT
630 PRINT
635 RETURN
640 END
```

DISCIPLINE MATHEMATICS

SUBJECT QUADRATIC EQUATIONS

PROGRAM NAME ROOTS2

DESCRIPTION:

This program describes the nature of the roots of a quadratic equation, and finds the roots whether real or complex.

OBJECTIVES:

A. To familiarize the student with quadratic function.

B. To review and drill exercise... to study the nature of roots.

C. To emphasize that roots of $f(x) = 0$ are the same as x-intercepts of $f(x) = y$.

D. To impress the student with geometric interpretation(s) of the nature of roots.

E. To provide "lead-in" material for the introduction of further study of the real number line, the real cartesian plane, complex numbers, quadratic inequalities, etc.

PRELIMINARY PREPARATION:

A. Student - The teacher can use the program to introduce the students to the quadratic formula, to conclude discussion of the quadratic formula... or both.

B. Materials - none

DISCUSSION:

The program uses the "discriminant" to determine the nature of the roots of the quadratic equation. Regardless of the nature of the roots, the student is asked to graph $y = F(x)$, and to compare his graph with the kind of roots he finds for a specific $F(x) = 0$. He should be impressed with the picture; and he should understand (ultimately) the reasonableness and validity of the analytic methods presented in class.

```
THIS PROGRAM HANDLES ALL POSSIBLE CASES OF SOLUTION OF THE EQUATION :

     A * X ↑ 2  +  B * X  +  C  =  0

TYPE IN YOUR VALUES FOR A, B, AND C : ? 1,2,3

DISCRIMINANT IS LESS THAN ZERO, SO ROOTS ARE IMAGINARY.
THEY ARE OF THE FORM :  P+I*Q , P-I*Q , WHERE :
P =-1          Q = 1.414214

                                   ***


DO YOU WANT ANOTHER RUN ( 0 = NO , 1 = YES ) : ? 1

TYPE IN YOUR VALUES FOR A, B, AND C : ? 1,7,3

DISCRIMINANT IS GREATER THAN ZERO, SO ROOTS ARE REAL.
ROOTS ARE X1 AND X2 .
X1 =-.4586187 X2 =-6.541381

                                   ***


DO YOU WANT ANOTHER RUN ( 0 = NO , 1 = YES ) : ? 1

TYPE IN YOUR VALUES FOR A, B, AND C : ? 1,6,9

DISCRIMINANT IS EQUAL TO ZERO, SO ROOTS ARE EQUAL.  X =-3

                                   ***


DO YOU WANT ANOTHER RUN ( 0 = NO , 1 = YES ) : ? 1

TYPE IN YOUR VALUES FOR A, B, AND C : ? 2,5,6

DISCRIMINANT IS GREATER THAN ZERO, SO ROOTS ARE REAL.
ROOTS ARE X1 AND X2 .
X1 =-1         X2 =-3

                                   ***


DO YOU WANT ANOTHER RUN ( 0 = NO , 1 = YES ) : ? 0


READY
```

```
100 REM  THE ULTIMATE QUADRATIC SOLVER, UNTIL THE NEXT VERSION
110 REM  CHARLES LOSIK, PIB, 7/21/70, BASIC
120 PRINT "THIS PROGRAM HANDLES ALL POSSIBLE CASES OF SOLUTION OF"
125 PRINT " THE EQUATION :"
130 PRINT
140 PRINT "     A * X ↑ 2  +  B * X  +  C  =  0"
150 PRINT
160 PRINT "TYPE IN YOUR VALUES FOR A, B, AND C : "
165 REM  INPUT VALUES FOR A,B,C
170 INPUT A,B,C
171 PRINT
175 REM  FOR ALL CASES, CHECK A=0. IF SO, THEN LINEARITY
180 IF A=0 THEN 802
185 REM  D IS THE DISCRIMINANT
190 LET D=B*B-4*A*C
195 LET Z=2*A
200 IF D=0 THEN 710
210 IF D>0 THEN 610
300 REM  D<0, IMAGINARY RESULTS
310 PRINT "DISCRIMINANT IS LESS THAN ZERO, SO ROOTS ARE IMAGINARY."
320 PRINT "THEY ARE OF THE FORM :  P+I*Q ; P-I*Q , WHERE I :"
330 PRINT "P =";-B/Z,"Q =";SQR(ABS(D))/Z
340 GO TO 900
600 REM  D>0, SO REAL ROOTS
610 PRINT "DISCRIMINANT IS GREATER THAN ZERO, SO ROOTS ARE REAL."
620 PRINT "ROOTS ARE X1 AND X2 ."
630 PRINT "X1 =";(-B+SQR(D))/Z,"X2 =";(-B-SQR(D))/Z
640 GO TO 900
700 REM  EQUAL ROOTS (D=0)
710 PRINT "DISCRIMINANT IS EQUAL TO ZERO, SO ROOTS ARE EQUAL.  X =";-B/Z
720 GO TO 900
800 REM  A=0, SO X=-C/B, UNLESS B=0
802 IF B<>0 THEN 810
803 IF C=0 THEN 807
804 PRINT "MEANINGLESS STATEMENT."
806 GO TO 900
807 PRINT "OK, ZERO = ZERO."
808 GO TO 900
810 PRINT "THE EQUATION IS LINEAR.  X =";-C/B
900 PRINT
901 PRINT TAB(30);"***"
905 PRINT
910 PRINT
920 PRINT "DO YOU WANT ANOTHER RUN ( 0 = NO , 1 = YES ) : ";
930 INPUT Z
940 IF Z=1 THEN 150
950 IF Z<>0 THEN 920
999 END
```

DISCIPLINE MATHEMATICS - JR. HIGH

SUBJECT INTERSECTION AND UNION OF SETS

PROGRAM NAME SETS

DESCRIPTION:

This program finds the intersection and union of any two numerical sets.

OBJECTIVES:

A. To motivate students to find the union and intersection of any two sets.

B. To learn the logic involved in finding the union and intersection.

PRELIMINARY PREPARATION:

A. Student - no special preparation necessary.

B. Materials - see discussion

DISCUSSION:

This program may be used with individuals, small groups, or class-size groups. The elements of the two sets are entered as per instructions. Incidently, one or both of the sets may be empty. The computer then types back the elements in the union and intersection. The speed with which the computer operates enables the students to see a great many examples, giving them the opportunity to make discoveries about what is the union and what is an intersection of two sets. The teacher may use the flow chart that follows to explain the logic behind finding the union and intersection.

It is suggested that when used with large groups, a supplementary device be used to display output.

```
THIS PROGRAM FINDS THE UNION AND INTERSECTION OF ANY TWO
NUMERICAL SETS.

  HOW MANY ELEMENTS IN THE FIRST SET? 5

THESE ARE - (HIT THE RETURN KEY AFTER ENTERING EACH ELEMENT).
? 1
? 2
? 3
? 4
? 5

  HOW MANY ELEMENTS IN THE SECOND SET? 5

THESE ARE:
? 2
? 4
? 6
? 8
? 10


THE INTERSECTION CONTAINS  2  4
THE UNION CONTAINS 2  4  6  8  10  1  3  5

DO YOU WANT ANOTHER RUN (1=YES, 0=NO) : ? 1

  HOW MANY ELEMENTS IN THE FIRST SET? 8

THESE ARE - (HIT THE RETURN KEY AFTER ENTERING EACH ELEMENT).
? 1
? 2
? 3
? 4
? 6
? 8
? 10
? 12

  HOW MANY ELEMENTS IN THE SECOND SET? 10

THESE ARE:
? 1
? 2
? 3
? 4
? 5
? 6
? 7
? 8
? 9
? 10


THE INTERSECTION CONTAINS  1  2  3  4  6  8  10
THE UNION CONTAINS 1  2  3  4  5  6  7  8  9  10  12

DO YOU WANT ANOTHER RUN (1=YES, 0=NO) : ? 0


READY
```

```
100REM W. TEPPER, WYANDANCH HS, 7/29/69
101 REM  REVISED BY C.LOSIK  8-10-70
103 DIM A(30),B(30)
110 REM  UP TO 30 ELEMENTS PER SET ARE ALLOWED(UNLESS DIM IS CHANGED)
120PRINT"THIS PROGRAM FINDS THE UNION AND INTERSECTION OF ANY TWO"
130PRINT"NUMERICAL SETS."
140PRINT
150PRINT"  HOW MANY ELEMENTS IN THE FIRST SET";
160INPUT N
163 IF N=0 THEN 230
166 IF N=INT(N) THEN 170
167 PRINT "ONLY AN INTEGER NUMBER OF ELEMENTS IS POSSIBLE."
169 GO TO 140
170 IF N<=30 THEN 180
173 PRINT "THE MACHINE CANNOT HOLD MORE THAN 30 ELEMENTS.";
175 PRINT "  SEE YOUR TEACHER."
177 GO TO 690
180 IF N>0 THEN 189
183 PRINT "THERE CANNOT BE A NEGATIVE NUMBER OF ELEMENTS."
186 GO TO 140
189 PRINT
190PRINT"THESE ARE - (HIT THE RETURN KEY AFTER ENTERING EACH ELEMENT)."
200FOR K=1TON
210INPUTA(K)
220NEXT K
230PRINT
240 PRINT "  HOW MANY ELEMENTS IN THE SECOND SET";
250INPUT J
253 IF J=0 THEN 550
256 IF J=INT(J) THEN 260
257 PRINT "ONLY AN INTEGER NUMBER OF ELEMENTS IS POSSIBLE."
259 GO TO 230
260 IF J>30 THEN 173
270 IF J>0 THEN 279
273 PRINT "THERE CANNOT BE A NEGATIVE NUMBER OF ELEMENTS."
276 GO TO 230
279 PRINT
280 PRINT"THESE ARE:"
290 FOR K1=1TO J
300 INPUT B(K1)
310 NEXT K1
311 PRINT
312 PRINT
315 IF N<=0 THEN 640
320 PRINT "THE INTERSECTION CONTAINS ";
330 FOR K=1 TO N
340 FOR L=1 TO J
```

```
350 IF A(K)=B(L)THEN 380
360 NEXT L
370 GO TO 400
380 PRINT A(K);
390 LET X=X+1
400 NEXT K
410 IF X>0THEN 430
420 PRINT"  EMPTY SET....NO ELEMENTS"
430PRINT
440 PRINT"THE UNION CONTAINS";
450 FOR L=1 TO J
460 PRINT B(L);
470 NEXT L
480 FOR K=1 TO N
490 FOR L=1 TO J
500 IF A(K)=B(L)THEN 530
510 NEXT L
520 PRINT A(K);
530 NEXT K
540 GO TO 690
550 IF N<=0 THEN 620
560 PRINT "INTERSECTION IS EMPTY"
570 PRINT "UNION CONTAINS";
580 FOR K=1 TO N
590 PRINT A(K);
600 NEXT K
610 GO TO 690
620 PRINT "UNION AND INTERSECTION ARE EMPTY"
630 GO TO 690
640 PRINT "INTERSECTION IS EMPTY"
650 PRINT "UNION CONTAINS";
660 FOR K=1 TO J
670 PRINT B(K);
680 NEXT K
690 PRINT
700 PRINT
720 PRINT "DO YOU WANT ANOTHER RUN (1=YES, 0=NO) : ";
730 INPUT N
740 IF N=1 THEN 140
750 IF N<>0 THEN 720
760 END
```

DISCIPLINE ALGEBRA

SUBJECT SIMULTANEOUS EQUATIONS

PROGRAM NAME SIMEQN

DESCRIPTION:

This program finds the simultaneous solution set for sets of simultaneous linear equations (up to 10x10)

OBJECTIVES:

1. To eliminate the tedium of solution of sets of simultaneous equations.

2. To provide a means for checking solutions obtained by other means.

PRELIMINARY PREPARATIONS:

Presentation of concepts of simultaneous equations and methods for finding solutions.

```
THIS PROGRAM SOLVES ANY NUMBER OF SETS OF SIMULTANEOUS
EQUATIONS OF UP TO 10 EQUATIONS PER SET.  ENTER YOUR SETS
OF EQUATIONS IN DATA STATEMENTS IN LINES 700-800,
PRECEDED BY THE NUMBER OF EQUATIONS IN EACH SET.
EXAMPLE:  TO SOLVE THE SYSTEM
  1*A(1) + 2*X(2) = 3
  4*A(1) + 9*X(2) = 10
ENTER DATA AS FOLLOWS:
  700 DATA 2
  701 DATA 1,2,3
  702 DATA 4,9,10
THEN TYPE:
  1 GO TO 110
  RUN
THE COMPUTER WILL PRINT A MATRIX OF YOUR EQUATIONS, FOLLOWED
BY THE SOLUTION TO THE EQUATIONS.


READY

700 DATA 2
701 DATA 1,2,3
702 DATA 4,9,10
1 GO TO 110
RUN

 1             2             3
 4             9             10

X( 1 )=        7
X( 2 )=       -2




READY

700 DATA 2
701 DATA 3,2,16
702 DATA -6,-4,-32
1 GO TO 110
RUN

 3             2            16
-6            -4           -32

NO UNIQUE SOLUTION
```

```
READY

700 DATA 3
701 DATA 3,2,5,10
702 DATA -1,4,7,-21
703 DATA 1,1,-1,14
1 GO TO 110
RUN

 3              2              5             10
-1              4              7            -21
 1              1             -1             14

X( 1 )=          7.413044
X( 2 )=          2.956522
X( 3 )=         -3.630435


READY
```

```
10 REMARK  D.SOBIN, BKLYN POLY, 11-69
15 REM  REVISED BY C.LOSIK, 9-25-70
20 PRINT "THIS PROGRAM SOLVES ANY NUMBER OF SETS OF SIMULTANEOUS"
25 PRINT "EQUATIONS OF UP TO 10 EQUATIONS PER SET.  ENTER YOUR SETS"
30 PRINT "OF EQUATIONS IN DATA STATEMENTS IN LINES 700-800,"
35 PRINT "PRECEDED BY THE NUMBER OF EQUATIONS IN EACH SET."
40 PRINT "EXAMPLE:  TO SOLVE THE SYSTEM"
45 PRINT "  1*X(1) + 2*X(2) = 3"
50 PRINT "  4*X(1) + 9*X(2) = 10"
60 PRINT "ENTER DATA AS FOLLOWS:"
62 PRINT "  700 DATA 2"
64 PRINT "  701 DATA 1,2,3"
66 PRINT "  702 DATA 4,9,10"
70 PRINT "THEN TYPE:"
72 PRINT"  1 GO TO 110"
74 PRINT "  RUN"
80 PRINT "THE COMPUTER WILL PRINT A MATRIX OF YOUR EQUATIONS, FOLLOWED"
85 PRINT "BY THE SOLUTION TO THE EQUATIONS."
90 STOP
100 DIM E(10,11), X(10)
110 READ N
120 IF N=0 THEN 999
130 FOR I=1 TO N
140      FOR K=1 TO N+1
150 READ E(I,K)
155 PRINT E(I,K),
160      NEXT K
165 PRINT " "
170 NEXT I
185 REMARK  EVALUATE MATRIX
190 FOR J=1 TO N-1
200 IF E(J,J)=0 THEN 560
210 FOR I=J+1 TO N
220 LET Q=E(I,J)/E(J,J)
230 FOR K=J TO N+1
240 LET E(I,K)=E(I,K)-E(J,K)*Q
250 NEXT K
260 NEXT I
270 NEXT J
340 REMARK SOLVE FOR X(N)
350 IF E(N,N)=0 THEN 520
360 LET I=N+1
370 LET X(N)=E(N,I)/E(N,N)
380      FOR J=1 TO N-1
390      LET S=0
400        FOR K=1 TO J
410        LET S=S+E(N-J,I-K)*X(I-K)
```

```
420         NEXT K
430         LET X(N-J)=(E(N-J,I)-S)/E(N-J,N-J)
440       NEXT J
450 REMARK PRINT VALUES
455 PRINT
460 FOR J=1 TO N
470 PRINT "X("J")=",X(J)
480 NEXT J
500 GO TO 530
520 PRINT
525 PRINT "NO UNIQUE SOLUTION"
530 PRINT
535 PRINT
540 PRINT
550 GOTO 110
560 FOR T= J+1 TO N
570 IF E(T,J)<>0 THEN 600
580 NEXT T
590 GOTO 520
600 FOR C=J TO N+1
610 LET A=E(J,C)
620 LET E(J,C)=E(T,C)
630 LET E(T,C)=A
640 NEXT C
650 GOTO 210
801 DATA 0
999 END
```

DISCIPLINE CALCULUS - GRADE 13

SUBJECT TANGENT SLOPE FOR ANY FUNCTION

PROGRAM NAME SLOPE

DESCRIPTION:

This program considers a function which is differentiable at $x = a$, and at all points in the interval $[a, a+1]$. The value of the derivative at $x = a$ is approximated through secant slopes.

OBJECTIVES:

A. The preliminary discussion of the method whereby the machine solves the problem enhances the students' comprehension of the techniques. These techniques are then used in developing the analytic method for finding the slope of the tangent line.

B. The type-out of successive approximations to the tangent slope clarifies and dramatizes the nature of the limiting processes.

C. Time-saving factor through the elimination of lengthy computations.

PRELIMINARY PREPARATION:

Materials

The diagram below may be shown to the students on a blackboard, or an overhead projector, to explain the computations geometrically.



DISCUSSION:

The use of the computer and the attendant discussion of the program dramatically introduces the idea of differentiation.

```
          SECANT SLOPE OF A CURVE - THE DERIVATIVE

THIS PROGRAM CONSIDERS A FUNCTION OF X (Y=F(X)) WHICH IS
DIFFERENTIABLE AT X=A AND AT ALL POINTS IN THE INTERVAL
(A,A+1).  THE VALUE OF THE DERIVATIVE AT X=A IS
APPROXIMATED THROUGH SECANT SLOPES.

AFTER THE PROGRAM STOPS, TYPE IN THE FOLLOWING:
(END EACH LINE, INCLUDING 'RUN', WITH A 'CARRIAGE RETURN')

          1 GO TO 300
          300 DEF FNY(X)=....(YOUR FUNCTION OF X)....
          RUN

FOR EXAMPLE, TO FIND THE SLOPE OF THE EQUATION Y=X↑3
YOU WOULD TYPE AS FOLLOWS:

          1 GO TO 300
          300 DEF FNY(X)=X↑3
          RUN

YOU MIGHT TRY THAT AS YOUR FIRST RUN.
FOR SUBSEQUENT RUNS, YOU NEED ONLY CHANGE LINE 300 FOR
A NEW FUNCTION, FOLLOWED BY 'RUN'.


READY

1 GO TO 300
300 DEF FNY(X)=X↑3
RUN

FOR WHAT VALUE OF A IS THE SLOPE TO BE EVALUATED? 2

'CHANGE IN X' IS THE DISTANCE FROM 'A', AND 'CHANGE IN Y'
IS THE DISTANCE FROM 'F(A)' UPON WHICH THE SLOPE IS CALCULATED.


CHANGE IN X    CHANGE IN Y    SECANT SLOPE   % CHANGE IN SLOPE
-----------    -----------    ------------   -----------------
 1             19             19             NO PREVIOUS VALUE
 .5            7.625          15.25          19.73684
 .25           3.390625       13.5625        11.06557
 .125          1.595703       12.76562       5.875576
 .0625         .7736816       12.37891       3.089376
 .03125        .3808899       12.18845       1.53634
 .015625       .1889687       12.09399       .7751783
 .0078125      .09411669      12.04694       .3891031
 .00390625     .04698667      12.02347       .1946049
 .001953125    .02346039      12.01172       .09771946
 .0009765625   .01172447      12.00586       .04878049
 .0004882813   5.860806E-3    12.00293       .02440815

*****

DO YOU WISH TO USE A DIFFERENT VALUE OF X (1=YES, 0=NO)? 0
TO CHANGE YOUR FUNCTION SEE THE INSTRUCTIONS.
IF YOU ARE FINISHED, TYPE '1', AND THE 'RETURN' KEY
AFTER THE PROGRAM STOPS.


READY

1
```

```
100 REM SECANT SLOPE OF A CURVE - G. J. O'CONNOR  8-12-68
101 REM REVISED 8-7-70 (D. PESSEL)  (COMBINATION OF SLOPE AND DIFFQ)
102 REM IMPORTANT VARIABLES:  S-SECANT SLOPE; P-PERCENT CHANGE;
103 REM D-CHANGE IN X; Y-CHANGE IN Y
105 LET S1=0
110 PRINT TAB(10);"SECANT SLOPE OF A CURVE - THE DERIVATIVE"
120 PRINT
130 PRINT "THIS PROGRAM CONSIDERS A FUNCTION OF X (Y=F(X)) WHICH IS"
131 PRINT "DIFFERENTIABLE AT X=A AND AT ALL POINTS IN THE INTERVAL"
132 PRINT "(A,A+1).  THE VALUE OF THE DERIVATIVE AT X=A IS"
133 PRINT "APPROXIMATED THROUGH SECANT SLOPES."
134 PRINT
139 PRINT "AFTER THE PROGRAM STOPS, TYPE IN THE FOLLOWING:"
140 PRINT "(END EACH LINE, INCLUDING 'RUN', WITH A 'CARRIAGE RETURN')"
141 PRINT
142 PRINT "          1 GO TO 300"
143 PRINT "          300 DEF FNY(X)=....(YOUR FUNCTION OF X)...."
145 PRINT "          RUN"
146 PRINT
147 PRINT "FOR EXAMPLE, TO FIND THE SLOPE OF THE EQUATION Y=X↑3"
148 PRINT "YOU WOULD TYPE AS FOLLOWS:"
149 PRINT
150 PRINT "          1 GO TO 300"
151 PRINT "          300 DEF FNY(X)=X↑3"
153 PRINT "          RUN"
154 PRINT
155 PRINT "YOU MIGHT TRY THAT AS YOUR FIRST RUN."
156 PRINT "FOR SUBSEQUENT RUNS, YOU NEED ONLY CHANGE LINE 300 FOR"
157 PRINT "A NEW FUNCTION, FOLLOWED BY 'RUN'."
160 STOP
299 REM CALCULATION OF SLOPE AND PRINTOUT
300 DEF FNY(X)=X↑3
305 PRINT "FOR WHAT VALUE OF A IS THE SLOPE TO BE EVALUATED";
306 INPUT A
310 PRINT
311 PRINT "'CHANGE IN X' IS THE DISTANCE FROM 'A', AND 'CHANGE IN Y'"
312 PRINT "IS THE DISTANCE FROM 'F(A)' UPON WHICH THE SLOPE IS CALCU";
313 PRINT "LATED."
316 PRINT
317 PRINT
320 PRINT "CHANGE IN X","CHANGE IN Y","SECANT SLOPE","% CHANGE IN SLOPE"
321 PRINT "------ -- -","------ -- -","------ -----","- ------ -- -----"
410 FOR N=0 TO 11
420 LET D=2↑N
430 LET Y=FNY(A+1/D)-FNY(A)
440 LET S=D*Y
444 IF S1>0 THEN 447
445 PRINT "1/"D,Y,S,"NO PREVIOUS VALUE"
446 GO TO 455
447 LET P=((ABS(S1-S))/S1)*100
450 PRINT "1/"D,Y,S,P
455 LET S1=S
460 NEXT N
470 PRINT
480 PRINT "*****"
490 PRINT
500 PRINT "DO YOU WISH TO USE A DIFFERENT VALUE OF X (1-YES, 0-NO)";
501 INPUT Q8
502 IF Q8>0 THEN 305
510 PRINT "TO CHANGE YOUR FUNCTION SEE THE INSTRUCTIONS."
520 PRINT "IF YOU ARE FINISHED, TYPE '1', AND THE 'RETURN' KEY"
530 PRINT "AFTER THE PROGRAM STOPS."
540 END
```

DISCIPLINE MATHEMATICS

SUBJECT ALGEBRA(9TH and 12TH GRADE)

PROGRAM NAME SQRT

DESCRIPTION:

This program finds the square root of counting numbers up to five decimal places.

OBJECTIVES:

A. To demonstrate and familiarize the students with square roots.

B. The method utilizes "pinching" $\sqrt{Z}$ between the endpoints of smaller and smaller domains.

PRELIMINARY PREPARATION:

A. Student - 1) The definition of square root as the inverse operation of squaring; and 2) Drill in estimating square roots to the nearest tenth, hundredth, etc.

B. Materials - none

DISCUSSION:

This program provides an "introduction to," and a "review of" evolution and involution. Limiting the neighborhood of $\sqrt{Z}$ to find successively closer approximations of the square root of a number, demonstrates to the student that he is able to determine the square root to any degree.

The program may be effectively utilized for introducing the limiting process.

```
PROGRAM FINDS SQUARE ROOT OF ANY POSITIVE NUMBER
BY 'PINCHING' IT WITHIN A SMALLER AND SMALLER INTERVAL.


WHAT IS THE NUMBER WHOSE SQUARE ROOT YOU SEEK? 54


LOWER LIMIT                                    UPPER LIMIT
----- -----                                    ----- -----
 0              < SQ.RT. OF 54 <                54
 5.4            < SQ.RT. OF 54 <                10.8
 7.02           < SQ.RT. OF 54 <                7.56
 7.344          < SQ.RT. OF 54 <                7.398
 7.344          < SQ.RT. OF 54 <                7.3494
 7.348319       < SQ.RT. OF 54 <                7.348659
 7.348427       < SQ.RT. OF 54 <                7.348481
 7.348465       < SQ.RT. OF 54 <                7.34847
 7.348469       < SQ.RT. OF 54 <                7.34847

APPROXIMATION NOW CORRECT TO AN ACCURACY OF 1.000000E-5
YOU MAY USE EITHER 7.348469 OR 7.34847 AS THE SQUARE ROOT OF 54


WANT TO TRY ANOTHER NUMBER (1=YES, 0=NO) : ? 1


WHAT IS THE NUMBER WHOSE SQUARE ROOT YOU SEEK? 39


LOWER LIMIT                                    UPPER LIMIT
----- -----                                    ----- -----
 0              < SQ.RT. OF 39 <                39
 3.9            < SQ.RT. OF 39 <                7.8
 6.24           < SQ.RT. OF 39 <                6.63
 6.24           < SQ.RT. OF 39 <                6.279
 6.2439         < SQ.RT. OF 39 <                6.2478
 6.24468        < SQ.RT. OF 39 <                6.24507
 6.244992       < SQ.RT. OF 39 <                6.245031
 6.244996       < SQ.RT. OF 39 <                6.245

APPROXIMATION NOW CORRECT TO AN ACCURACY OF 1.000000E-5
YOU MAY USE EITHER 6.244996 OR 6.245 AS THE SQUARE ROOT OF 39


WANT TO TRY ANOTHER NUMBER (1=YES, 0=NO) : ? 0


READY
```

```
100  REM  T. BURNS, JOHN GLENN HS,  8-6-69
110  REM  REVISED BY C.LOSIK   8-27-70
120  REM  A=LOWER LIMIT, B=UPPER LIMIT, Z=STEP IN INTERVAL
121  REM  E IS THE ACCURACY YOU DESIRE
125 LET E=.00001
130 PRINT "PROGRAM FINDS SQUARE ROOT OF ANY POSITIVE NUMBER"
140 PRINT "BY 'PINCHING' IT WITHIN A SMALLER AND SMALLER INTERVAL."
150 PRINT
160 PRINT
170 PRINT "WHAT IS THE NUMBER WHOSE SQUARE ROOT YOU SEEK";
180 INPUT Z
185 PRINT
190 IF Z>0 THEN 220
200 PRINT "YOUR NUMBER MUST BE POSITIVE !!!"
210 GO TO 160
220 PRINT
230 PRINT "LOWER LIMIT"," "," ","UPPER LIMIT"
235 PRINT "----- -----"," "," ","----- -----"
240 LET A=0
250 LET B=Z
260 LET S=(B-A)/10
270 PRINT A,"< SQ.RT. OF"Z"<",B
275 IF ABS(A*B-Z)<E THEN 360
280 FOR I=A TO B STEP S
290 IF Z<I*I THEN 310
300 NEXT I
301 LET B=B*10
302 GO TO 260
310 LET B=I
320 LET A=I-S
350 GO TO 260
360 PRINT
370 PRINT "APPROXIMATION NOW CORRECT TO AN ACCURACY OF"E
380 PRINT "YOU MAY USE EITHER"A"OR"B"AS THE SQUARE ROOT OF"Z
390 PRINT
400 PRINT
410 PRINT "WANT TO TRY ANOTHER NUMBER (1=YES, 0=NO) : ";
420 INPUT Z
430 IF Z=1 THEN 150
440 IF Z<>0 THEN 400
450 END
```

DISCIPLINE MATHEMATICS-TEACHER ASSISTANCE

SUBJECT ARITHMETIC MEAN (AVERAGE)

PROGRAM STATAL

DESCRIPTION:

This program finds the average (arithmetic mean), median, and standard deviation of up to one hundred numbers.

OBJECTIVES:

A. To familiarize the student with the concepts of arithmetic mean (average), median, and standard deviation of a group of numbers.

B. To impress him with the speed and accuracy of the computer as a calculating device.

C. To provide teachers with handy means of computing averages.

PRELIMINARY PREPARATION:

A. Student - "Arithmetic mean", "average", "median", and "standard deviation" must be well-defined.

B. Materials - None

DISCUSSION:

Given N terms, "A(1), A(2),..., A(N-1), A(N)", students will have learned the average of these N terms is "$\frac{A(1)+A(2)+...A(N-1)+A(N)}{N}$".

The program prints out the median value of the user's data when there is an odd number of data values. When there is an even number, the median value printed is the average between the N/2 and the (N+2)/2 terms.

The program serves as an excellent vehicle for drill in division and addition, and helps strengthen the concept of arithmetic mean (average).

This program is useful in demonstrating a simple "loop" routine for students interested in programming.

MEAN, MEDIAN, AND DEVIATION OF A SET OF NUMBERS.

ENTER YOUR NUMBERS IN DATA STATEMENTS ON LINES
1000 - 2000. FOR EXAMPLE, YOU MIGHT TYPE :

1000 DATA 1,2,3,4 ETC. (YOUR DATA GOES HERE)

WHEN YOUR DATA HAS BEEN ENTERED, TYPE :

1 GO TO 300
RUN

THEN RELAX WHILE THE MACHINE GRINDS OUT THE ANSWERS.

IF A 'SUBSCRIPT ERROR' APPEARS, INCREASE THE SIZE OF THE
ARRAY IN LINE 295.

WARNING -- THE NUMBER 9999 IS USED AS AN INTERNAL DATA
VALUE. IF THIS VALUE IS ONE OF YOUR DATA VALUES, SIMPLY
RE-TYPE LINES 999 AND 2001 WITH A COMMON DATA VALUE WHICH
YOU WILL NOT USE.

READY

1000 DATA 244,182,112,2,198,10,314,169,18,38
1 GO TO 300
RUN

THESE ARE YOUR NUMBERS :
244 182 112 2 198 10 314 169 18 38

THESE ARE YOUR NUMBERS (HIGHEST TO LOWEST) :
314 244 198 182 169 112 38 18 10 2

NUMBER OF VALUES IS 10
SUM OF THE VALUES IS 1287
THE MEAN VALUE IS 128.7
THE MEDIAN VALUE IS 140.5
THE STANDARD DEVIATION IS 209.5409

FOR ANOTHER RUN, RE-ENTER DATA ON LINES
1000 - 2000, TAKING CARE TO ELIMINATE OLD DATA
BY TYPING THOSE LINE NUMBERS WHICH YOU DO NOT USE AGAIN;
THEN TYPE 'RUN'.

READY

1

1000

```
100  REM  CHARLES M. LOSIK, BKLYN POLY, MEAN-MEDIAN-DEVIATION
110  REM  (7-66 IN FORTRAN II) ; (8-26-70 IN BASIC)
115 REM REVISED 9-24-70
120  REM  YOU PUT YOUR NUMBERS IN DATA STATEMENTS AND
130  REM  YOU GET WHAT YOU PAY FOR.
140 PRINT " ","MEAN, MEDIAN, AND DEVIATION OF A SET OF NUMBERS."
150 PRINT
160 PRINT " ENTER YOUR NUMBERS IN DATA STATEMENTS ON LINES"
170 PRINT " 1000 - 2000.  FOR EXAMPLE, YOU MIGHT TYPE :"
171 PRINT
172 PRINT " ","1000 DATA  1,2,3,4  ETC. (YOUR DATA GOES HERE)"
173 PRINT
174 PRINT " WHEN YOUR DATA HAS BEEN ENTERED,  TYPE :"
180 PRINT
190 PRINT " ","1 GO TO 300"
200 PRINT " ","RUN"
210 PRINT
220 PRINT " THEN RELAX WHILE THE MACHINE GRINDS OUT THE ANSWERS."
222 PRINT
225 PRINT " IF A 'SUBSCRIPT ERROR' APPEARS, INCREASE THE SIZE OF THE"
227 PRINT " ARRAY IN LINE 295."
230 PRINT
240  REM  A(I) ARE THE NUMBERS, S IS THEIR SUM,
250  REM  S2 IS THE SUM OF THEIR SQUARES.
260 REM
270 PRINT " WARNING -- THE NUMBER 9999 IS USED AS AN INTERNAL DATA"
275 PRINT " VALUE.  IF THIS VALUE IS ONE OF YOUR DATA VALUES, SIMPLY"
280 PRINT " RE-TYPE LINES 999 AND 2001 WITH A COMMON DATA VALUE WHICH"
285 PRINT " YOU WILL NOT USE."
290 STOP
295 DIM A(100)
300 PRINT
303 PRINT " THESE ARE YOUR NUMBERS :"
305 LET I=1
310 READ E
315 LET S=0
316 LET S2=0
320 READ A(I)
330 IF E = A(I) THEN 370
340 PRINT A(I) ;
345 LET S = S + A(I)
347 LET S2 = S2 + A(I) * A(I)
350 LET I = I + 1
360 GO TO 320
370 LET N = I - 1
380 PRINT
390 PRINT
399  REM  ***** BUBBLE SORT*****
400 PRINT " THESE ARE YOUR NUMBERS (HIGHEST TO LOWEST) :"
405 FOR I = 1 TO N - 1
```

```
 410 FOR J = I + 1 TO N
 420 IF A(I) > A(J) THEN 460
 430 LET T = A(I)
 440 LET A(I) = A(J)
 450 LET A(J) = T
 460 NEXT J
 465 PRINT A(I) ;
 470 NEXT I
 475 PRINT A(N)
 480 PRINT
 490 PRINT
 500 PRINT " NUMBER OF VALUES IS";N
 510 PRINT " SUM OF THE VALUES IS";S
 520 PRINT " THE MEAN VALUE IS" ; S / N
 530 PRINT " THE MEDIAN VALUE IS" ;
 540 IF N / 2 <> INT ( N / 2 ) THEN 570
 550 PRINT ( A(N/2) + A((N+2)/2))/2
 560 GO TO 600
 570 PRINT A((N+1)/2)
 600 PRINT " THE STANDARD DEVIATION IS" ; SQR ( N * S2 + S * S ) / N
 610 PRINT
 620 PRINT
 630 PRINT " FOR ANOTHER RUN, RE-ENTER DATA ON LINES"
 640 PRINT " 1000 - 2000, TAKING CARE TO ELIMINATE OLD DATA"
 642 PRINT " BY TYPING THOSE LINE NUMBERS WHICH YOU DO NOT USE AGAIN;"
 645 PRINT " THEN TYPE 'RUN'."
 650 STOP
 999 DATA 9999
2001 DATA 9999
2010 END

READY
```

DISCIPLINE MATHEMATICS, SOCIAL STUDIES

SUBJECT THE STOCK MARKET

PROGRAM NAME STOCK

DESCRIPTION:

This program simulates the stock market. Each student is given $10,000 with which he may buy and/or sell shares in five fictitious issues.

OBJECTIVES:

A. To give the student a simple understanding of the operations of the stock market.

B. To motivate the student to reinforce his basic arithmetic skills.

C. To give an example of the use of everyday mathematics and economics in everyday life.

PRELIMINARY PREPARATION:

A. Student - no special preparation

B. Materials - possibly graph paper

DISCUSSION:

This program can be used as a good motivation device in the teaching of basic stock-market concepts, and the basic mathematical skills involved. The computer starts each student with $10,000, and allows him to buy and/or sell shares. Precautionary tests are included for the student who tries to purchase more shares than he has money for, or to sell more shares than he actually owns. The program continues for as many trading days as the student desires.

The stock values rise and fall on a semi-random basis. On each trading day all stocks undergo a small random price change, a trend change (based on a random trend), and the possibility--on a random basis--of a large price change. The structure of the formula is:

new price=old price + (trend x old price) + (small random price change) + (possible large price change)

The trend is a random number between -.1 and +.1. It remains constant for a random number of days, at which time the trend is changed randomly. The trend affects all stocks equally, and attempts to simulate general market trends. The small random change ranges between -3 and +3 points. It occurs every day to every stock. The possible large price change is either +10 or -10 points. The + and - changes each occur at random day intervals, and to random stocks. That is, there may be no large change on some trading days, only a +10 change on others, a -10 change on still others, and both large and small changes on others. In all large-change cases, the change affects only one random stock when it occurs.

Because of the random generation of stock values and their fluctuations, the program does not exactly simulate the real market. It does, however, provide a simplified view of what does happen, and familiarizes the student with the basic functions involved. This should be explained to the students, along with some real causes of stock-market fluctuations.

Graph paper might be used to plot the daily stock values and the exchange average. In this way, the trend will become evident.

```
                    THE STOCK MARKET
DO YOU WANT THE INSTRUCTIONS (YES-TYPE 1, NO-TYPE 0)? 1


THIS PROGRAM PLAYS THE STOCK MARKET.  YOU WILL BE GIVEN
$10,000 AND MAY BUY OR SELL STOCKS.  THE STOCK PRICES WILL
BE GENERATED RANDOMLY AND THEREFORE THIS MODEL DOES NOT
REPRESENT EXACTLY WHAT HAPPENS ON THE EXCHANGE.  A TABLE
OF AVAILABLE STOCKS, THEIR PRICES, AND THE NUMBER OF SHARES
IN YOUR PORTFOLIO WILL BE PRINTED.  FOLLOWING THIS, THE
INITIALS OF EACH STOCK WILL BE PRINTED WITH A QUESTION
MARK.  HERE YOU INDICATE A TRANSACTION.  TO BUY A STOCK
TYPE +NNN, TO SELL A STOCK TYPE -NNN, WHERE NNN IS THE
NUMBER OF SHARES.  A BROKERAGE FEE OF 1% WILL BE CHARGED
ON ALL TRANSACTIONS.  NOTE THAT IF A STOCK'S VALUE DROPS
TO ZERO IT MAY REBOUND TO A POSITIVE VALUE AGAIN.  YOU
HAVE $10,000 TO INVEST.  USE INTEGERS FOR ALL YOUR INPUTS.
(NOTE:  TO GET A 'FEEL' FOR THE MARKET RUN FOR AT LEAST
10 DAYS)
-----GOOD LUCK!-----


STOCK                          INITIALS      PRICE/SHARE
INT. BALLISTIC MISSILES          IBM           65.75
RED CROSS OF AMERICA             RCA           85.5
LICHTENSTEIN, BUNRAP & JOKE      LBJ           155.25
AMERICAN BANKRUPT CO.            ABC           136
CENSORED BOOKS STORE             CBS           104.25

NEW YORK STOCK EXCHANGE AVERAGE:  113.75

TOTAL STOCK ASSETS ARE    $ 0
TOTAL CASH ASSETS ARE     $ 10000
TOTAL ASSETS ARE          $ 10000

WHAT IS YOUR TRANSACTION IN
IBM? 2
RCA? 3
LBJ? 1
ABC? 1
CBS? 1


**********  END OF DAY'S TRADING

STOCK         PRICE/SHARE   HOLDINGS      VALUE         NET PRICE CHANGE
IBM             96.5          2            193            10.75
RCA             81            3            243           -4.5
LBJ             153.5         1            153.5         -1.75
ABC             135.5         1            135.5         -0.5
CBS             99            1            99            -5.25


NEW YORK STOCK EXCHANGE AVERAGE:  113.1    NET CHANGE: -.65

TOTAL STOCK ASSETS ARE    $ 824
TOTAL CASH ASSETS ARE     $ 9166.25
TOTAL ASSETS ARE          $ 9990.25

DO YOU WISH TO CONTINUE (YES-TYPE 1, NO-TYPE 0)? 1
WHAT IS YOUR TRANSACTION IN
IBM? 5
RCA? 1
LBJ? 1
ABC? 1
CBS? 0
```

```
**********  END OF DAY'S TRADING

STOCK          PRICE/SHARE    HOLDINGS       VALUE          NET PRICE CHANGE
IBM             98.75          7              691.25          2.85
RCA             62.5           4              330             1.5
LBJ             154            2              308              .5
ABC             133.5          2              267             -2
CBS             108.75         1              108.75          3.75


NEW YORK STOCK EXCHANGE AVERAGE:  114.3    NET CHANGE:  1.2

TOTAL STOCK ASSETS ARE    $ 1699
TOTAL CASH ASSETS ARE     $ 8305.23
TOTAL ASSETS ARE          $ 10004.23

DO YOU WISH TO CONTINUE (YES-TYPE 1, NO-TYPE 0)? 1
WHAT IS YOUR TRANSACTION IN
IBM? 3
RCA? 2
LBJ? 5
ABC? -1
CBS? 3

**********  END OF DAY'S TRADING

STOCK          PRICE/SHARE    HOLDINGS       VALUE          NET PRICE CHANGE
IBM             99.25          10             992.5           .5
RCA             62.25          6              493.5           -.25
LBJ             154.75         7              1083.25          .75
ABC             133.5          1              133.5           0
CBS             103.25         4              413              .5


NEW YORK STOCK EXCHANGE AVERAGE:  114.6    NET CHANGE:  .3

TOTAL STOCK ASSETS ARE    $ 3115.75
TOTAL CASH ASSETS ARE     $ 6882.5
TOTAL ASSETS ARE          $ 9998.25

DO YOU WISH TO CONTINUE (YES-TYPE 1, NO-TYPE 0)? 1
WHAT IS YOUR TRANSACTION IN
IBM?  5

RCA? 3
LBJ? 5
ABC? 3
CBS? 4

**********  END OF DAY'S TRADING

STOCK          PRICE/SHARE    HOLDINGS       VALUE          NET PRICE CHANGE
IBM             96.75          15             1451.25         -2.5
RCA             60.5           9              724.5           -1.75
LBJ             150            12             1800            -4.75
ABC             138            4              552             -1.5
CBS             98.75          8              790             -4.5


NEW YORK STOCK EXCHANGE AVERAGE:  111.6    NET CHANGE:  -3

TOTAL STOCK ASSETS ARE    $ 5893.75
TOTAL CASH ASSETS ARE     $ 4586.95
TOTAL ASSETS ARE          $ 9622.7
```

```
DO YOU WISH TO CONTINUE (YES-TYPE 1, NO-TYPE 0)? 1
WHAT IS YOUR TRANSACTION IN
IBM? 0
RCA? -5
LBJ? -7
ABC? 0
CBS? -5


**********  END OF DAY'S TRADING

STOCK          PRICE/SHARE    HOLDINGS       VALUE          NET PRICE CHANGE
IBM             98.75          15             1451.25        0
RCA             66.75          4              267            -13.75
LBJ             150.75         5              753.75         .75
ABC             132            4              528            0
CBS             95.75          3              287.25         -3


NEW YORK STOCK EXCHANGE AVERAGE:  108.4    NET CHANGE: -3.8

TOTAL STOCK ASSETS ARE    $ 3287.25
TOTAL CASH ASSETS ARE     $ 6455.74
TOTAL ASSETS ARE          $ 9742.99

DO YOU WISH TO CONTINUE (YES-TYPE 1, NO-TYPE 0)? 1
WHAT IS YOUR TRANSACTION IN
IBM? -10
RCA? -2
LBJ? 2
ABC? 2
CBS? 0


**********  END OF DAY'S TRADING

STOCK          PRICE/SHARE    HOLDINGS       VALUE          NET PRICE CHANGE
IBM             87.5           5              437.5          -9.25
RCA             58             2              116            -8.75
LBJ             135.25         7              946.75         -15.5
ABC             132.5          6              735            -9.5
CBS             98.75          3              296.25         3


NEW YORK STOCK EXCHANGE AVERAGE:  100.4    NET CHANGE: -8

TOTAL STOCK ASSETS ARE    $ 2531.5
TOTAL CASH ASSETS ARE     $ 6974.58
TOTAL ASSETS ARE          $ 9506.08

DO YOU WISH TO CONTINUE (YES-TYPE 1, NO-TYPE 0)? 1
WHAT IS YOUR TRANSACTION IN
IBM? -4
RCA? -1
LBJ? -6
ABC? -8
CBS? -8
```

```
**********  END OF DAY'S TRADING

STOCK          PRICE/SHARE    HOLDINGS       VALUE          NET PRICE CHANGE
IBM             80             1              80             -7.5
RCA             51             1              51             -7
LBJ             121.75         1              121.75         -13.5
ABC             109.5          4              438            -13
CBS             91.5           1              91.5           -7.25


NEW YORK STOCK EXCHANGE AVERAGE:  90.75    NET CHANGE: -9.65

TOTAL STOCK ASSETS ARE    $ 762.25
TOTAL CASH ASSETS ARE     $ 8619.96
TOTAL ASSETS ARE          $ 9408.21

DO YOU WISH TO CONTINUE (YES-TYPE 1, NO-TYPE 0)? 1
WHAT IS YOUR TRANSACTION IN
IBM? 0
RCA? 0
LBJ? 0
ABC? -3
CBS? 0

**********  END OF DAY'S TRADING

STOCK          PRICE/SHARE    HOLDINGS       VALUE          NET PRICE CHANGE
IBM             77.5           1              77.5           -2.5
RCA             52.25          1              52.25           1.25
LBJ             119.25         1              119.25         -2.5
ABC             107            1              107            -2.5
CBS             92.25          1              92.25           .75


NEW YORK STOCK EXCHANGE AVERAGE:  89.65    NET CHANGE: -1.1

TOTAL STOCK ASSETS ARE    $ 446.25
TOTAL CASH ASSETS ARE     $ 8945.18
TOTAL ASSETS ARE          $ 9393.43

DO YOU WISH TO CONTINUE (YES-TYPE 1, NO-TYPE 0)? 1
WHAT IS YOUR TRANSACTION IN
IBM? 0
RCA? 0
LBJ? 0
ABC? 0
CBS? 10

**********  END OF DAY'S TRADING

STOCK          PRICE/SHARE    HOLDINGS       VALUE          NET PRICE CHANGE
IBM             74.5           1              74.5           -3
RCA             54             1              54              1.75
LBJ             107            1              107            -12.25
ABC             108            1              108             1
CBS             90.75          11             998.25         -1.5
```

```
NEW YORK STOCK EXCHANGE AVERAGE:  86.85    NET CHANGE: -2.8

TOTAL STOCK ASSETS ARE    $ 1341.75
TOTAL CASH ASSETS ARE     $ 8013.46
TOTAL ASSETS ARE          $ 9355.21

DO YOU WISH TO CONTINUE (YES-TYPE 1, NO-TYPE 0)? 1
WHAT IS YOUR TRANSACTION IN
IBM? 5
RCA? 6
LBJ? 10
ABC? 10
CBS? 10


**********  END OF DAY'S TRADING

STOCK          PRICE/SHARE    HOLDINGS       VALUE          NET PRICE CHANGE
IBM             72             6              432            -2.5
RCA             52.5           7              367.5          -1.5
LBJ             105            11             1155           -2
ABC             103.25         11             1135.75        -4.75
CBS             91.5           21             1921.5          .75


NEW YORK STOCK EXCHANGE AVERAGE:  84.85    NET CHANGE: -2

TOTAL STOCK ASSETS ARE    $ 5011.75
TOTAL CASH ASSETS ARE     $ 4821.92
TOTAL ASSETS ARE          $ 9833.67

DO YOU WISH TO CONTINUE (YES-TYPE 1, NO-TYPE 0)? 0
HOPE YOU HAD FUN!!


READY
```

```
100 REM STOCK MARKET SIMULATION      -STOCK-
101 REM REVISED 8/18/70 (D. PESSEL, L. BRAUN, C. LOSIK)
102 REM IMP VRBLS: A-MRKT TRND SLP; B5-BRKRGE FEE; C-TTL CSH ASSTS;
103 REM C5-TTL CSH ASSTS (TEMP); C(I)-CHNG IN STK VAL; D-TTL ASSTS;
104 REM E1,E2-LRG CHNG MISC; I-STCK #; I1,I2-STCKS W LRG CHNG;
105 REM N1,N2-LRG CHNG DAY CNTS; P5-TTL DAYS PRCHSS; P(I)-PRTFL CNTNTS;
106 REM Q9-NEW CYCL?; S4-SGN OF A; S5-TTL DYS SLS; S(I)-VALUE/SHR;
107 REM T-TTL STCK ASSTS; T5-TTL VAL OF TRNSCTNS;
108 REM W3-LRG CHNG; X1-SMLL CHNG(<$1); Z4,Z5,Z6-NYSE AVE.; Z(I)-TRNSCTN
109  PRINT TAB(20);"THE STOCK MARKET"
110  DIM S(5),P(5),Z(5),C(5)
112 REM SLOPE OF MARKET TREND:A  (SAME FOR ALL STOCKS)
113  RANDOMIZE
114  LET A=INT((RND(X)/10)*100+.5)/100
115  LET T5=0
116  LET X9=0
117  LET N1=0
118  LET N2=0
119  LET E1=0
120  LET E2=0
121 REM INTRODUCTION
122  PRINT "DO YOU WANT THE INSTRUCTIONS (YES-TYPE 1, NO-TYPE 0)";
123  INPUT Z9
124  PRINT
125  PRINT
126  IF Z9<1 THEN 200
130  PRINT "THIS PROGRAM PLAYS THE STOCK MARKET.  YOU WILL BE GIVEN"
132  PRINT "$10,000 AND MAY BUY OR SELL STOCKS.  THE STOCK PRICES WILL"
134  PRINT "BE GENERATED RANDOMLY AND THEREFORE THIS MODEL DOES NOT"
135  PRINT "REPRESENT EXACTLY WHAT HAPPENS ON THE EXCHANGE.  A TABLE"
136  PRINT "OF AVAILABLE STOCKS, THEIR PRICES, AND THE NUMBER OF SHARES"
137  PRINT "IN YOUR PORTFOLIO WILL BE PRINTED.  FOLLOWING THIS, THE"
138  PRINT "INITIALS OF EACH STOCK WILL BE PRINTED WITH A QUESTION"
139  PRINT "MARK.  HERE YOU INDICATE A TRANSACTION.  TO BUY A STOCK"
140  PRINT "TYPE +NNN, TO SELL A STOCK TYPE -NNN, WHERE NNN IS THE"
141  PRINT "NUMBER OF SHARES.  A BROKERAGE FEE OF 1% WILL BE CHARGED"
142  PRINT "ON ALL TRANSACTIONS.  NOTE THAT IF A STOCK'S VALUE DROPS"
143  PRINT "TO ZERO IT MAY REBOUND TO A POSITIVE VALUE AGAIN.  YOU"
144  PRINT "HAVE $10,000 TO INVEST.  USE INTEGERS FOR ALL YOUR INPUTS."
145  PRINT "(NOTE:  TO GET A 'FEEL' FOR THE MARKET RUN FOR AT LEAST"
146  PRINT "10 DAYS)"
147  PRINT "------GOOD LUCK!------"
200 REM GENERATION OF STOCK TABLE; INPUT REQUESTS
210 REM INITIAL STOCK VALUES
220  LET S(1)=100
230  LET S(2)=85
240  LET S(3)=150
250  LET S(4)=140
260  LET S(5)=110
265 REM INITIAL T8 - # DAYS FOR FIRST TREND SLOPE (A)
266  LET T8=INT(4.99*RND(X)+1)
267 REM RANDOMIZE SIGN OF FIRST TREND SLOPE (A)
268  IF RND(X)>.5 THEN 270
269  LET A=-A
270 REM RANDOMIZE INITIAL VALUES
280  GOSUB 890
285 REM INITIAL PORTFOLIO CONTENTS
290  FOR I=1 TO 5
300  LET P(I)=0
305  LET Z(I)=0
310  NEXT I
320  PRINT
```

```
330  PRINT
333 REM INITIALIZE CASH ASSETS:C
335  LET C=10000
338 REM PRINT INITIAL PORTFOLIO
340  PRINT "STOCK";" ","INITIALS","PRICE/SHARE"
350  PRINT "INT. BALLISTIC MISSILES","  IBM",S(1)
358  PRINT "RED CROSS OF AMERICA","  RCA",S(2)
354  PRINT "LICHTENSTEIN, BUMRAP & JOKE","  LBJ",S(3)
356  PRINT "AMERICAN BANKRUPT CO.","  ABC",S(4)
358  PRINT "CENSURED BOOKS STORE","  CBS",S(5)
360  PRINT
361 REM NYSE AVERAGE:Z5; TEMP. VALUE:Z4; NET CHANGE:Z6
363  LET Z4=Z5
364  LET Z5=0
365  LET T=0
370  FOR I=1 TO 5
375  LET Z5=Z5+S(I)
380  LET T=T+S(I)*P(I)
390  NEXT I
391  LET Z5=INT(100*(Z5/5)+.5)/100
392  LET Z6=INT((Z5-Z4)*100+.5)/100
393 REM TOTAL ASSETS:D
394  LET D=T+C
395  IF X9>0 THEN 398
396  PRINT "NEW YORK STOCK EXCHANGE AVERAGE: "Z5
397  GO TO 399
398  PRINT "NEW YORK STOCK EXCHANGE AVERAGE: "Z5"   NET CHANGE: "Z6
399  PRINT
400  LET T=INT(100*T+.5)/100
401  PRINT "TOTAL STOCK ASSETS ARE   $";T
403  LET C=INT(100*C+.5)/100
405  PRINT "TOTAL CASH ASSETS ARE    $";C
407  LET D=INT(100*D+.5)/100
408  PRINT "TOTAL ASSETS ARE         $";D
410  PRINT
411  IF X9=0 THEN 416
412  PRINT "DO YOU WISH TO CONTINUE (YES-TYPE 1, NO-TYPE 0)=";
413  INPUT Q9
414  IF Q9<1 THEN 998
416 REM INPUT TRANSACTIONS
420  PRINT "WHAT IS YOUR TRANSACTION IN"
430  PRINT "IBM";
440  INPUT Z(1)
450  PRINT "RCA";
460  INPUT Z(2)
470  PRINT "LBJ";
480  INPUT Z(3)
490  PRINT "ABC";
500  INPUT Z(4)
510  PRINT "CBS";
520  INPUT Z(5)
525  PRINT
530 REM TOTAL DAY'S PURCHASES IN $:P5
540  LET P5=0
550 REM TOTAL DAY'S SALES IN $:S5
560  LET S5=0
570  FOR I=1 TO 5
575  LET Z(I)=INT(Z(I)+.5)
580  IF Z(I)<=0 THEN 610
590  LET P5=P5+Z(I)*S(I)
600  GO TO 620
610  LET S5=S5-Z(I)*S(I)
612  IF -Z(I)<=P(I) THEN 620
614  PRINT "YOU HAVE OVERSOLD A STOCK; TRY AGAIN."
616  GO TO 420
620  NEXT I
```

```
622 REM TOTAL VALUE OF TRANSACTIONS:T5
625  LET T5=P5+S5
630 REM BROKERAGE FEE:B5
640  LET B5=INT(.01*T5*100+.5)/100
650 REM CASH ASSETS=OLD CASH ASSETS-TOTAL PURCHASES
652 REM -BROKERAGE FEES+TOTAL SALES:C5
654  LET C5=C-P5-B5+S5
656  IF C5>=0 THEN 674
658  PRINT "YOU HAVE USED $"-C5" MO
RE THEN YOU HAVE."
660  GO TO 480
674  LET C=C5
675 REM CALCULATE NEW PORTFOLIO
680  FOR I=1 TO 5
690  LET P(I)=P(I)+Z(I)
700  NEXT I
710 REM CALCULATE NEW STOCK VALUES
720  GOSUB 830
750 REM PRINT PORTFOLIO
751 REM BELL RINGING-DIFFERENT ON MANY COMPUTERS
752  FOR I=1 TO 20
753  PRINT CHR$(135);
754  NEXT I
755  PRINT
756  PRINT "**********  END OF DAY'S TRADING"
757  PRINT
758  PRINT
759  IF X9<1 THEN 769
769  PRINT "STOCK","PRICE/SHARE","HOLDINGS","VALUE","NET PRICE CHANGE"
770  PRINT "IBM", S(1), P(1), S(1)*P(1), C(1)
771  PRINT "RCA", S(2), P(2), S(2)*P(2), C(2)
772  PRINT "LBJ", S(3), P(3), S(3)*P(3), C(3)
773  PRINT "ABC", S(4), P(4), S(4)*P(4), C(4)
774  PRINT "CBS", S(5), P(5), S(5)*P(5), C(5)
775  LET X9=1
780  PRINT
790  PRINT
810  GO TO 360
829 REM NEW STOCK VALUES - SUBROUTINE
830 REM RANDOMLY PRODUCE NEW STOCK VALUES BASED ON PREVIOUS
831 REM DAY'S VALUES
832 REM N1,N2 ARE RANDOM NUMBERS OF DAYS WHICH RESPECTIVELY
833 REM DETERMINE WHEN STOCK I1 WILL INCREASE 10 PTS. AND STOCK
834 REM I2 WILL DECREASE 10 PTS.
840 REM IF N1 DAYS HAVE PASSED, PICK AN I1, SET E1, DETERMINE NEW N1
841  IF N1>0 THEN 850
845  LET I1=INT(4.99*RND(X)+1)
846  LET N1=INT(4.99*RND(X)+1)
847  LET E1=1
850 REM IF N2 DAYS HAVE PASSED, PICK AN I2, SET E2, DETERMINE NEW N2
851  IF N2>0 THEN 860
855  LET I2=INT(4.99*RND(X)+1)
856  LET N2=INT(4.99*RND(X)+1)
857  LET E2=1
860 REM DEDUCT ONE DAY FROM N1 AND N2
861  LET N1=N1-1
862  LET N2=N2-1
890 REM LOOP THROUGH ALL STOCKS
900  FOR I=1 TO 5
910  LET X1=RND(X)
915  IF X1>.25 THEN 920
```

```
916  LET X1=.25
917  GO TO 935
920  IF X1>.50 THEN 925
921  LET X1=.50
922  GO TO 935
925  IF X1>.75 THEN 930
926  LET X1=.75
927  GO TO 935
930  LET X1=0.0
931 REM BIG CHANGE CONSTANT:W3  (SET TO ZERO INITIALLY)
935  LET W3=0
936  IF E1<1 THEN 945
937  IF INT(I1+.5)<>INT(I+.5) THEN 945
938 REM ADD 10 PTS. TO THIS STOCK;  RESET E1
939  LET W3=10
943  LET E1=0
945  IF E2<1 THEN 955
947  IF INT(I2+.5)<>INT(I+.5) THEN 955
948 REM SUBTRACT 10 PTS. FROM THIS STOCK;  RESET E2
949  LET W3=W3-10
953  LET E2=0
954 REM C(I) IS CHANGE IN STOCK VALUE
955  LET C(I)=INT(A*S(I))+X1+INT(3-6*RND(X)+.5)+W3
956  LET C(I)=INT(100*C(I)+.5)/100
957  LET S(I)=S(I)+C(I)
960  IF S(I)>0 THEN 967
964  LET C(I)=0
965  LET S(I)=0
966  GO TO 970
967  LET S(I)=INT(100*S(I)+.5)/100
970  NEXT I
972 REM AFTER T8 DAYS RANDOMLY CHANGE TREND SIGN AND SLOPE
973  LET T8=T8-1
974  IF T8<1 THEN 985
980  RETURN
985 REM RANDOMLY CHANGE TREND SIGN AND SLOPE (A), AND DURATION OF
986 REM OF TREND (T8)
990  LET T8=INT(4.99*RND(X)+1)
992  LET A=INT((RND(X)/10)*100+.5)/100
993  LET S4=RND(X)
994  IF S4<=.5 THEN 997
995  LET A=-A
997  RETURN
998  PRINT "HOPE YOU HAD FUN!!"
999  END
```

DISCIPLINE CALCULUS-GRADE 13

SUBJECT AREA OF A SURFACE OF REVOLUTION

PROGRAM NAME SURFAR

DESCRIPTION:

This program approximates the area of a surface of revolution, by computing lateral areas of frustrums of cones of revolution.

OBJECTIVES:

A. The saving of time in computations.

B. The speedy demonstration of limiting processes.

C. The focusing of attention upon those processes needed to develop the analytic approach.

PRELIMINARY PREPARATION:

Before running this program, the lateral area of a frustrum of a cone should be discussed. Many students in the Advanced Placement Program have not taken a course in Solid Geometry and may be unfamiliar with the formula:

$$\text{Lateral Area} = \pi\, l\, (r_1 + r_2)$$



Frustrum of a Cone

Whether or not this formula is derived in class will depend on the amount of time available. Most likely it will merely be stated; students who have not taken Solid Geometry may be asked to look up the derivation on their own.

```
                    AREA OF A SURFACE OF REVOLUTION

    THIS PROGRAM APPROXIMATES THE AREA OF A SURFACE OF
REVOLUTION BY COMPUTING LATERAL AREAS OF FRUSTUMS OF CONES
OF REVOLUTION.  TYPE IN YOUR FUNCTION OF X (Y=F(X)),
WHOSE GRAPH WILL BE ROTATED ABOUT THE X AXIS, AS FOLLOWS:

        1 GO TO 300
        300 DEF FNY(X)=...(YOUR FUNCTION OF X)...
        RUN

FOR EXAMPLE, TO USE THE FUNCTION Y=X↑2 YOU WOULD TYPE:

        1 GO TO 300
        300 DEF FNY(X)=X↑2
        RUN

YOU MIGHT TRY THAT AS YOUR FIRST RUN.
END EACH LINE, INCLUDING 'RUN', WITH THE 'RETURN' KEY.


READY

1 GO TO 300
300 DEF FNY(X)=X↑2
RUN

WHAT ARE THE ABSCISSAS OF THE END POINTS OF THE SECTION
TO BE CONSIDERED (SMALLER FIRST: P,Q)? -3,2

NUMBER OF       SUM OF                    % CHANGE
SUBINTERVALS    APPROXIMATING AREAS       IN SUM
------------    -------------------       --------
 1              868.7871                  NO PREVIOUS VALUE
 2              324.6289                   11.68411
 4              317.6819                   2.161263
 8              315.3346                   .7416313
 16             314.7434                   .1876635
 32             314.5933                   .04769154
 64             314.5557                   .01197374
 128            314.5461                   3.025796E-3

WOULD YOU LIKE TO TRY NEW END POINTS (1-YES, 0-NO)? 0
TO ENTER A NEW FUNCTION YOU NEED ONLY RETYPE LINE
300 AND 'RUN'.  SEE INSTRUCTIONS FOR MORE DETAILS.
IF YOU ARE FINISHED, TYPE '1' AND THE 'RETURN' KEY.


READY

1
```

```
100 REM AREA OF A SURFACE OF REVOLUTION, G. J. O'CONNOR, 7/12/68
101 REM REVISED 8/21/70 (D. PESSEL)
105 PRINT TAB(17); "AREA OF A SURFACE OF REVOLUTION"
106 PRINT
110 PRINT "    THIS PROGRAM APPROXIMATES THE AREA OF A SURFACE OF"
120 PRINT "REVOLUTION BY COMPUTING LATERAL AREAS OF FRUSTUMS OF CONES"
130 PRINT "OF REVOLUTION.  TYPE IN YOUR FUNCTION OF X (Y=F(X)),"
131 PRINT "WHOSE GRAPH WILL BE ROTATED ABOUT THE X AXIS, AS FOLLOWS:"
150 PRINT
160 PRINT "       1 GO TO 200"
170 PRINT "       300 DEF FNY(X)=...(YOUR FUNCTION OF X)..."
180 PRINT "       RUN"
185 PRINT
186 PRINT "FOR EXAMPLE, TO USE THE FUNCTION Y=X↑2 YOU WOULD TYPE:"
187 PRINT
188 PRINT "       1 GO TO 200"
189 PRINT "       300 DEF FNY(X)=X↑2"
190 PRINT "       RUN"
191 PRINT
192 PRINT "YOU MIGHT TRY THAT AS YOUR FIRST RUN."
193 PRINT "END EACH LINE, INCLUDING 'RUN', WITH THE 'RETURN' KEY."
195 STOP
200 REM COMPUTATION SECTION OF PROGRAM
220 PRINT "WHAT ARE THE ABSCISSAS OF THE END POINTS OF THE SECTION"
230 PRINT "TO BE CONSIDERED (SMALLER FIRST: P,Q)";
240 INPUT P,Q
245 IF P<=Q THEN 250
246 PRINT "P CANNOT BE GREATER THAN Q!"
247 GO TO 220
250 PRINT
260 PRINT "NUMBER OF       SUM OF                        % CHANGE"
270 PRINT "SUBINTERVALS    APPROXIMATING AREAS           IN SUM"
280 PRINT "-------------   -------------------           --------"
285 LET E1=0
300 DEF FNY(X)=X↑2
305 FOR N=1 TO 9
310 LET E=2↑(N-1)
320 LET H=(Q-P)/E
330 LET S=0
340 FOR I=0 TO (E-1)
350 LET G=FNY(P+I*H+H)+FNY(P+I*H)
360 LET M=FNY(P+I*H+H)-FNY(P+I*H)
370 LET L=3.14159*G*SQR(M*M+H*H)
380 LET S=S+L
390 NEXT I
395 IF S1=0 THEN 405
396 LET W=100*(ABS(S-S1))/((S+S1)/2)
399 IF S1=0 THEN 405
400 PRINT E,S," ",W
402 IF W<1E-2 THEN 420
404 GO TO 407
405 PRINT E,S," ","NO PREVIOUS VALUE"
407 LET S1=S
410 NEXT N
420 PRINT
430 PRINT "WOULD YOU LIKE TO TRY NEW END POINTS (1-YES, 0-NO)";
431 INPUT Q1
432 IF Q1>0 THEN 220
440 PRINT "TO ENTER A NEW FUNCTION YOU NEED ONLY RETYPE LINE"
450 PRINT "300 AND 'RUN'.  SEE INSTRUCTIONS FOR MORE DETAILS."
460 PRINT "IF YOU ARE FINISHED, TYPE '1' AND THE 'RETURN' KEY."
500 END
```

DISCIPLINE CALCULUS - GRADE 13

SUBJECT VOLUME OF ANY SOLID OF REVOLUTION, (ANALYTICALLY DEFINED)

PROGRAM NAME VOLSOL

DESCRIPTION:

Through the use of cylindrical discs, the program approximates the volume of a solid of revolution generated by rotating about the x-axis the area bounded by $y = f(x)$, the x-axis, and the vertical lines $x = a$ and $x = b$.

OBJECTIVES:

To help the student understand the analytic procedures and to appreciate the nature of the limiting process.

PRELIMINARY PREPARATION:

The class should be reminded of the formula for the volume of a cylinder, and the way in which a cylinder is generated by rotating a rectangle about one of its sides.

DISCUSSION:

It would be desirable to make use of an overhead projector transparency to display the cylindrical discs generated.



Approximation of Volume of Revolution by Cylindrical Discs

VOLUME OF A SOLID OF REVOLUTION

THIS PROGRAM USES CYLINDRICAL DISCS TO APPROXIMATE THE VOLUME OF A SOLID OF REVOLUTION. THE SOLID IS GENERATED BY ROTATING ABOUT THE X-AXIS THE AREA BOUNDED BY Y=F(X), THE LINES X=A AND X=B, AND THE X-AXIS.

TO INPUT YOUR FUNCTION OF X (Y=F(X)) TYPE AS FOLLOWS:

```
1 GO TO 200
220 DEF FNY(X)=....(YOUR FUNCTION OF X)....
RUN
```

FOR EXAMPLE, TO USE THE FUNCTION Y=X↑2 YOU WOULD TYPE:

```
1 GO TO 200
220 DEF FNY(X)=X↑2
RUN
```

YOU MIGHT TRY THAT AS YOUR FIRST EXAMPLE.
END EACH LINE, INCLUDING 'RUN', WITH THE 'RETURN' KEY.

READY

```
1 GO TO 200
220 DEF FNY(X)=X↑2
RUN
```

WHAT ARE YOUR VALUES FOR A AND B (SMALLER FIRST: A,B)? 0,5

NUMBER OF CYLINDERS	SUM OF CYLINDER VOLUMES	% CHANGE IN SUM
1	0	NO PREV. VALUE, OR IT WAS ZERO
2	306.7959	NO PREV. VALUE, OR IT WAS ZERO
4	939.5624	206.25
8	1400.955	49.10714
16	1669.476	19.16702
32	1813.291	8.614392
64	1887.594	4.097653
128	1925.344	1.999911
256	1944.369	.9881206
512	1953.918	.4911339

```
WOULD YOU LIKE TO TRY YOUR OWN 'NUMBER OF CYLINDERS' (1-YES, 0-NO)? 1
HOW MANY CYLINDERS WOULD YOU LIKE TO TRY? 700

FOR  700  CYLINDERS THE VOLUME IS  1956.487  .

WOULD YOU LIKE TO TRY AGAIN (1-YES, 0-NO)? 0

*****

WOULD YOU LIKE TO TRY NEW VALUES OF A AND B (1-YES, 0-NO)? 0
TO USE A NEW FUNCTION YOU NEED ONLY RETYPE LINE 220 AND
'RUN'.  SEE INSTRUCTIONS FOR MORE DETAILS.
IF YOU ARE FINISHED, TYPE '1' AND THE 'RETURN' KEY.


READY

1
```

```
100 REM VOLUME OF A SOLID OF REVOLUTION, Q. J. O'CONNOR, 8/1/68
101 REM REVISED 8/24/70 (D. PESSEL)
110 PRINT TAB(15);"VOLUME OF A SOLID OF REVOLUTION"
111 PRINT
115 PRINT"     THIS PROGRAM USES CYLINDRICAL DISCS TO APPROXIMATE"
117 PRINT"THE VOLUME OF A SOLID OF REVOLUTION.  THE SOLID IS GENE-"
120 PRINT"RATED BY ROTATING ABOUT THE X-AXIS THE AREA BOUNDED BY"
130 PRINT"Y=F(X), THE LINES X=A AND X=B, AND THE X-AXIS."
135 PRINT
140 PRINT"TO INPUT YOUR FUNCTION OF X (Y=F(X)) TYPE AS FOLLOWS:"
141 PRINT
145 PRINT"           1 GO TO 200"
150 PRINT"           220 DEF FNY(X)=....(YOUR FUNCTION OF X)...."
160 PRINT"           RUN"
161 PRINT
165 PRINT"FOR EXAMPLE, TO USE THE FUNCTION Y=X↑2 YOU WOULD TYPE:"
166 PRINT
167 PRINT"           1 GO TO 200"
168 PRINT"           220 DEF FNY(X)=X↑2"
169 PRINT"           RUN"
170 PRINT
175 PRINT"YOU MIGHT TRY THAT AS YOUR FIRST EXAMPLE."
176 PRINT"END EACH LINE, INCLUDING 'RUN', WITH THE 'RETURN' KEY."
180 STOP
199 PRINT
200 PRINT"WHAT ARE YOUR VALUES FOR A AND B (SMALLER FIRST: A,B)";
210 INPUT A,B
211 IF A<B THEN 214
212 PRINT"A MUST BE SMALLER THAN B!"
213 GO TO 200
214 PRINT
215 PRINT"NUMBER OF          SUM OF                  % CHANGE"
216 PRINT"CYLINDERS      CYLINDER VOLUMES             IN SUM"
217 PRINT"---------      ----------------            --------"
218 LET V1=0
220 DEF FNY(X)=X↑2
230 FOR N=1 TO 10
240 LET D=2↑(N-1)
250 LET H=(B-A)/D
260 LET V=0
270 FOR I=0 TO (D-1)
280 LET Y=FNY(A+I*H)
290 LET V=V+3.14159*Y*Y*H
300 NEXT I
305 IF V1=0 THEN 315
307 LET P=100*(ABS(V-V1))/V1
310 PRINT D,V," ",P
```

```
312 IF P<.5 THEN 330
313 GO TO 318
315 PRINT D,V,"    NO PREV. VALUE, OR IT WAS ZERO"
318 LET V1=V
320 NEXT N
330 PRINT
333 PRINT
334 PRINT"WOULD YOU LIKE TO TRY YOUR OWN 'NUMBER OF CYLINDERS' (1-YES";
335 PRINT", 0-NO)";
336 INPUT Q5
337 IF Q5<1 THEN 377
338 PRINT"HOW MANY CYLINDERS WOULD YOU LIKE TO TRY";
339 INPUT D1
340 IF D1>1 THEN 343
341 PRINT"NUMBER OF CYLINDERS MUST BE GREATER THAN ZERO!"
342 GO TO 338
343 IF D1<1000 THEN 347
344 PRINT"THIS IS A VERY LARGE NUMBER OF CYLINDERS AND MAY TAKE"
345 PRINT"A LONG TIME TO RUN."
347 LET V2=0
348 LET H1=(B-A)/D1
349 LET D1=INT(D1+.5)
350 FOR I=0 TO (D1-1)
352 LET Y1=FNY(A+I*H1)
354 LET V2=V2+3.14159*Y1*Y1*H1
356 NEXT I
358 PRINT
360 PRINT "FOR "D1" CYLINDERS THE VOLUME IS "V2" ."
362 PRINT
363 PRINT"WOULD YOU LIKE TO TRY AGAIN (1-YES, 0-NO)";
364 INPUT Q6
365 IF Q6>0 THEN 338
377 PRINT
378 PRINT"*****"
379 PRINT
380 PRINT"WOULD YOU LIKE TO TRY NEW VALUES OF A AND B (1-YES, 0-NO)";
382 INPUT Q1
384 IF Q1>0 THEN 199
386 PRINT"TO USE A NEW FUNCTION YOU NEED ONLY RETYPE LINE 220 AND"
388 PRINT"'RUN'.  SEE INSTRUCTIONS FOR MORE DETAILS."
390 PRINT"IF YOU ARE FINISHED, TYPE '1' AND THE 'RETURN' KEY."
500 END
```

DISCIPLINE PHYSICS

SUBJECT MAGNETIC FIELDS

PROGRAM NAME BFIELD

DESCRIPTION:

Student may visualize the effects of current on the magnetic field produced about a single conductor. The student may also explore the fields produced by the current flow in two parallel wires. The current in the two wires may be chosen in the same direction or in opposite directions.

OBJECTIVES:

To acquaint student with the magnetic fields produced by current carrying conductors.

PRELIMINARY PREPARATION:

A. Student - Prior preparation involving currents and fields.

B. Materials - None

DISCUSSION:

Student may qualitatively explore the effects of currents on the production of magnetic fields by successively increasing or decreasing the current. The resulting magnetic field is printed out showing the relative magnitude of the field in relation to the position of the current.

The student may also view the magnetic field due to two currents in the same or opposite direction.

This program may also be used to introduce groups to the field concept. In addition, minor modification of the program will produce a series of plots which will demonstrate an expanding field resulting from an increasing current.

```
WOULD YOU LIKE TO TRY TWO DIFFERENT CURRENTS AT THE SAME
TIME (YES=1; NO=0)? 0
WOULD YOU RATHER TRY ANOTHER CURRENT (YES=1; NO=0)? 1
ENTER YOUR VALUE OF CURRENT? 6

THE MAGNITUDE OF THE FIELD DECREASES FROM 9 TO 0.
9 IS THE HIGHEST POSSIBLE FIELD STRENGTH, AND 0
(WHICH MEANS A ZERO FIELD) THE LOWEST.
```



```
WOULD YOU LIKE TO TRY TWO DIFFERENT CURRENTS AT THE SAME
TIME (YES=1; NO=0)? 1
THE TWO CURRENTS WILL BE SEPARATED BY 1.0 METER. (NOTE: IF
THE CURRENTS ARE TO BE OPPOSITELY DIRECTED, STATE ONE OF THEM
AS A NEGATIVE VALUE).
ENTER THE TWO CURRENTS? 4,-8

                              METERS
              -1.2       -.6         0          .6        1.2
   METERS   .+.........+.........+.........+.........+.
      1.2   . 11111111111       22222222222222222222  .
      1.1   . 111111111      2222222222222222222222222.
        1   . 11111111     2222222                222222.
       .9   . 1111111     22222     333333333333     222.
       .8   . 111111    2222    3333333       333333    2.
       .7   . 11111    222  3333     4444444444    3333  .
       .6   . 11111  222 333  4444    555555  444  333 .
       .5   . 1111  22  3  44   555  666666  55 44  33.
       .4   .  111  2 3344 555  666 77     7 6 55 4  3.
       .3   .  111 2 3 45 66   777 8 9        76 5 4  .
       .2   .  111 2   678   8888 9           876 544 .
       .1   .  111 23469      9999             8 65 4 .
        0   .  1112 3 7  +                -     8765 4 .
      -.1   .  111 23469      9999             8 65 4 .
      -.2   .  111 2   678   8888 9           876 544 .
      -.3   .  111 2 3 45 66   777 8 9        76 5 4  .
      -.4   .  111  2 3344 555  666 77     7 6 55 4  3.
      -.5   . 1111  22  3  44   555  666666  55 44  33.
      -.6   . 11111  222 333  4444   555555  444  333 .
      -.7   . 11111   222  3333     4444444444    3333  .
      -.8   . 111111    2222    3333333       333333    2.
      -.9   . 1111111     22222     333333333333     222.
        -1  . 11111111     2222222                222222.
      -1.1  . 111111111      2222222222222222222222222.
      -1.2  . 11111111111       22222222222222222222  .
            .+.........+.........+.........+.........+.

WANT TO TRY AGAIN (YES=1; NO=0)? 1
THE TWO CURRENTS WILL BE SEPARATED BY 1.0 METER. (NOTE: IF
THE CURRENTS ARE TO BE OPPOSITELY DIRECTED, STATE ONE OF THEM
AS A NEGATIVE VALUE).
ENTER THE TWO CURRENTS? 4,4
```

```
                    METERS
          -1.2      -.6        0         .6        1.2
METERS    .+.........+.........+.........+.........+.
    1.2   .22222222222222222222222222222222222222222.
    1.1   .222222222                         222222222.
    1     .22222                                 22222.
    .9    .22          333333333333333333          22.
    .8    .       3333333333333333333333333333       .
    .7    .   333333333333333333333333333333333333   .
    .6    .  33333           3333333           33333  .
    .5    .3333   44444444   33333   44444444   3333.
    .4    .333  444  555 444 33333 444 555  444  333.
    .3    .33  44 5  666 5 4 33333 4 5 666  5 44  33.
    .2    .3  44 56  8 8 6543  2  3456 8 8  65 44  3.
    .1    .3 44 5  8      8 432 1 234 8      8  5 44 3.
    0     .3 44 567    +  64  101  46  +    765 44 3.
   -.1    .3 44 5  8      8 432 1 234 8      8  5 44 3.
   -.2    .3  44 56  8 8 6543  2  3456 8 8  65 44  3.
   -.3    .33  44 5  666 5 4 33333 4 5 666  5 44  33.
   -.4    .333  444  555 444 33333 444 555  444  333.
   -.5    .3333   44444444   33333   44444444   3333.
   -.6    .  33333           3333333           33333  .
   -.7    .   333333333333333333333333333333333333   .
   -.8    .       3333333333333333333333333333       .
   -.9    .22          333333333333333333          22.
   -1     .22222                                 22222.
   -1.1   .222222222                         222222222.
   -1.2   .22222222222222222222222222222222222222222.
          .+.........+.........+.........+.........+.
```

WANT TO TRY AGAIN (YES=1; NO=0)? 1
THE TWO CURRENTS WILL BE SEPARATED BY 1.0 METER. (NOTE: IF THE CURRENTS ARE TO BE OPPOSITELY DIRECTED, STATE ONE OF THEM AS A NEGATIVE VALUE).
ENTER THE TWO CURRENTS? 4,-4




```
WANT TO TRY AGAIN (YES=1; NO=0)? 0
WELL I GUESS YOU'RE ALL THROUGH. THANKS-- SEE Y'A


READY
```

```
1REM A.C. CAGGIANO; PATCHOGUE H.S.; PHYSICS; 7-'69
2 REM  THIS PROGRAM PERMITS A STUDENT TO VISUALIZE THE MAGNETIC
3 REM INDUCTION ABOUT A SINGLE CONDUCTER AND THE INFLUENCE OF THE
4 REM CURRENT ON THE MAGNITUDE OF THE FIELD. THE STUDENT MAY ALSO
5 REM VIEW THE MAGNETIC FIELD DUE TO TWO CURRENTS IN THE
6 REM SAME OR OPPOSITE DIRECTIONS.
7 REM
8REM IT SHOULD BE NOTED THAT THE PRINTOUT FOR EACH FIELD PLOT TAKES
9 REM ABOUT 4 MINUTES
10 REM
11 REM  REVISED BY L. BRAUN AND C. LOSIK  7-28-70
12 REM
13 GO TO 240
20 PRINT"THIS PROGRAM WILL PERMIT YOU TO EXPLORE THE MAGNETIC FIELD"
30 PRINT"ABOUT A CURRENT DIRECTED INTO THE PAGE AS A FUNCTION OF THE"
40 PRINT"CURRENT MAGNITUDE."
50 PRINT
60 PRINT"WHAT WILL BE YOUR INITIAL CURRENT (SELECT POSITIVE VALUES"
70 PRINT"BETWEEN 1 AND 8 AMPERES)."
80 PRINT"ENTER YOUR VALUE OF CURRENT";
90 INPUT I1
95 IF ABS(I1)>8 THEN 110
100 IF ABS(I1)>=1 THEN 130
110 PRINT"C'MON NOW-- ENTER PROPER VALUES."
120 GOTO 80
130 IF K>0 THEN 180
140 PRINT
150 PRINT "THE MAGNITUDE OF THE FIELD DECREASES FROM 9 TO 0. "
155 PRINT "9 IS THE HIGHEST POSSIBLE FIELD STRENGTH, AND 0"
160 PRINT "(WHICH MEANS A ZERO FIELD) THE LOWEST."
170 PRINT
180 GOSUB 440
190 LET K=K+1
200 PRINT
210 IF K=2 THEN 240
220 PRINT"SELECT A DIFFERENT CURRENT."
230 GOTO 80
240 PRINT"WOULD YOU LIKE TO TRY TWO DIFFERENT CURRENTS AT THE SAME"
250 PRINT"TIME (YES=1; NO=0)";
260 INPUT Q
270 IF Q=1 THEN 330
275 IF Q<>0 THEN 240
280 PRINT"WOULD YOU RATHER TRY ANOTHER CURRENT (YES=1; NO=0)";
290 INPUT P
300 IF P=1 THEN 80
305 IF P<>0 THEN 280
310 PRINT"WELL I GUESS YOU'RE ALL THROUGH. THANKS-- SEE Y'A"
320 STOP
330 PRINT"THE TWO CURRENTS WILL BE SEPARATED BY 1.0 METER. (NOTE: IF"
340 PRINT"THE CURRENTS ARE TO BE OPPOSITELY DIRECTED, STATE ONE OF THEM"
350 PRINT"AS A NEGATIVE VALUE)."
360 PRINT"ENTER THE TWO CURRENTS";
370 INPUT I1,I2
380 PRINT
390 GOSUB 440
400 PRINT"WANT TO TRY AGAIN (YES=1; NO=0)";
410 INPUT P
420 IF P=1 THEN 330
423 IF P=0 THEN 310
425 GO TO 400
440 PRINT
450 PRINT " ","  ","METERS"
460 PRINT" ","-1.8      -.6        0        .6        1.8"
```

```
470 PRINT "     METERS",".+.........+.........+.........+.........+."
480 FOR Y=1.2 TO -1.2 STEP(-.1)
500 PRINT "         ";INT(10*Y+.5)/10,".";
530 FOR Z=-2 TO 2 STEP(.1)
540 LET X=.6*Z
545 LET Y2=Y*Y
550 IF Q<>1 THEN 730
560 IF ABS(Y)>.001 THEN 640
570 IF ABS(Z+.8)>.001 THEN 610
580 IF I1>0 THEN 600
590 PRINT "-";
595 GO TO 945
600 PRINT "+";
605 GO TO 945
610 IF ABS(Z-.8)>.001 THEN 640
620 IF I2>0 THEN 600
630 GO TO 590
635 REM  R1 AND R2 ARE SQUARES !!!
640 LET X1=X+.5
650 LET X2=X-.5
660 LET R1=X1*X1+Y2
670 LET R2=X2*X2+Y2
680 REM  PARALLEL WIRES
690 LET H1=I1*Y/R1+I2*Y/R2
695 REM  H2 IS REALLY MINUS OF WHAT WE HAVE
700 LET H2=I1*X1/R1+I2*X2/R2
710 LET B=SQR(H1*H1+H2*H2)
720 GO TO 760
730 LET R=SQR(X*X+Y2)
740 IF ABS(R)<.001 THEN 580
750 LET B=ABS(I1/R)
760 IF B>.001 THEN 800
770 PRINT "0";
780 GO TO 945
800 FOR J=1 TO 9
810 IF ABS(B-2.5*J)<.75 THEN 840
820 NEXT J
825 PRINT " ";
830 GO TO 945
840 IF J>5 THEN 900
850 IF J<>1 THEN 860
853 PRINT "1";
856 GO TO 945
860 IF J<>2 THEN 870
863 PRINT "2";
866 GO TO 945
870 IF J<>3 THEN 880
873 PRINT "3";
876 GO TO 945
880 IF J<>4 THEN 890
883 PRINT "4";
886 GO TO 945
890 IF J<>5 THEN 900
893 PRINT "5";
896 GO TO 945
900 IF J<>6 THEN 910
903 PRINT "6";
906 GO TO 945
910 IF J<>7 THEN 920
913 PRINT "7";
916 GO TO 945
920 IF J<>8 THEN 930
923 PRINT "8";
926 GO TO 945
930 IF J<>9 THEN 825
933 PRINT "9";
945 NEXT Z
950 PRINT "."
955 NEXT Y
960 PRINT " ",".+.........+.........+.........+.........+."
970 PRINT
980 RETURN
990 END
```

DISCIPLINE PHYSICS

SUBJECT THE BOHR ATOM AND PHOTON EMISSION

PROGRAM NAME BOHR

DESCRIPTION:

The student may choose to have the Lyman, Balmer, or Paschen Series of the hydrogen emission spectrum displayed. He then must decide which energy level transitions are responsible for the lines of the spectrum that he has chosen. If he is successful, an energy-level diagram is presented and he must determine the energies of the photons emitted by the electron as it falls between randomly-selected energy levels.

OBJECTIVES:

To give an increased understanding of the Bohr atom and of how emission spectra are formed.

PRELIMINARY PREPARATION:

A. Student - The student should have been introduced to the Bohr atom, quantum theory, and ideally, have measured the wavelengths of the bright lines of the hydrogen spectrum.

B. Materials - A piece of paper and a pencil.

DISCUSSION:

After the student selects the series he wishes to see, it is displayed and he tries to discover which quantum level jumps by the electron are responsible for the first two of three lines in the series. If he is successful three times, a statement as to how the lines of that series are formed is printed and he may then elect to try another series or move on to work with the energy-level diagram for hydrogen.

After a brief explanation concerning the energy of a photon emitted during the transition of the electron from a higher to a lower energy level, the student is given a chance to show what he has learned. Energy levels are randomly selected and he must calculate the energy of the emitted photon. If the student is not successful, he gets a further explanation. After six trials the program ends.

```
YOU MAY VIEW THE 1. LYMAN  2.BALMER  OR 3. PASCHEN
SERIES BY TYPING IN THE NUMBER OF THE SERIES YOU WANT
DISPLAYED. OR TYPE 4 FOR AN ENERGY LEVEL DIAGRAM.

  CHOOSE THE NUMBER OF THE PART YOU WOULD LIKE TO SEE.
? 2

 7000  A
 6900  A
 6800  A
 6700  A
 6600  A
 6500  A---------- 6564.706
 6400  A
 6300  A
 6200  A
 6100  A
 6000  A
 5900  A
 5800  A
 5700  A
 5600  A
 5500  A
 5400  A
 5300  A
 5200  A
 5100  A
 5000  A
 4900  A
 4800  A---------- 4862.745
 4700  A
 4600  A
 4500  A
 4400  A
 4300  A---------- 4341.737
 4200  A
 4100  A---------- 4102.941
 4000  A
 3900  A---------- 3971.842
 3800  A---------- 3890.196
 3700  A
 3600  A---------- 3647.059 -----SERIES LIMIT
 3500  A
 3400  A
 3300  A
 3200  A
 3100  A
 3000  A
```

ACCORDING TO THE BOHR THEORY EACH OF THESE LINES RESULTS FROM THE EMISSION OF A PHOTON DURING THE TRANSITION OF THE ORBITAL ELECTRON OF AN EXCITED HYDROGEN ATOM FROM A HIGHER ENERGY STATE (ORBIT) TO A LOWER ONE. IN A PARTICULAR SERIES THE TRANSITION (JUMP) IS ALWAYS INTO THE SAME LOWER LEVEL (ORBIT) FROM ANY HIGHER ONE.
LET'S SEE IF YOU CAN DETERMINE WHICH TWO ORBITS THE ELECTRON JUMPED BETWEEN TO GIVE THE LINES THAT HAVE BEEN DISPLAYED.
THE LOWEST ENERGY LEVEL (GROUND STATE) IS NUMBERED ONE. HIGHER ENERGY LEVELS HAVE HIGHER NUMBERS IN SEQUENCE.

FOR EXAMPLE: FROM 4 TO 1 ENTER AS 4,1? 3,4

HEY!! FROM A HIGHER TO A LOWER ENERGY LEVEL.

FOR EXAMPLE: FROM 4 TO 1 ENTER AS 4,1? 3,2

GOOD START. THAT GIVES A WAVELENGTH OF 6564.706
THE NEXT LINE IS FORMED BY WHICH TRANSITION? 4?,2

BY GEORGE!! I THINK YOU'VE GOT IT!! THE WAVELENGTH IS 4862.745
TRY ONE MORE - THE NEXT ONE. ENTER NOW.? 5,2

ANY TRANSITION FROM A HIGHER ENERGY LEVEL INTO THE SECOND ENERGY LEVEL YIELDS A PHOTON OF THE BALMER SERIES.

IF YOU WOULD LIKE TO TRY ANOTHER SERIES TYPE IN THE NUMBER OF THAT SERIES. IF YOU WANT TO GO ON TO A NEW PART OF THE PROGRAM TYPE 4
WHICH?? 4

YOU WILL NOW GET AN ENERGY LEVEL DIAGRAM FOR HYDROGEN. IT SHOWS THE ENERGY OF THE ELECTRON IN THE VARIOUS ENERGY LEVELS. THE DIFFERENCE BETWEEN THE ENERGY OF THE ELECTRON IN A HIGHER LEVEL AND THAT IN A LOWER LEVEL IS THE ENERGY OF THE EMITTED PHOTON. E(PHOTON) = E (HIGHER) - E (LOWER)

CONTINUUM

BOHR

```
N= .....INFINITY..... E=    000000
N= 6    --------------- E= - .370001
N= 5    --------------- E= - .540001
N= 4    --------------- E= - .850001

N= 3    --------------- E= - 1.510001



N= 2    --------------- E= - 3.400001
```

```
N= 1    --------------- E= - 13.6

FIND THE ENERGIES OF THE PHOTONS GIVEN OFF FOR THE
TRANSITIONS GIVEN BELOW.

FROM LEVEL 2  TO LEVEL 1  THE ENERGY OF THE PHOTON IS?? 10.2

GOOD. TRY ANOTHER

FROM LEVEL 4  TO LEVEL 2  THE ENERGY OF THE PHOTON IS?? 2.65

THE ENERGY OF LEVEL 4  IS-.85
THE ENERGY OF LEVEL 2  IS-3.4

THEIR DIFFERENCE = PHOTON ENERGY = 2.55

FROM LEVEL 4  TO LEVEL 2  THE ENERGY OF THE PHOTON IS?? 2.55

GOOD. TRY ANOTHER

FROM LEVEL 2  TO LEVEL 1  THE ENERGY OF THE PHOTON IS?? 10.2

GOOD. TRY ANOTHER

FROM LEVEL 5  TO LEVEL 2  THE ENERGY OF THE PHOTON IS?? 2.86

GOOD. TRY ANOTHER

FROM LEVEL 5  TO LEVEL 2  THE ENERGY OF THE PHOTON IS?? 2.86

THANK YOU, AND GOODBYE.

READY
```

```
1 REM JOHN HOSIE - NORTHPORT HIGH - 7/23/69
90 LET T=8
100 REM  REVISED BY C.LOSIK  8-21-70
105 REM  M IS WHICH PROBLEM, L IS A SERIES WAVELENGTH,A AND B ARE
106 REM  UPPER AND LOWER LIMITS OF POSSIBLE SERIES VALUES
120 RANDOMIZE
130 PRINT "YOU MAY VIEW THE 1. LYMAN  2.BALMER  OR 3. PASCHEN"
140 PRINT "SERIES BY TYPING IN THE NUMBER OF THE SERIES YOU WANT"
150 PRINT "DISPLAYED, OR TYPE 4 FOR AN ENERGY LEVEL DIAGRAM."
160 PRINT
170 PRINT "  CHOOSE THE NUMBER OF THE PART YOU WOULD LIKE TO SEE.";
180 PRINT
190 LET G=1
200 DIM S(10)
210 LET J=0
220 INPUT M
222 FOR Q=1 TO 4
224 IF M=Q THEN 235
226 NEXT Q
228 PRINT "1, 2, 3, OR 4 ONLY, PLEASE!"
230 GO TO 220
235 PRINT
240 IF M=4 THEN 1200
250 LET N=M+1
260 LET D=12400*M↑2/13.6
270 LET D1=INT (.01*D)
280 GO TO 450
290 FOR I=A TO B STEP -1
300 LET E=13.6*(1/M↑2-1/N↑2)
310 LET L=12400/E
320 LET P=INT (.01*L)
330 IF I=D1 THEN 430
340 IF I=P THEN 370
350 PRINT 100*I" A"
360 GO TO 410
370 LET J=J+1
380 LET S(J)=L
390 PRINT 100*I" A----------"L
400 LET N=N+1
410 NEXT I
420 GO TO 590
430 PRINT 100*I" A----------"12400*N*M/13.6"-----SERIES LIMIT"
440 GO TO 410
450 LET Y=12400*(M↑2*N↑2)/(13.6*(N↑2-M↑2))
460 LET Y=INT(.01*Y)
470 IF Y<15 THEN 500
480 IF Y<70 THEN 530
490 IF Y<190 THEN 560
500 LET A=15
510 LET B=5
520 GO TO 290
530 LET A=70
540 LET B=30
550 GO TO 290
560 LET A=190
570 LET B=78
575 IF G>1 THEN 680
580 GO TO 290
590 PRINT
600 PRINT "  ACCORDING TO THE BOHR THEORY EACH OF THESE LINES RESULTS"
```

```
610 PRINT "FROM THE EMISSION OF A PHOTON DURING THE TRANSITION OF THE"
620 PRINT "ORBITAL ELECTRON OF AN EXCITED HYDROGEN ATOM FROM A HIGHER"
630 PRINT "ENERGY STATE (ORBIT) TO A LOWER ONE.  IN A PARTICULAR SERIES"
640 PRINT "THE TRANSITION (JUMP) IS ALWAYS INTO THE SAME LOWER LEVEL"
650 PRINT "(ORBIT) FROM ANY HIGHER ONE."
655 LET G=G+1
660 PRINT "  LET'S SEE IF YOU CAN DETERMINE WHICH TWO ORBITS THE ELECTRO
670 PRINT "JUMPED BETWEEN TO GIVE THE LINES THAT HAVE BEEN DISPLAYED."
680 PRINT "  THE LOWEST ENERGY LEVEL (GROUND STATE) IS NUMBERED ONE."
690 PRINT "HIGHER ENERGY LEVELS HAVE HIGHER NUMBERS IN SEQUENCE."
700 LET F=1
710 PRINT
720 PRINT
730 PRINT "  FOR EXAMPLE: FROM 4 TO 1 ENTER AS 4,1";
735 LET T=T+1
740 INPUT N,M1
750 PRINT
760 IF N>M1 THEN 820
770 IF F<>1 THEN 800
780 PRINT "HEY!! FROM A HIGHER TO A LOWER ENERGY LEVEL."
790 GO TO 710
800 PRINT "OOPS - FROM HIGHER TO LOWER."
810 GO TO 710
820 LET L1=12400*(M1↑2*N↑2)/(13.6*(N↑2-M1↑2))
830 IF ABS(L1-S(F))<.005 THEN 890
840 IF F>1 THEN 870
850 PRINT "NOPE!! TRY AGAIN."
860 GO TO 710
870 PRINT "SORRY.  TRY AGAIN!  YOUR FINAL ENERGY LEVEL SHOULD BE"M
880 GO TO 710
890 IF F=1 THEN 930
900 IF M=1 THEN 1010
910 IF F=2 THEN 970
920 GO TO 1010
930 PRINT "GOOD START.  THAT GIVES A WAVELENGTH OF"L1
940 PRINT "THE NEXT LINE IS FORMED BY WHICH TRANSITION";
950 LET F=F+1
960 GO TO 740
970 PRINT "BY GEORGE!! I THINK YOU'VE GOT IT!! THE WAVELENGTH IS"L1
980 PRINT "TRY ONE MORE - THE NEXT ONE.  ENTER NOW.";
990 LET F=F+1
1000 GO TO 740
1010 PRINT
1020 PRINT "  ANY TRANSITION FROM A HIGHER ENERGY LEVEL INTO THE"
1030 IF M=1 THEN 1080
1040 IF M=2 THEN 1110
1050 PRINT "THIRD ENERGY LEVEL CAUSES THE EMISSION OF A PHOTON OF THE"
1060 PRINT "PASCHEN SERIES."
1070 GO TO 1120
1080 PRINT "GROUND STATE IS ACCOMPANIED BY THE EMISSION OF A PHOTON OF"
1090 PRINT "LIGHT BELONGING TO THE LYMAN SERIES."
1100 GO TO 1120
1110 PRINT "SECOND ENERGY LEVEL YIELDS A PHOTON OF THE BALMER SERIES."
1120 PRINT
1130 PRINT "IF YOU WOULD LIKE TO TRY ANOTHER SERIES TYPE IN THE NUMBER"
1140 PRINT "OF THAT SERIES.  IF YOU WANT TO GO ON TO A NEW PART OF"
1150 PRINT "THE PROGRAM TYPE 4"
1160 PRINT "WHICH?";
1190 GO TO 220
1200 PRINT
1210 PRINT "  YOU WILL NOW GET AN ENERGY LEVEL DIAGRAM FOR HYDROGEN."
1220 PRINT "IT SHOWS THE ENERGY OF THE ELECTRON IN THE VARIOUS ENERGY"
1230 PRINT "LEVELS.  THE DIFFERENCE BETWEEN THE ENERGY OF THE ELECTRON"
1240 PRINT "IN A HIGHER LEVEL AND THAT IN A LOWER LEVEL IS THE ENERGY"
```

```
1250 PRINT "OF THE EMITTED PHOTON.   E(PHOTON) = E (HIGHER) - E (LOWER)"
1260 PRINT
1270 PRINT "         CONTINUUM"
1280 PRINT
1290 PRINT "N= .....INFINITY..... E=    000000"
1300 LET N=6
1310 FOR I=1 TO 40
1320 LET Y=INT(40/N↑2+.56)
1330 IF I=Y THEN 1360
1340 PRINT
1350 GO TO 1380
1360 PRINT "N="N"  ---.------------ E= -"INT(1360/N↑2)/100+.000001
1370 LET N=N-1
1380 NEXT I
1390 PRINT
1400 PRINT "FIND THE ENERGIES OF THE PHOTONS GIVEN OFF FOR THE"
1410 PRINT "TRANSITIONS GIVEN BELOW."
1420 LET J=0
1430 FOR I=0 TO T*Q
1440 LET Y3=RND(I)
1450 NEXT I
1460 FOR I=1 TO 20
1470 LET Y3=INT(1+(5*RND(I)))
1480 LET Y4=INT(1+(5*RND(I+1)))
1490 IF Y4>Y3 THEN 1510
1500 NEXT I
1510 PRINT
1520 PRINT "FROM LEVEL"Y4" TO LEVEL"Y3" THE ENERGY OF THE PHOTON IS?";
1530 INPUT E1
1
1540 LET J=J+1
1550 PRINT
1560 LET E=-13.6*(1/(Y4↑2)-1/(Y3↑2))
1570 IF ABS(E1-E)>.005 THEN 1610
1580 IF J=6 THEN 1660
1590 PRINT "GOOD. TRY ANOTHER"
1600 GO TO 1460
1610 PRINT "THE ENERGY OF LEVEL"Y4 " IS"-13.6/(Y4↑2)
1620 PRINT "THE ENERGY OF LEVEL"Y3 " IS"-13.6/(Y3↑2)
1630 PRINT
1640 PRINT "THEIR DIFFERENCE = PHOTON ENERGY ="E
1650 GO TO 1460
1660 PRINT "THANK YOU, AND GOODBYE."
1670 END
```

DISCIPLINE PHYSICS

SUBJECT CALORIMETRY

PROGRAM NAME CALORI

DESCRIPTION:

Calorimetry experiments are simulated by the computer permitting the student to enter the mass and temperatures of two quantities of water. The computer calculates and prints out the equilibrium temperature of the mixture. The student must then determine the heat energy, in calories, to be supplied (or removed) from each mass to obtain the equilibrium temperature.

OBJECTIVES:

A. To acquaint the students with conservation of energy concepts involving calorimetry.

B. To determine the equations governing these relationships.

PRELIMINARY PREPARATION:

A. Student - Must know definitions for calorie and specific heat.

B. Materials - Table of Specific heats

DISCUSSION:

Calorimetry, in its simplest form, is presented as part of a class lesson. The concept of heat energy balance is developed by presenting several examples, with the computer, based on the definition of the "calorie." Specific heat is introduced by a similar approach (replacing the water of the initial examples, with alcohol; specific heat of .6 cal/gm-0°C.)

The program can be modified (with relative ease) to incorporate different materials or combinations of different materials.

When this program was used as an introduction to calorimetry, it was noted that many students were able to determine the equations describing the phenomenon by utilizing the stated results from the computer.

HEAT AND CALORIMETRY

YOU HAVE TWO BEAKERS OF WATER .
WHAT IS THE MASS (IN GRAMS) AND THE TEMP (IN DEGREES) OF THE
WATER IN THE FIRST BEAKER? 80,50

WHAT IS THE MASS (IN GRAMS) AND THE TEMP (IN DEGREES) OF THE
WATER IN THE SECOND BEAKER? 40,60

THE FINAL TEMPERATURE OF THE MIXTURE IS 53.33 DEGREES.

HOW MANY CALORIES WERE INVOLVED IN CHANGING THE TEMP OF
THE FIRST BEAKER FROM 50 TO 53.33 DEGREES? 260

YOU'RE CLOSE ENOUGH. THE CORRECT ANSWER IS 266.4 CALORIES.

HOW MANY CALORIES WERE INVOLVED IN CHANGING THE TEMP OF
THE SECOND BEAKER FROM 60 TO 53.33 DEGREES? 240

YOU'RE MORE THAN 3 PERCENT OFF. YOU SHOULD HAVE SAID
266.8 CALORIES.

WANT TO TRY AGAIN (1=YES, 0=NO) ? ? 1
CHOOSE A LIQUID : 0=WATER, 1=ALCOHOL. WHICH? 1

YOU HAVE TWO BEAKERS OF ALCOHOL .
WHAT IS THE MASS (IN GRAMS) AND THE TEMP (IN DEGREES) OF THE
ALCOHOL IN THE FIRST BEAKER? 100,50

WHAT IS THE MASS (IN GRAMS) AND THE TEMP (IN DEGREES) OF THE
ALCOHOL IN THE SECOND BEAKER? 100,70

THE FINAL TEMPERATURE OF THE MIXTURE IS 60 DEGREES.

HOW MANY CALORIES WERE INVOLVED IN CHANGING THE TEMP OF
THE FIRST BEAKER FROM 50 TO 60 DEGREES? 1000

YOU'RE MORE THAN 3 PERCENT OFF. YOU SHOULD HAVE SAID
600 CALORIES.

HOW MANY CALORIES WERE INVOLVED IN CHANGING THE TEMP OF
THE SECOND BEAKER FROM 70 TO 60 DEGREES? 600

YOU'RE CLOSE ENOUGH. THE CORRECT ANSWER IS 600 CALORIES.

WANT TO TRY AGAIN (1= , 0=NO) : ? 0

READY

```
1 REM A.C. CAGGIANO; PATCHOGUE H.S.; PHYSICS; 8-'69
2 REM THIS PROGRAM INVOLVES CALORIMETRY EXPERIMENTS OR THEIR
3 REM SIMULATION.
5 REM  REVISED BY C.LOSIK 8-25-70
6 REM  K TELLS WHICH LIQUID, J TELLS WHICH BEAKER.
7 REM  M(J) ARE THE MASSES OF LIQUID, T(J) ARE THEIR TEMPERATURES
80 LET K=0
90 DIM M(2),T(2)
100 PRINT " ","HEAT AND CALORIMETRY"
110 PRINT
112 PRINT
114 PRINT "##########"
116 PRINT
120 PRINT"YOU HAVE TWO BEAKERS OF";
130 GOSUB 590
140 PRINT"."
150 FOR J=1 TO 2
160 PRINT"WHAT IS THE MASS (IN GRAMS) AND THE TEMP (IN DEGREES) OF THE"
170 GOSUB 590
180 PRINT"IN THE";
190 GOSUB 540
200 PRINT"BEAKER";
210 INPUT M(J),T(J)
220 PRINT
230 NEXT J
240 LET T3=(M(1)*T(1)+M(2)*T(2))/(M(1)+M(2))
245 LET T3=INT(100*T3+.5)/100
250 PRINT"THE FINAL TEMPERATURE OF THE MIXTURE IS "T3" DEGREES."
260 PRINT
270 FOR J=1 TO 2
280 PRINT"HOW MANY CALORIES WERE INVOLVED IN CHANGING THE TEMP OF"
290 PRINT"THE ";
300 GOSUB 540
310 PRINT" BEAKER ";
320 PRINT"FROM "T(J)" TO "T3" DEGREES";
330 INPUT H
335 PRINT
340 LET G=ABS(H)
350 LET T=ABS(T3-T(J))
360 IF G<>0 THEN390
370 IF S*M(J)*T=0 THEN 400
380 GOTO 430
390 IF ABS((G-S*M(J)*T)/G)>.03 THEN 430
400 PRINT"YOU'RE CLOSE ENOUGH. THE CORRECT ANSWER IS ";
405 GOTO 440
410 PRINT
420 GOTO460
430 PRINT"YOU'RE MORE THAN 3 PERCENT OFF. YOU SHOULD HAVE SAID"
440 PRINT INT(100*S*M(J)*T+.5)/100 " CALORIES."
450 PRINT
460 NEXT J
465 PRINT
470 PRINT "WANT TO TRY AGAIN (1=YES, 0=NO) ? ";
475 INPUT Q
480 IF Q=0 THEN 660
483 IF Q<>1 THEN 465
485 PRINT "CHOOSE A LIQUID ? 0=WATER, 1=ALCOHOL.  WHICH";
490 INPUT K
500 IF K*(K-1)<>0 THEN 485
510 GO TO 110
540 IF J=2 THEN 570
550 PRINT" FIRST ";
560 GOTO 580
570 PRINT" SECOND ";
580 RETURN
590 IF K=1 THEN 630
600 PRINT" WATER ";
610 LET S=1
620 GOTO650
630 PRINT" ALCOHOL ";
640 LET S=.6
650 RETURN
660 END
```

DISCIPLINE PHYSICS

SUBJECT RADIOACTIVE DECAY

PROGRAM NAME DECAY1

DESCRIPTION:

Radioactive decay is treated pseudo-quantitatively, by permitting the student to determine the approximate number of radioactive particles remaining after various times.

OBJECTIVES:

To induce a "feel" for exponential decay, by repeated exercises.

PRELIMINARY PREPARATION:

A. Student - Awareness of terms: half-life, exponential, and radioactivity

B. Materials - none

DISCUSSION:

The concept of radioactive decay is presented playfully as a game, allowing the student to challenge his own ability in determining (with 5, 10, or 20% error), the number of radioactive "chips" remaining after various times. The number of chips successively decreases with each trial, increasing the level of difficulty as the program runs. In each case, the exact number remaining is given, following the students' entered value.

Individuals or small groups, find this program exciting. They enjoy the game approach, at least the first time through it, and seem to be motivated by the opportunity to " break the bank."

This program can be used as an integral part of a class lesson to introduce the concept, or to motivate group discussion and participation concerning the phenomena.

```
                ---THE NEW CLEA CASINO---

     MR. A. TOM MICK, GENERAL MANAGER OF THE NEW CLEA CASINO,
HAS, AT TIME  T=0,  DISCOVERED  100,000 RADIOACTIVE PLAYING
CHIPS AT HIS TABLE. THEIR HALF-LIFE IS 10 MINUTES. EACH CHIP
TRANSMUTES SPONTANEOUSLY AND COMPLETELY IN A RANDOM FASHION.

     AT VARIOUS TIMES T, AFTER T=0, YOU MUST DETERMINE WITHIN
A CERTAIN PERCENTAGE, HOW MANY CHIPS ARE LEFT.

     TO FURTHER THE INTEREST OF THE GAME, YOU WILL START WITH
$1,000 AND THE HOUSE WITH AN UNSPECIFIED AMOUNT.  HALF THE
MONEY YOU HAVE WILL RIDE ON EACH GUESS YOU TAKE.  LET'S SEE
IF YOU CAN BREAK THE HOUSE BEFORE THE CHIPS RUN OUT.

THE HOUSE OFFERS THE FOLLOWING ODDS:
     2) 2 TO 1 ODDS FOR GUESSING WITHIN 20 PERCENT
     4) 4 TO 1 ODDS FOR GUESSING WITHIN 10 PERCENT
     8) 8 TO 1 ODDS FOR GUESSING WITHIN  5 PERCENT.

ENTER THE NUMBER 2, 4, OR 8 FOR THE ODDS YOU WANT AFTER THE
QUESTION MARK IN THE COLUMN LABELLED ODDS.

YOUR $          HOUSE $         TIME (MIN)      ODDS

 1000           1.000000E+6     7.2             ? 8
HOW MANY CHIPS LEFT ? 60700
ACTUAL NUMBER LEFT IS  60716
YOU WON. TRY AGAIN.

 5000           996000          13.9            ? 8
HOW MANY CHIPS LEFT ? 38150
ACTUAL NUMBER LEFT IS  38164
YOU WON. TRY AGAIN.

 25000          976000          26.9            ? 8
HOW MANY CHIPS LEFT ? 15500
ACTUAL NUMBER LEFT IS  15502
YOU WON. TRY AGAIN.

 125000         876000          30.7            ? 8
HOW MANY CHIPS LEFT ? 11900
ACTUAL NUMBER LEFT IS  11913
YOU CAN BREAK THE HOUSE IF YOU TRY A LONG SHOT.

 625000         376000          48.4            ? 8
HOW MANY CHIPS LEFT ? 3500
ACTUAL NUMBER LEFT IS  3494
YOU BROKE THE HOUSE. YOU NEEDED ONLY THE MINIMUM NUMBER OF GUESSES.
CONGRATULATIONS.
YOU MUST KNOW A LOT ABOUT RADIOACTIVITY AND THINGS.
THANKS FOR PLAYING..

-----------------------------------------------------------

                                        CHECK NO. 3499

                                      DATE: ----------19--

 PAY TO THE ORDER OF---------CASH-------$ 1.001000E+6

 THE NEW CLEA CASINO                    A. TOM MICK
                                        GENERAL MANAGER

-----------------------------------------------------------

DONT SPEND IT ALL IN ONE PLACE.

READY
```

```
100 REM  RICHARD P. PAV, PATCHOGUE H.S., (PHYSICS)  REVISED NOV. 26,1968
105 RANDOMIZE
110 REM  THIS IS A GAME BASED ON RADIOACTIVE DECAY.
120 PRINT "               ---THE NEW CLEA CASINO---"
130 PRINT
140 PRINT "     MR. A. TOM MICK, GENERAL MANAGER OF THE NEW CLEA CASINO"
150 PRINT "HAS, AT TIME  T=0,  DISCOVERED  100,000 RADIOACTIVE PLAYING"
160 PRINT "CHIPS AT HIS TABLE. THEIR HALF-LIFE IS 10 MINUTES. EACH CHIP"
170 PRINT "TRANSMUTES SPONTANEOUSLY AND COMPLETELY IN A RANDOM FASHION."
180 PRINT
190 PRINT "     AT VARIOUS TIMES T, AFTER T=0, YOU MUST DETERMINE WITHIN
200 PRINT "A CERTAIN PERCENTAGE, HOW MANY CHIPS ARE LEFT."
210 PRINT
220 PRINT "     TO FURTHER THE INTEREST OF THE GAME, YOU WILL START WITH
230 PRINT "$1,000 AND THE HOUSE WITH AN UNSPECIFIED AMOUNT.  HALF THE"
240 PRINT "MONEY YOU HAVE WILL RIDE ON EACH GUESS YOU TAKE.  LET'S SEE"
250 PRINT "IF YOU CAN BREAK THE HOUSE BEFORE THE CHIPS RUN OUT."
260 PRINT
270 PRINT "THE HOUSE OFFERS THE FOLLOWING ODDS:"
280 PRINT "     2) 2 TO 1 ODDS FOR GUESSING WITHIN 20 PERCENT"
290 PRINT "     4) 4 TO 1 ODDS FOR GUESSING WITHIN 10 PERCENT"
300 PRINT "     8) 8 TO 1 ODDS FOR GUESSING WITHIN  5 PERCENT."
310 PHINT
320 PRINT "ENTER THE NUMBER 2, 4, OR 8 FOR THE ODDS YOU WANT AFTER THE"
330 PRINT "QUESTION MARK IN THE COLUMN LABELLED ODDS."
340 PRINT
350 PRINT "YOUR $",  "HOUSE $",  "TIME (MIN)",  "ODDS"
360 LET A=0
370 LET B=0
380 LET T=0
390 LET Y=1000
400 LET C=0
410 PRINT
420 IF ABS(G-D)<1500 THEN 450
430 LET G=5
440 LET D=2
450 LET B=B+1
460 FOR I=1 TO 3+A+ABS(G-D)
470 LET T3=INT(100*RND(-Y))/10
480 NEXT I
490 LET T=T+T3
500 LET D=INT(1E5*EXP(-.0693*T))
510 IF D=0 THEN 860
520 PRINT Y,1001000-Y,T,
530 INPUT A
540 IF A=2 THEN 610
550 IF A=4 THEN 610
560 IF A=8 THEN 610
570 PRINT "SORRY PAL, WE DONT OFFER THOSE ODDS."
580 IF C=1 THEN 620
590 LET C=1
600 GOTO 520
610 PRINT "HOW MANY CHIPS LEFT ";
620 INPUT G
630 PRINT "ACTUAL NUMBER LEFT IS "ID
640 IF A=2 THEN 700
650 IF A=4 THEN 680
660 LET P=.05
670 GOTO 710
680 LET P=.1
690 GOTO 710
700 LET P=.2
```

```
710 LET T=10*B
720 IF ABS(D-G)<=P*D THEN 770
730 LET Y=INT(Y-Y/2)
740 IF Y<=50 THEN 820
750 PRINT "TOO BAD. YOU LOST. TRY AGAIN."
760 GOTO 400
770 LET Y=INT(Y+A*Y/2)
780 IF 1000000-Y<1 THEN 890
790 IF Y>2E5 THEN 840
800 PRINT "YOU WON. TRY AGAIN."
810 GOTO 400
820 PRINT "IT SEEMS YOU JUST CANT GET THE HANG OF IT. SAVE YOUR BREAD."
830 GOTO 960
840 PRINT "YOU CAN BREAK THE HOUSE IF YOU TRY A LONG SHOT."
850 GOTO 400
860 PRINT "OOOOPS... SORRY PAL. THE LAST CHIP JUST DISINTEGRATED."
870 PRINT "THE HOUSE IS CLOSED."
880 GOTO 960
890 PRINT "YOU BROKE THE HOUSE. YOU NEEDED ONLY ";
895 LET Y=1001000
900 IF B>5 THEN 930
910 PRINT "THE MINIMUM NUMBER OF GUESSES."
920 GOTO 940
930 PRINT B;"GUESSES."
940 PRINT "CONGRATULATIONS."
950 PRINT "YOU MUST KNOW A LOT ABOUT RADIOACTIVITY AND THINGS."
960 PRINT "THANKS FOR PLAYING.."
970 PRINT
980 PRINT "------------------------------------------------------------"
990 PRINT
1000 PRINT "                                        CHECK NO."B+D
1010 PRINT
1020 PRINT "                                      DATE: ";
1030 PRINT "-------------19--"
1040 PRINT
1050 PRINT
1060 PRINT " PAY TO THE ORDER OF--";
1070 PRINT "------CASH------";
1080 PRINT "$";Y
1090 PRINT
1100 PRINT
1110 PRINT " THE NEW CLEA CASINO                    A. TOM MICK"
1120 PRINT "                                        GENERAL MANAGER"
1130 PRINT
1140 PRINT "------------------------------------------------------------"
1150 PRINT
1160 PRINT "DONT SPEND IT ALL IN ONE PLACE."
1170END
```

DISCIPLINE CHEMISTRY-PHYSICS

SUBJECT NUCLEAR DECAY

PROGRAM NAME DECAY2

DESCRIPTION:

This program will do the following:

A. Calculate half-life from 2 readings on a geiger counter, and the time between them.

B. Calculate mass of a radioactive sample remaining after some given amount of time.

C. Prints out a table showing mass or number of particles of a radioactive sample remaining vs. some range of time.

OBJECTIVES:

A. To provide tables and graphs for a better understanding of the exponential decay of a radioactive substance.

B. To provide a calculator for determining the amount of mass of a radioactive sample remaining after some given amount of time.

C. To provide a calculator for half-life experiments.

PRELIMINARY PREPARATION:

A. Student - The student should have a general introduction to half-life before the use of the program.

B. Materials - none

DISCUSSION:

It is difficult to teach about the exponential (logarithmic) manner by which radioactive elements decay without meaningful illustrations and simulations.

DISCUSSION: (con' t)

With this program, a number of interesting possibilities are available. For example, if the initial mass is 100 g and the time is equal to 10 half-lives with an increment equal to the half-life, the student will see the mass decrease to 0.1 g during that time. More important, the example may be generalized to show that for any radioactive sample:

after 1 half-life 50% of the substance remains
after 2 half-life 25% of the substance remains
after 3 half-life 12.5% of the substance remains
after 10 half-life 0.1% of the substance remains

You may also illustrate nuclear decay by using particles instead of mass. Use Avogadro' s number of particles with students who feel comfortable with scientific notation. For the othe , you may use a number up to 1,000,000 without having exponential n bers print out in the table.

The fact that the teletype unit takes about 8 seconds to type out a line provides you with cute little gimmicks. Set up a run with 8 seconds (or any multiple of 8) and the print-out of the table will keep time with the decay of the sample substance.

Please note that the half-life calculations are not accurate for a small number of particles, thus it is misleading to make runs go to zero mass or zero particles.

```
DO YOU WANT INSTRUCTIONS (1=YES, 0=NO) : ? 1
  THIS PROGRAM WILL DO THE FOLLOWING:
       CHOICE 1 - CALCULATES HALF-LIFE FROM TWO READINGS
                  ON A GEIGER COUNTER.
       CHOICE 2 - CALCULATES HOW MUCH OF A RADIOACTIVE SAMPLE
                  WILL REMAIN AFTER SOME GIVEN AMOUNT OF TIME
       CHOICE 3 - PRINTS OUT A TABLE SHOWING MASS OF SAMPLE
                  VS. TIME OR NO. OF PARTICLES VS. TIME.
                  (GRAPH OPTIONAL) NOTE: FOR THE TABLE YOU
                  MUST INPUT TOTAL TIME AND TIME INCREMENT.
                   EXAMPLE: IF TOTAL TIME=100 AND TIME
                  INCREMENT=10, THEN TIME IN THE TABLE WILL
                  BE 10,20,30,.............,100.
        CHOICE 4 - END OF PROGRAM

                  NOTE: IN ANY ONE PROBLEM, TIME MUST
                  ALWAYS BE INPUTED IN THE SAME UNITS
                  OF MEASURE (IE: SECS.,MINS.,ETC.)

**********

WHAT IS YOUR CHOICE? 1

 WHAT IS THE INITIAL READING ON THE GEIGER COUNTER,
 THE SECOND READING, AND THE TIME BETWEEN READINGS.
? 1500,3000,36

INITIAL READING= 3000 SECOND READING= 1500 TIME= 36
HALF-LIFE= 35.99755

**********

WHAT IS YOUR CHOICE? 1

 WHAT IS THE INITIAL READING ON THE GEIGER COUNTER,
 THE SECOND READING, AND THE TIME BETWEEN READINGS.
? 775,1256,212
INITIAL READING= 1256 SECOND READING= 775 TIME= 212
HALF-LIFE= 304.3285

**********

WHAT IS YOUR CHOICE? 2

WHAT IS THE HALF-LIFE, INITIAL MASS OF SAMPLE, AND
TOTAL TIME OF DECAY? 18,56,76

HALF-LIFE= 18 INITIAL MASS= 56 TOTAL TIME= 76
MASS OF SAMPLE REMAINING= 3.000952

**********

WHAT IS YOUR CHOICE? 3

DO YOU WANT TO WORK WITH PARTICLES OR MASS? (ANSWER 1 FOR
 PARTICLES OR 2 FOR MASS) ? 1

 WHAT IS THE HALF-LIFE, INITIAL NUMBER OF PARTICLES IN THE
 SAMPLE, TOTAL ELAPSED TIME FOR DECAY, AND THE
 INCREMENT OF ELAPSED TIME? 10,6.02E23,100,10

HALF-LIFE= 10 INITIAL NO. OF PARTICLES= 6.020000E+23
TOTAL TIME= 100 INCREMENT= 10
```

```
TIME          PARTICLES     PART. LOSS     TOTAL PART. LOSS
----          ---------     ----- ----     ----- ----- ----

0             6.020000E+23  0              0
10            3.010142E+23  3.009858E+23   3.009858E+23
20            1.505142E+23  1.505000E+23   4.514858E+23
30            7.526065E+22  7.525355E+22   5.267393E+23
40            3.763210E+22  3.762855E+22   5.643679E+23
50            1.881694E+22  1.881516E+22   5.831831E+23
60            9.408913E+21  9.408026E+21   5.925911E+23
70            4.704679E+21  4.704235E+21   5.972953E+23
80            2.352450E+21  2.352228E+21   5.996475E+23
90            1.176281E+21  1.176170E+21   6.008237E+23
100           5.881561E+20  5.881126E+20   6.014118E+23


DO YOU WANT THE ABOVE DATA GRAPHED? (1-YES, 0-NO)? 1


                              MASS (OR PARTICLES) REMAINING

              0
 6.020000E+23
TIME          I---------I---------I---------I---------I---------I
 0            I                                                  *
 10           I                              *
 20           I                   *
 30           I         *                                        .
 40           I   *
 50           I *
 60           I*
 70           *
 80           *
 90           *
 100          *

**********

WHAT IS YOUR CHOICE? 3
DO YOU WANT TO WORK WITH PARTICLES OR MASS? (ANSWER 1 FOR
 PARTICLES OR 2 FOR MASS) ? 2

 WHAT IS THE HALF-LIFE, INITIAL MASS OF SAMPLE, TOTAL
ELAPSED TIME FOR DECAY, AND THE INCREMENT OF
ELAPSED TIME? 15,100,150,15

HALF-LIFE= 15 INITIAL MASS= 100 TOTAL TIME= 150 INCREMENT= 15

TIME          MASS          MASS LOSS      TOTAL MASS LOSS
----          ----          ---- ----      ----- ---- ----

 0            100           0              0
 15           50.00236      49.99764       49.99764
 30           25.00236      25             74.99764
 45           12.50177      12.50059       87.49823
 60           6.25118       6.25059        93.74882
 75           3.125737      3.125443       96.87426
 90           1.562942      1.562795       98.43706
 105          .7815081      .7814344       99.21849
 120          .3907725      .3907356       99.60923
 135          .1953955      .195377        99.8046
 150          .09770234     .09769313      99.90229
```

```
DO YOU WANT THE ABOVE DATA GRAPHED? (1-YES, 0-NO)? 1

                        MASS (OR PARTICLES) REMAINING

                0                                                       100
TIME            I---------I---------I---------I---------I---------I
 0              I                                                      *
 15             I                                    *
 30             I                          *
 45             I                *
 60             I       *
 75             I     *
 90             I   *
 105            I *
 120            I *
 135            I *
 150            I *

**********

WHAT IS YOUR CHOICE? 4


READY
```

```
100 REM R. DORFMAN+ C. MARCHISOTTO PTS 7/24/69
105 REM REVISED BY C.LOSIK 8-12-70
110 REM CALCULATION OF HALF-LIFE AND REMAINING MASS INCLUDING
120 REM TABLES AND GRAPHS.
122 PRINT "DO YOU WANT INSTRUCTIONS (1=YES, 0=NO) : ";
124 INPUT A
126 IF A=0 THEN 300
128 IF A<>1 THEN 122
130 PRINT "  THIS PROGRAM WILL DO THE FOLLOWING:"
140 PRINT "         CHOICE 1 - CALCULATES HALF-LIFE FROM TWO READINGS"
150 PRINT "                    ON A GEIGER COUNTER."
160PRINT"         CHOICE 2 - CALCULATES HOW MUCH OF A RADIOACTIVE SAMPLE"
170PRINT"                    WILL REMAIN AFTER SOME GIVEN AMOUNT OF TIME"
180 PRINT "         CHOICE 3 - PRINTS OUT A TABLE SHOWING MASS OF SAMPLE"
190 PRINT"                    VS. TIME OR NO. OF PARTICLES VS. TIME."
200 PRINT"                    (GRAPH OPTIONAL) NOTE: FOR THE TABLE YOU"
210 PRINT"                    MUST INPUT TOTAL TIME AND TIME INCREMENT."
220PRINT"                    EXAMPLE: IF TOTAL TIME=100 AND TIME"
230 PRINT"                    INCREMENT=10, THEN TIME IN THE TABLE WILL"
240 PRINT"                    BE 10,20,30,.............,100."
250 PRINT"         CHOICE 4 - END OF PROGRAM"
260 PRINT
270 PRINT"                    NOTE: IN ANY ONE PROBLEM,TIME MUST"
280 PRINT"                    ALWAYS BE INPUTED IN THE SAME UNITS"
290 PRINT"                    OF MEASURE (IE: SECS.,MINS.,ETC.)"
300 PRINT
310 PRINT "**********"
320 PRINT
330 PRINT "WHAT IS YOUR CHOICE";
340 INPUT A
350PRINT
360 IF A=1 THEN 410
370 IF A=2 THEN 490
380 IF A=3 THEN 570
390 IF A<>4 THEN 320
400 STOP
410PRINT" WHAT IS THE INITIAL READING ON THE GEIGER COUNTER,"
420 PRINT" THE SECOND READING, AND THE TIME BETWEEN READINGS."
430 INPUT B,A,C
433 IF A>B THEN 440
435 PRINT "INITIAL READING IS ALWAYS LESS THAN FINAL READING."
437  GO TO 430
440 LET D=(.6931*C)/LOG(A/B)
450 PRINT
460 PRINT "INITIAL READING="A;"SECOND READING="B;"TIME="C
470 PRINT "HALF-LIFE="D
480 GO TO 300
490 PRINT "WHAT IS THE HALF-LIFE, INITIAL MASS OF SAMPLE, AND"
500 PRINT "TOTAL TIME OF DECAY";
510 INPUT E,F,G
520 LET H=F*EXP(-.6931*G/E)
530 PRINT
540PRINT"HALF-LIFE="E;"INITIAL MASS="F;"TOTAL TIME="G
550 PRINT "MASS OF SAMPLE REMAINING="H
560 GO TO 300
570 PRINT "DO YOU WANT TO WORK WITH PARTICLES OR MASS? (ANSWER 1 FOR"
580 PRINT" PARTICLES OR 2 FOR MASS) ";
590 INPUT J
600 PRINT
610 IF J=1 THEN 760
615 IF J<>2 THEN 570
620 PRINT" WHAT IS THE HALF-LIFE, INITIAL MASS OF SAMPLE,TOTAL"
630 PRINT"ELAPSED TIME FOR DECAY, AND THE INCREMENT OF "
640 PRINT"ELAPSED TIME";
650 INPUT E,F,K,M
660LET W=0
670 LET Q=0
```

128

```
600 LET Z=F
690 PRINT
700 IF U=1 THEN 800
710 PRINT"HALF-LIFE="B;"INITIAL MASS="F;"TOTAL TIME="A;"INCREMENT="M
720 PRINT
730 PRINT "TIME", "MASS", "MASS LOSS", "TOTAL MASS LOSS"
740 PRINT "----", "----", "---- ----", "--- - ---- ----"
750 GO TO 850
760 PRINT" WHAT IS THE HALF-LIFE, INITIAL NUMBER OF PARTICLES IN THE"
770 PRINT" SAMPLE, TOTAL ELAPSED TIME FOR DECAY, AND THE "
780 PRINT" INCREMENT OF ELAPSED TIME";
790GO TO 650
795 PRINT
800 PRINT"HALF-LIFE="B;"INITIAL NO. OF PARTICLES="F
810 PRINT"TOTAL TIME="A;"INCREMENT="M
820 PRINT
830 PRINT" TIME", " PARTICLES", "PART. LOSS", "TOTAL PART. LOSS"
840 PRINT" ----", " ----------", "----- ----", "----- ----- ----"
850 PRINT
860 FOR G = 0 TO A STEP M
870 LET n=F*EXP(-.6931*G/B)
880 LET W=ABS(n-Z)
890 LET Q=Q+W
900 IF F >1E6 THEN 920
910 IF U=1 THEN 940
920 PRINT G,n,W,Q
930 GO TO 950
940 PRINT INT(G+.5),INT(n+.5),INT(W+.5),INT(Q+.5)
950 LET Z=n
960 NEXT G
970 PRINT
980 PRINT
990 PRINT
1000 PRINT" DO YOU WANT THE ABOVE DATA GRAPHED? (1-YES, 0-NO)";
1010 INPUT K
1020 IF K=0 THEN 300
1023 IF K<>1 THEN 1000
1030 PRINT
1040 PRINT
1050 PRINT
1060 PRINT TAB(30);"MASS (OR PARTICLES) REMAINING"
1070 PRINT
1080 PRINT " ","0";TAB(62);F
1100PRINT"TIME","I---------I---------I---------I---------I---------I"
1120 FOR G = 0 TO K STEP M
1130 LET H=F*EXP(-.6931*G/B)
1140 LET H1=INT(H/F*50+.5)
1150 IF H1<=50 THEN 1170
1160 LET H1=50
1170 PRINT G,"I";TAB(H1+14.5);"*"
1250 NEXT G
1260 GO TO 300
1280 END
```

DISCIPLINE PHYSICS

TOPIC ELECTRIC FIELD STRENGTH

PROGRAM NAME EFILL0

DESCRIPTION:

The electric-field strength at a point near a fixed charge is calculated and printed. A line of charge is then generated by adding charges to either side of the fixed charge. As each additional charge is added, the new electric-field strength is calculated and selected values are printed.

Similarly, the field strength at a point near a plane of charge is calculated and printed as the plane is generated with the addition of other lines to the previous line of charge.

In both cases, the fields can be seen to approach a limiting value which is then printed for an infinite line and plane.

OBJECTIVES:

A. To show that the electric-field strength approaches limiting values for a line and a plane of charge.

B. To let the student discover how the field strength depends upon the distance from a point to a line of and to a plane of charge.

PRELIMINARY PREPARATION:

A. Student - A knowledge of Coulomb's law and the vector addition of electric fields.

B. Materials - none

DISCUSSION:

The operator chooses a distance (y) away from a fixed charge (Q_2) at which he wishes to know the field strength. He also chooses the number of charges (N), and their spacing (C), that he wishes to add to each side of the fixed charge to generate a line of charge. After the line has been generated, the operator enters the number of such lines (M) that he wishes to use in building up the plane of charge.

Actual values of force are not given, only relative values. When the fixed charge (Q_2) is at a distance Y= 1 from the test charge (Q_1), the force is 1 unit. The force may be calculated in Newtons if all distances are in meters, and the program

is slightly changed so that Q_1 and Q_2 are in coulombs. If both of these charges were to be taken as single elementary charges, then the following changes should be made:

```
280  LET Q1 = 1.6*E-19
290  LET Q2 = 1.6*E-19
300  LET K = 9*E9
```

If the spacing (C) is taken as .1 and the number of charges (N) as 1000, then three runs through the program using the distance between the test charge and the fixed charge (y) as 1, 2, and 4 should be sufficient for the relationships to be determined. A casual inspection of the exact values of the field strength for these three distances should yield the following conclusions:

1. The field strength varies inversely with the square of the distance away from a single point charge.
2. The field strength varies inversely with the distance from a line of charge.
3. The field strength remains constant even though the distance from a plane of charge changes.

It should be noted in 2 and 3 above, that the spacing between charges must be small as compared to the distance away from the line or plane of charge, and of course that the line be so long and the plane so broad that any further increase in length or breadth be insignificant.

An interesting bonus to this program is discovered when distances from test charge to plane is decreased to .001, .0001, and .00001. Here it can be seen that the field no longer is constant, but changes as an inverse square law for a single charge because the test charge begins to "see" the fixed charge instead of the whole plane. The "EXACT VALUE...," is calculated for charges smeared over the whole plane and not in discrete point charges as we have here; hence, the disagreement with actual field values.

This program may be run by an individual student after proper introductory explanation concerning vector addition of electric fields, contributions of the charges being added in the line or lines to the plane. It may also be used as a class demonstration and discussion. When used with a whole class it is best to have a television camera and monitor available for immediate display of print out. A summary table constructed either by the teacher on the board or by students at their desks is useful in analysis of the data.

```
THIS PROGRAM WILL CALCULATE THE FORCE ON A TEST CHARGE
THAT IS PLACED SOME DISTANCE, Y, AWAY FROM ANOTHER CHARGE;
A LINE OF CHARGE; AND A PLANE OF CHARGE.
YOU MUST ENTER THE DISTANCE AWAY, Y; THE SPACING DESIRED
BETWEEN CHARGES, C, AND ALSO BETWEEN LINES OF CHARGE THAT
MAKE UP THE PLANE OF CHARGE.
YOU MUST ALSO CHOOSE THE NUMBER OF CHARGES (N) IN THE LINE OF
CHARGE THAT YOU WOULD LIKE TO USE (500 IS A GOOD VALUE IF
YOU USE A SPACING OF .1 FOR C.
JUST SO THE CALCULATIONS DON'T GO TOO FAR I'VE INCLUDED
A STOP THAT DEPENDS UPON THE ANGLE FROM TEST CHARGE TO THE
LAST CHARGE TO BE CALCULATED.  IF THE ANGLE IS LESS
THAN 2 DEGREES, CALCULATIONS WILL CEASE.

INPUT Y,C,N? 1,.1,1000


NO. OF CHGS.
ON EACH SIDE  FORCE
------------  -----
 0             1
 1             2.97
 2             4.86
 3             6.61
 4             8.21
 5             9.65
 6             10.91
 7             12.01
 8             12.9
 9             13.78
 10            14.49
 20            17.98
 30            19
 40            19.42
 50            19.62
 60            19.73
 70            19.8
 80            19.85
 90            19.88
 100           19.9
 200           19.98
 287           19.99
EXCESSIVE COMPUTER TIME WOULD BE REQUIRED TO CALCULATE
THE FORCE FOR ADDITIONAL CHARGES.


THE EXACT VALUE FOR AN INFINITELY LONG LINE OF CHARGE IS 20
```

```
NOW ADD ROWS ON EITHER SIDE OF THE LINE OF CHARGE JUST
CALCULATED.  THE SPACING BETWEEN ROWS WILL BE THE SAME AS
THE SPACING BETWEEN THE CHARGES.
ENTER THE NUMBER OF EQUALLY SPACED ROWS YOU WANT ON EACH SIDE
? 500


NO. OF LINES
ON EACH SIDE  FORCE
------------  -----
 0             20
 1             59.58
 2             96.02
 3             134.69
 4             169.16
 5             201.14
 6             230.53
 7             257.36
 8             281.73
 9             303.82
 10            323.81
 20            446.55
 30            501.31
 40            531.18
 50            549.8
 60            562.47
 70            571.62
 80            578.54
 90            583.96
 100           588.3
 200           608.03
 287           614.05
EXCESSIVE COMPUTER TIME WOULD BE REQUIRED TO CALCULATE
THE FORCE FOR ADDITIONAL LINES OF CHARGE.


THE EXACT VALUE FOR AN INFINITE PLANE OF CHARGE IS 628.318


DO YOU WANT ANOTHER RUN (1=YES, 0=NO) ! ? 0


READY
```

```
100 REM   JOHN HOSIE   -  NORTHPORT HIGH -- 3/14/69
105 REM  REVISED BY C.LOSIK  8-25-70
106 REM  SEE BELOW FOR IMPORTANT VARIABLES
107 REM  'F' TYPE VARIABLES ARE FORCES
110 REM I'VE CHOSEN THE ELECTROSTATIC CONSTANT TO BE ONE AND
120 REM ALL CHARGES TO BE ONE SO THAT THE FORCE CALCULATED
130 REM IS JUST A RELATIVE FORCE.  IF YOU WOULD LIKE YOU MAY CHANGE
140 REM THINGS WHEN YOU RUN IT TO GET EXACT FORCES IN NEWTONS AND
150 REM USE CHARGES IN MICROCCULOMBS OR WHAT EVER ELSE YOU WISH.
160 PRINT "THIS PROGRAM WILL CALCULATE THE FORCE ON A TEST CHARGE"
170 PRINT "THAT IS PLACED SOME DISTANCE, Y, AWAY FROM ANOTHER CHARGE;"
180 PRINT "A LINE OF CHARGE; AND A PLANE OF CHARGE."
190 PRINT "YOU MUST ENTER THE DISTANCE AWAY, Y; THE SPACING DESIRED"
200 PRINT "BETWEEN CHARGES, C, AND ALSO BETWEEN LINES OF CHARGE THAT"
210 PRINT "MAKE UP THE PLANE OF CHARGE."
220 PRINT "YOU MUST ALSO CHOOSE THE NUMBER OF CHARGES (N) IN THE LINE OF
230 PRINT "CHARGE THAT YOU WOULD LIKE TO USE (500 IS A GOOD VALUE IF"
240 PRINT "YOU USE A SPACING OF .1 FOR C."
250 PRINT "JUST SO THE CALCULATIONS DON'T GO TOO FAR I'VE INCLUDED"
260 PRINT "A STOP THAT DEPENDS UPON THE ANGLE FROM TEST CHARGE TO THE"
270 PRINT "LAST CHARGE TO BE CALCULATED.  IF THE ANGLE IS LESS"
275 REM  Q1 AND Q2 ARE THE CHARGES
280 LET Q1=1
290 LET Q2=1
295 REM  K IS THE ELECTROSTATIC CONSTANT
300 LET K=1
305 REM  A IS THE CUTOFF ANGLE. THIS MAY BE CHANGED TO YOUR PREFERENCE
310 LET A=2
315 PRINT "THAN"A"DEGREES, CALCULATIONS WILL CEASE."
320 LET S=SIN(3.14159*A/180)
330 PRINT
340 PRINT "INPUT Y,C,N";
350 LET F1=0
353 INPUT Y,C,N
356 IF C<=0 THEN 365
358 IF Y<=0 THEN 365
360 IF N>=0 THEN 370
365 PRINT "ONE OF YOUR VALUES IS UNREASONABLE."
367 GO TO 330
370 PRINT
380 PRINT
390 PRINT "NO. OF CHGS."
400 PRINT "ON EACH SIDE","FORCE"
410 PRINT "------------","-----"
420 FOR I=0 TO N
430 LET X=I*C
440 LET R=SQR(X*X+Y*Y)
450 LET F=K*Q1*Q2/(R*R)
```

```
460 IF I<>0 THEN 490
470 LET F1=F
480 GO TO 510
490 LET F1=F1+2*F*(Y/
510 IF I<=10 THEN 580
520 IF I=1000*INT(I/1000) THEN 580
530 IF I>1000 THEN 600
540 IF I=100*INT(I/100) THEN 580
550 IF I>100 THEN 600
560 IF I=10*INT (I/10) THEN 580
570 GO TO 600
580 PRINT I,INT(100*F1+.5)/100
590 IF N=0 THEN 330
600 IF Y/R<S THEN 612
610 NEXT I
611 GO TO 620
612 PRINT I,INT(100*F1+.5)/100
613 PRINT "EXCESSIVE COMPUTER TIME WOULD BE REQUIRED TO CALCULATE"
614 PRINT "THE FORCE FOR ADDITIONAL CHARGES."
620 PRINT
625 PRINT
630 PRINT "THE EXACT VALUE FOR AN INFINITELY LONG LINE OF CHARGE IS";
640 PRINT 2*(K*Q1/C)/Y
650 PRINT
660 PRINT
670 PRINT
680 PRINT "NOW ADD ROWS ON EITHER SIDE OF THE LINE OF CHARGE JUST"
683 PRINT "CALCULATED.  THE SPACING BETWEEN ROWS WILL BE THE SAME AS"
686 PRINT "THE SPACING BETWEEN THE CHARGES."
690PRINT"ENTER THE NUMBER OF EQUALLY SPACED ROWS YOU WANT ON EACH SIDE"
700 INPUT M
702 IF M>=0 THEN 710
704 PRINT "NO NEGATIVE VALUES, PLEASE."
706 GO TO 690
710 PRINT
720 PRINT
730 PRINT "NO. OF LINES"
740 PRINT "ON EACH SIDE","FORCE"
750 PRINT "------------","------"
760 FOR P=0 TO M
770 LET Z=P*C
780 LET R1=SQR(Z*Z+Y*Y)
790 IF P<>0 THEN 850
830 LET F3=2*(K*Q1/C)/Y
840 GO TO 870
850 LET F3=F3+2*F1*(Y↑2)/(R1↑2)
870 IF P=1000*INT(P/1000) THEN 930
880 IF P>1000 THEN 940
890 IF P=100*INT(P/100) THEN 930
900 IF P>100 THEN 940
910 IF P=10*INT(P/10) THEN 930
920 IF P>10 THEN 940
930 PRINT P,INT(100*F3+.5)/100
```

```
940 IF Y/R1<S THEN 952
950 NEXT P
951 GO TO 960
952 PRINT P,INT(100*F3+.5)/100
953 PRINT "EXCESSIVE COMPUTER TIME WOULD BE REQUIRED TO CALCULATE"
954 PRINT "THE FORCE FOR ADDITIONAL LINES OF CHARGE."
960 PRINT
965 PRINT
970 PRINT "THE EXACT VALUE FOR AN INFINITE PLANE OF CHARGE IS";
980 PRINT 2*3.14159*(K*Q1)/(C*C)
990 PRINT
1000 PRINT
1010 PRINT "DO YOU WANT ANOTHER RUN (1=YES, 0=NO) : ";
1020 INPUT C
1030 IF C>0 THEN 330
1050 END
```

DISCIPLINE PHYSICS

SUBJECT KINEMATICS REVIEW

PROGRAM NAME KINERV

DESCRIPTION:

Questions are asked concerning the motion of a ball thrown vertically upwards at various velocities. Neglecting air resistance, the student is to determine such quantities as 1) maximum obtainable height; 2) time of flight; and 3) the height reached at different times.

OBJECTIVES:

To develop and review basic skills in solving projectile motion problems.

PRELIMINARY PREPARATION:

A. Student - previous classroom instruction and a working knowledge of algebra.

B. Materials - none

DISCUSSION:

The student is presented with various problems concerning the motion of the ball. In each case, the initial velocity V_0 of the ball is given. There are five basic questions asked:

1. Determine maximum height reached;
2. Find the height after t seconds;
3. Find the velocity when the ball is at height h;
4. Determine the time of flight; and
5. Find the velocity after t seconds.

The quantities V_0, h, and t are randomly determined for each question asked and the correct answers are given following the student response.

The program is designed to serve as a review of typical motion problems discussed in class and to aid in overcoming student "uncertainty" in the solution of numerical problems.

The program may be modified to cover other areas of review by entering new questions in place of those presently offered (see listing).

```
                    ---REVIEW OF KINEMATICS---

     A BALL IS THROWN STRAIGHT UP AT VARIOUS VELOCITIES.
AIR FRICTION IS NEGLIGIBLE. THE UPWARD DIRECTION IS TAKEN
AS POSITIVE, AND THE DOWNWARD DIRECTION AS NEGATIVE.

THE LOCAL ACCELERATION DUE TO GRAVITY IS -10 METERS/SECOND/SEC.

ALL VALUES ARE IN M.K.S. METRIC UNITS.

     FOR VARIOUS THROWING SPEEDS, YOU MUST ANSWER CERTAIN
QUESTIONS ABOUT THE BALL IN FLIGHT.


 1 . THE UPWARD THROWING SPEED IS  25  METERS/SECOND.
WHAT IS THE VELOCITY WHEN IT REACHES A HEIGHT OF  18.45
METERS ABOVE THE GROUND ? 16
YOU'RE CORRECT WITHIN 5 PERCENT. THE CORRECT ANSWER IS  16  .


 2 . THE UPWARD THROWING SPEED IS  16  METERS/SECOND.
WHAT IS THE VELOCITY AFTER  2  SECONDS OF FLIGHT? 14
YOU'RE OFF MORE THAN 5 PERCENT. THE CORRECT ANSWER IS -4  .


 3 . THE UPWARD THROWING SPEED IS  35  METERS/SECOND.
HOW HIGH ABOVE THE GROUND WILL THE BALL GO? 70
YOU'RE OFF MORE THAN 5 PERCENT. THE CORRECT ANSWER IS  61.25  .


 4 . THE UPWARD THROWING SPEED IS  29  METERS/SECOND.
HOW LONG WILL IT TAKE THE BALL TO RETURN TO THE GROUND? 6.0
YOU'RE CORRECT WITHIN 5 PERCENT. THE CORRECT ANSWER IS  5.8  .


 5 . THE UPWARD THROWING SPEED IS  21  METERS/SECOND.
WHAT IS THE VELOCITY WHEN IT REACHES A HEIGHT OF  10.35
METERS ABOVE THE GROUND ? 12
YOU'RE OFF MORE THAN 5 PERCENT. THE CORRECT ANSWER IS  15.29706  .


OUT OF  5  QUESTIONS, YOU GOT  2  RIGHT.
DON'T YOU KNOW ANYTHING ABOUT THROWING THINGS UP???

WANT TO TRY ANOTHER 5 PROBLEMS (1=YES, 0=NO) : ? 0


READY
```

```
100 REM RICHARD F. PAVJ PATCHOGUE H.S.J 1-24-69J PHYSICS
110REM THIS PROGRAM IS DESIGNED TO SERVE AS A REVIEW TEST IN KINEMATICS.
120 REM  REVISED BY C.LOSIK  8-25-70
130 REM  V IS VERTICAL VELOCITY, ALL ELSE IS 'A' (FOR COMPARISONS)
140 REM  WE GENERATE A V RANDOMLY AND RANDOMLY PICK A QUESTION
145 RANDOMIZE
150 LET P=0
155 LET R=0
160 PRINT"                    ---REVIEW OF KINEMATICS---"
170 PRINT
180 PRINT"     A BALL IS THROWN STRAIGHT UP AT VARIOUS VELOCITIES."
190 PRINT"AIR FRICTION IS NEGLIGIBLE. THE UPWARD DIRECTION IS TAKEN"
200 PRINT"AS POSITIVE, AND THE DOWNWARD DIRECTION AS NEGATIVE."
210 PRINT
220 PRINT"THE LOCAL ACCELERATION DUE TO GRAVITY IS -10 METERS/SECOND/SEC
230 PRINT
240 PRINT"ALL VALUES ARE IN M.K.S. METRIC UNITS."
250 PRINT
260 PRINT"     FOR VARIOUS THROWING SPEEDS, YOU MUST ANSWER CERTAIN"
270 PRINT"QUESTIONS ABOUT THE BALL IN FLIGHT."
280PRINT
290PRINT
300 LET U=RND(X)
330 IF Q=0THEN 350
340 IF Q/5=INT(Q/5) THEN 770
350LET V=5+INT(35*U)
360LETZ=1+INT(4.999*U)
370 IF(Z-P)*(V-R)=0 THEN 300
380 LETP=Z
390 LET Q=Q+1
400 LET R=V
410 PRINT
420 PRINT Q". THE UPWARD THROWING SPEED IS "V" METERS/SECOND."
430 IFZ=1 THEN 540
440 IF Z=2 THEN 590
450 IF Z=3 THEN 630
460 IF Z=4 THEN 500
470 LET A=.05*V*V
480 PRINT"HOW HIGH ABOVE THE GROUND WILL THE BALL GO"J
490 GOTO670
500 LET A=V/5
510 PRINT"HOW LONG WILL IT TAKE THE BALL TO RETURN TO THE GROUND"J
520 GOTO670
540 LET T=1+ INT(2*V*U)/10
550LET A=V*T-5*T*T
560 PRINT"HOW HIGH ABOVE THE GROUND WILL THE BALL BE AFTER "JT
570 PRINT"SECONDS OF FLIGHT"J
580 GOTO670
590 LET T=1+INT(2*V*U)/10
600 LET A=V-10*T
610 PRINT"WHAT IS THE VELOCITY AFTER "T" SECONDS OF FLIGHT"J
620 GO TO 670
630 LET S=.5*INT(V*V*U)/10
640 LET A=SQR(V*V-20*S)
650 PRINT"WHAT IS THE VELOCITY WHEN IT REACHES A HEIGHT OF "JS
660 PRINT"METERS ABOVE THE GROUND "J
670 INPUT G
680 PRINT"YOU'RE"J
690 IF ABS((G-A)/A)>.05 THEN 730
700 LET C=C+1
710 PRINT " CORRECT WITHIN "J
720 GOTO740
730 PRINT" OFF MORE THAN "J
740 PRINT"5 PERCENT. THE CORRECT ANSWER IS "A" ."
750PRINT
760GOTO300
770 PRINT
780 PRINT"OUT OF "Q" QUESTIONS, YOU GOT "C" RIGHT."
790 IFC/Q>=.7 THEN810
800 PRINT"DON'T YOU KNOW ANYTHING ABOUT THROWING THINGS UP???"
810 PRINT
820 PRINT "WANT TO TRY ANOTHER 5 PROBLEMS (1=YES, 0=NO) ? "J
830 INPUT M
835 LET U=RND(X)
840 IF M=1 THEN 350
850 IF M<>0 THEN 810
860 END
```

DISCIPLINE PHYSICS

SUBJECT LENSES

PROGRAM NAME LENSES

DESCRIPTION:

The focal length, object distance, image distance, image size, or object size, may be calculated if sufficient information is entered by the student.

OBJECTIVES:

A. To solve for focal length of a lens from laboratory data.

B. To check image position and size from lab data.

C. To solve lens problems.

PRELIMINARY PREPARATION:

A. Student - Data from a lens experiment.

B. Materials - none

DISCUSSION:

If this program is used in conjunction with a lens laboratory, the student may check his calculations of focal length.

He may also check his image size and position from known object size and position.

```
THIS PROGRAM MAY BE USED TO SOLVE LENS PROBLEMS.
IN THE ORDER GIVEN ENTER THE VALUES FOR THE FOLLOWING:
FOCAL LENGTH, OBJECT DISTANCE, IMAGE DISTANCE, OBJECT
SIZE, IMAGE SIZE.  INPUT 0 (ZERO) FOR UNKNOWN VALUES.
EVERY TIME THE COMPUTER ASKS 'READY?', ENTER 1 IF YOU HAVE
MORE PROBLEMS TO DO, OR 0 TO END THE PROGRAM.

*** READY ? 1
WHAT ARE YOUR VALUES FOR F, P, Q, O, I? 1,2,3,4,5

YOUR Q IS NOT CORRECT FOR THE F AND P YOU HAVE ENTERED
NOTE CHANGED Q.  THE PERCENT ERROR ON YOUR Q IS:
 50  PERCENT.
YOUR I IS NOT CORRECT FOR THE O YOU HAVE ENTERED
NOTE CHANGED I.  THE PERCENT ERROR ON YOUR I IS:
 25  PERCENT.

F= 1          P= 2          Q= 2          O= 4          I= 4


*** READY ? 1
WHAT ARE YOUR VALUES FOR F, P, Q, O, I? 1,2,2.04,4,3.9

YOUR Q IS CORRECT TO WITHIN 2 %
NOTE CORRECTED Q.
YOUR I IS CORRECT TO WITHIN 4.411765 %
NOTE CORRECTED I.

F= 1          P= 2          Q= 2.04       O= 4          I= 4.08


*** READY ? 1
WHAT ARE YOUR VALUES FOR F, P, Q, O, I? 0,5,8,4,4
```

```
YOUR I IS NOT CORRECT FOR THE O YOU HAVE ENTERED
NOTE CHANGED I.  THE PERCENT ERROR ON YOUR I IS:
 37.5  PERCENT.

F= 3.076923   P= 5          Q= 8          O= 4          I= 6.4


*** READY ? 1

WHAT ARE YOUR VALUES FOR F, P, Q, O, I? 46,53,0,34,32

YOUR I IS NOT CORRECT FOR THE O YOU HAVE ENTERED
NOTE CHANGED I.  THE PERCENT ERROR ON YOUR I IS:
 85.67775  PERCENT.

F= 46         P= 53         Q= 348.2857   O= 34         I= 223.4286


*** READY ? 1

WHAT ARE YOUR VALUES FOR F, P, Q, O, I? 4857,2,6,0,0

YOUR Q IS NOT CORRECT FOR THE F AND P YOU HAVE ENTERED
NOTE CHANGED Q.  THE PERCENT ERROR ON YOUR Q IS:
 399.8765  PERCENT.

F= 4857       P= 2          Q=-2.000824
```

```
100 REM JOHN W. HOSIE - NORTHPORT HIGH - PHYSICS
101 REM REVISED 11-18-70, L.BRAUN
110 PRINT "THIS PROGRAM MAY BE USED TO SOLVE LENS PROBLEMS."
120PRINT
130PRINT"IN THE ORDER GIVEN ENTER THE VALUES FOR THE FOLLOWING:"
140PRINT
150PRINT"FOCAL LENGTH, OBJECT DISTANCE, IMAGE DISTANCE, OBJECT"
160PRINT
170PRINT"SIZE, IMAGE SIZE.  INPUT 0 (ZERO) FOR UNKNOWN VALUES."
180PRINT
182 PRINT "EVERY TIME THE COMPUTER ASKS 'READY?', ENTER 1 IF YOU HAVE"
183 PRINT
184 PRINT "MORE PROBLEMS TO DO, OR 0 TO END THE PROGRAM."
186 PRINT
188 PRINT
190 PRINT "*** READY ";
191 INPUT F
192 IF F=0 THEN 780
193 IF F<>1 THEN 188
195PRINT
196 PRINT "WHAT ARE YOUR VALUES FOR F, P, Q, O, I";
200 INPUT F,P,Q,O,I
210 LET P=ABS(P)
220 LET O=ABS (O)
230 PRINT
240 IF F=0 THEN 390
250 IF P<>F THEN 280
260 PRINT "THE IMAGE IS AT INFINITY"
270 GO TO 186
280 IF P =0 THEN 360
290 LET Z=P*F/(P-F)
300 IF Q=0 THEN 340
310 IF Z=Q THEN 420
312 IF ABS(Q-Z)<0.05*Z THEN 344
320 PRINT "YOUR Q IS NOT CORRECT FOR THE F AND P YOU HAVE ENTERED"
330 PRINT "NOTE CHANGED Q.  THE PERCENT ERROR ON YOUR Q IS:"
335 PRINT 100*ABS(Z-Q)/ABS(Z); " PERCENT."
340 LET Q=Z
342 GO TO 420
344 PRINT"YOUR Q IS CORRECT TO WITHIN";100*ABS(Q-Z)/ABS(Z)"%"
346 PRINT"NOTE CORRECTED Q."
350 GOTO 420
360 IF Q=0 THEN 560
370 LET P=Q*F/(Q-F)
380 GO TO 420
390 IF P=0 THEN 550
400 IF Q=0 THEN 490
410 LET F=Q*P/(Q+P)
420 IF O=0 THEN 730
430 IF I=0 THEN 470
```

```
435 LET Z9=O*Q/P
440 IF I=Z9 THEN 660
445 IF ABS(I-Z9)<0.05*Z9 THEN 474
450 PRINT "YOUR I IS NOT CORRECT FOR THE O YOU HAVE ENTERED"
460 PRINT "NOTE CHANGED I.  THE PERCENT ERROR ON YOUR I IS:"
465 PRINT 100*ABS(I-Z9)/ABS(Z9); " PERCENT."
470 LET I=Z9
472 GO TO 660
474 PRINT"YOUR I IS CORRECT TO WITHIN";100*ABS(I-Z9)/ABS(Z9)"%"
476 PRINT"NOTE CORRECTED I."
478 LET I=Z9
480 GO TO 660
490 IF O<>0 THEN 520
500 PRINT "IF F,Q, AND O OR I = 0; CANNOT CALCULATE - TRY AGAIN."
510 GO TO 186
520 IF I=0 THEN 500
530 LET Q=P*I/O
540 GO TO 650
550 IF Q<>0 THEN 590
560 PRINT "THE FOCAL LENGTH CANNOT BE CALCULATED IF BOTH OBJECT"
570 PRINT "AND IMAGE DISTANCES ARE ZERO."
580 GO TO 186
590 IF O=0 THEN 610
600 IF I<>0 THEN 640
610 PRINT "MUST KNOW BOTH OBJECT AND IMAGE SIZE TO FIND FOCAL"
620 PRINT "LENGTH AND OBJECT DISTANCE."
630 GO TO 186
640 LET P=Q*O/I
650 LET F=Q*P/(P+Q)
660 PRINT
665 PRINT "F="F, "P="P, "Q="Q, "O="O, "I="I
670 PRINT
680 PRINT
690 GO TO 186
700 PRINT "INFORMATION ABOUT EITHER FOCAL LENGTH OR BOTH OBJECT"
710 PRINT "AND IMAGE SIZE NEEDED FOR COMPLETE SOLUTION."
720 GO TO 186
730 IF I=0 THEN 760
740 LET O=I*P/Q
750 GOTO 660
760 PRINT
765 PRINT "F="F, "P="P, "Q="Q
770 GO TO 670
780 END
```

DISCIPLINE PHYSICS

SUBJECT MASS DEFECT

PROGRAM NAME MASSD

DESCRIPTION:

A classroom presentation that could be used to calculate mass defect, and give the answer in terms of usable energy (kw-hr. of electricity).

OBJECTIVES:

A. To calculate and explain mass defect.

B. To introduce the concept of binding energy.

C. Conversion of mass to energy. (atomic power)

PRELIMINARY PREPARATION:

A. Student - The student should have an understanding of nuclear particles, and the law of conservation of mass and energy.

B. Materials - The teacher should make available a table of isotopes that lists the actual mass. (Handbook of Chemistry and Physics, Chemical Rubber Company)

DISCUSSION:

It should be noted that the masses used here include the electrons. The very small difference which would be obtained if the bare nuclear mass were known is negligible for the purpose of this calculation.

Time permitting, it would be beneficial to have the student investigate the conversion of atomic mass units(AMU) to calories and kilowatt-hours in order to recognize the significance of the units and the magnitude of the numbers involved.

```
   THIS PROGRAM IS DESIGNED TO INVESTIGATE MASS DEFECT

   WHICH OF THE ELEMENTS WOULD YOU LIKE TO CONSIDER?
 REMEMBER WE ARE DEALING WITH A SINGLE ATOM, THEREFORE
 IN ADDITION TO THE ATOMIC NUMBER WE ARE GOING TO NEED THE
 ACTUAL MASS (IN AMU) AND THE MASS NUMBER OF THE ISOTOPE
 YOU WANT TO WORK WITH.

 WHEN THE MACHINE TYPES A QUESTION MARK (?) TYPE IN
 YOUR ANSWER THEN HIT RETURN KEY. USE NUMBERS OF UP TO
 SIX SIGNIFICANT FIGURES. ROUND IF NECESSARY TO 6 DIGITS.
IN THE VALUES FOR MASS DEFECT.


 THE ATOMIC NUMBER IS ? 8
 THE ACTUAL MASS IS ? 15.9949
 THE MASS NUMBER IS ? 16

 THE SUM OF THE MASS OF THE 8 PROTONS AND THE 8 NEUTRONS
 PLUS THE WEIGHT OF THE 8 ELECTRONS IS THE CALCULATED
 MASS.

  CALCULATED MASS  -  ACTUAL MASS  =  MASS DEFECT
      16.13199     -    15.9949     =     .1371

 THE MASS DEFECT IN TERMS OF ENERGY IS THE EQUIVALENT OF
 2936 X 10^9 CAL. PER MOLE OF THIS SUBSTANCE,
OR 184 X 10^9 CAL. PER GRAM.

 IF WE DIVIDE THIS BINDING ENERGY BY THE NUMBER OF
 PARTICLES IN THE NUCLEUS, WE GET A RATIO KNOWN AS THE
 BINDING ENERGY PER NUCLEON, WHICH IS A MEASURE OF THE
 STABILITY OF THE NUCLEUS.  THE MORE 'BINDING'
 PER NUCLEON, THE MORE STABLE IS THE NUCLEUS.
 THE BINDING ENERGY PER NUCLEON IS : 1.276744E-5 ERGS. PER NUCLEON, OR
 3.047121E-13 CAL. PER NUC.,
 WHICH IS MORE COMMONLY EXPRESSED AS 800 MEV.

 THE AMOUNT OF ENERGY (BINDING ENERGY) CONTAINED IN ONE
 GRAM OF THIS SUBSTANCE WOULD BE SUFFICIENT TO SUPPLY ALL
 THE ELECTRICAL NEEDS IN AN AVERAGE ONE FAMILY HOUSE USING
 15 KW-HRS. PER DAY FOR A PERIOD OF 14245 DAYS OR
 39 YEARS.

 IF YOU WOULD LIKE TO RUN ANOTHER PROBLEM TYPE IN 1,
 IF NOT TYPE IN 0.
? 0

               ***************

READY
```

```
100 REM JOHN MARCHISOTTO  PIB SUMMER 69   BASIC
105 REM  REVISED BY C.LOSIK  7-22-70
106 REM  AT NO=A, MASS=B, MASS NO=C
107 REM  MASS DEFECT IS F
130 PRINT"   THIS PROGRAM IS DESIGNED TO INVESTIGATE MASS DEFECT"
140 PRINT
150 PRINT"   WHICH OF THE ELEMENTS WOULD YOU LIKE TO CONSIDER? "
160 PRINT" REMEMBER WE ARE DEALING WITH A SINGLE ATOM, THEREFORE"
170 PRINT" IN ADDITION TO THE ATOMIC NUMBER WE ARE GOING TO NEED THE"
180 PRINT" ACTUAL MASS (IN AMU) AND THE MASS NUMBER OF THE ISOTOPE"
190 PRINT" YOU WANT TO WORK WITH."
200 PRINT
210 PRINT " WHEN THE MACHINE TYPES A QUESTION MARK (?) TYPE IN"
220 PRINT " YOUR ANSWER THEN HIT RETURN KEY. USE NUMBERS OF UP TO"
230 PRINT " SIX SIGNIFICANT FIGURES. ROUND IF NECESSARY TO 6 DIGITS."
237 PRINT "IN THE VALUES FOR MASS DEFECT."
238 PRINT
240 PRINT
250 PRINT " THE ATOMIC NUMBER IS ";
260 INPUT A
270 PRINT " THE ACTUAL MASS IS ";
280 INPUT C
290 PRINT " THE MASS NUMBER IS ";
300 INPUT B
310 PRINT
320 REM  G IS AVOGADRO'S NUMBER
330 LET G=6.023E23
340 LET D = B - A
350 LET E=(1.00728*A)+(1.00867*D)+(5.48597E-4*A)
360 LET F=INT(1E4*(E-C)+.5)/1E4
370 PRINT " THE SUM OF THE MASS OF THE"A"PROTONS AND THE"D"NEUTRONS"
380 PRINT " PLUS THE WEIGHT OF THE"A"ELECTRONS IS THE CALCULATED"
390 PRINT " MASS."
400 PRINT
410 PRINT"  CALCULATED MASS  -  ACTUAL MASS  =  MASS DEFECT"
420 PRINT"      "E,"     -  "C;"     =    "F
430 PRINT
440 REM   CONVERSION FACTORS:
450 REM         1.49 X 10-3 ERGS PER AMU
460 REM         4.19 X 10 7 ERGS PER CAL.
470 REM         3.6  X 10 13 ERGS PER KW-H
475 REM          931.0 MEV PER AMU
480 LET H=(1.49E-3*F*G)/4.19E7
490 PRINT " THE MASS DEFECT IN TERMS OF ENERGY IS THE EQUIVALENT OF"
500 PRINT INT(H/1E9+.5)"X 10↑9 CAL. PER MOLE OF THIS SUBSTANCE,"
510 PRINT "OR"INT((H/C)/1E9+.5)"X 10↑9 CAL. PER GRAM."
```

```
511 PRINT
512 PRINT " IF WE DIVIDE THIS BINDING ENERGY BY THE NUMBER OF"
513 PRINT " PARTICLES IN THE NUCLEUS, WE GET A RATIO KNOWN AS THE"
514 PRINT " BINDING ENERGY PER NUCLEON, WHICH IS A MEASURE OF THE"
515 PRINT " STABILITY OF THE NUCLEUS.  THE MORE 'BINDING'"
516 PRINT " PER NUCLEON, THE MORE STABLE IS THE NUCLEUS."
517 PRINT " THE BINDING ENERGY PER NUCLEON IS :";
518 PRINT 1.49E-3*F/B"ERGS. PER NUCLEON, OR";
519 PRINT 1.49E-3*F/(B*4.19E7)"CAL. PER NUC.,"
520PRINT" WHICH IS MORE COMMONLY EXPRESSED AS"100*INT(931*F/B+.5)"MEV."
522 LET J = ((A/C)*4.19E7/3.6E13)/15
525 PRINT
530 PRINT " THE AMOUNT OF ENERGY (BINDING ENERGY) CONTAINED IN ONE"
540 PRINT " GRAM OF THIS SUBSTANCE WOULD BE SUFFICIENT TO SUPPLY ALL"
550 PRINT " THE ELECTRICAL NEEDS IN AN AVERAGE ONE FAMILY HOUSE USING"
560 PRINT " 15 KW-HRS. PER DAY FOR A PERIOD OF"INT(J+.5)"DAYS OR"
565 PRINT INT((J/365)+.5)"YEARS."
570 PRINT
580 PRINT " IF YOU WOULD LIKE TO RUN ANOTHER PROBLEM TYPE IN 1,"
590 PRINT " IF NOT TYPE IN 0."
600 INPUT M
610 PRINT
620 PRINT " ","***************"
630 IF M=1 THEN 240
640 IF M<>0 THEN 580
650 END
```

DISCIPLINE PHYSICS

SUBJECT FORCES + DISPLACEMENTS

PROGRAM NAME NEWTN2

DESCRIPTION:

A problematic situation is presented to the student which requires repeated applications of Newton's 2nd law. By selecting various angles and forces, the operator can observe the resulting motion produced. To successfully complete the program, the student must complete a specified displacement within ten attempts.

OBJECTIVES:

To aid in the development of skills in applying the equations of motion.

PRELIMINARY PREPARATION:

A. Student - An awareness of Newton's 2nd law is required. In addition, some familiarity of force components (resolution of vectors) is necessary.

B. Materials - Graph paper is helpful to students with below-to-average ability.

DISCUSSION:

A. Operational Suggestions

This program was designed for operation by individual students or small groups, but the program may be used with a class as a "lead-in" demonstration of forces and vectors. The presentation is also helpful in describing two dimensional motion under the influence of a constant external force.

When executed by small groups of average students, it has been noted that programs of this type stimulate discussions and involvement for those participating.

B. Suggested Follow-up

The student is confronted with a situation which requires that he overcome a given force (the wind), in moving a boat across a channel 10 Km. wide. The magnitude of the force produced by the wind on the boat varies with each "run", but the direction of the vector is always southwest, i.e. 45 deg. with respect to the direction EAST. The student may vary his paddling force (limited to values less than 200 Newtons), and direction at intervals during his displacement. After each choice of variables, he is given his position, as well as the resulting speed and direction of the boat. A certificate is presented for successful completion of the task.

```
HINT: GRAPH PAPER IS HELPFUL IN RUNNING THIS PROGRAM.

                F' MA  SPEED....
                ----------------

YOU'RE TRYING TO ESCAPE FROM DEVIL'S ISLAND ON A SMALL BOAT.
DEVIL'S ISLAND IS LOCATED AT COORDINATES (0,0).
TO SUCCEED, YOU MUST REACH A CHANNEL 50 METERS WIDE AND
10000 METERS DUE EAST, AT ABOUT (10000,0).

IN ADDITION, YOU MUST GET THERE IN FIVE MINUTES OR LESS OR
SUFFER RECAPTURE ---  (HEH,HEH,HEH--)

WHAT DO YOU WEIGH (IN POUNDS)? 170

YOUR SITUATION IS AS FOLLOWS:

THE WIND IS BLOWING FROM THE NORTHEAST (45 DEGREES) EXERTING
A FORCE OF 100 NEWTONS ON YOUR BOAT. YOU MAY PADDLE WITH
ANY FORCE IN THE EASTWARD DIRECTION (ZERO DEGREES IS EAST)
TO ACCELERATE YOUR BOAT ACROSS THE BAY AND THUS
REACH THE OPPOSITE SHORE (AND FREEDOM).
(NOTE: THE MASS OF THE BOAT WITH YOU ABOARD IS 177 KILOGRAMS).

WITH WHAT FORCE (IN NEWTONS) AND DIRECTION (IN DEGREES) WILL
YOU PADDLE? 199,25
T= .5          X= 276         Y= 32          V(X)= 18       V(Y)= 8

WANT TO CHANGE FORCE OR DIRECTION (1=YES, 0=NO) ? ? 1

WITH WHAT FORCE (IN NEWTONS) AND DIRECTION (IN DEGREES) WILL
YOU PADDLE? 199,23
T= 1           X= 1111        Y= 112         V(X)= 37       V(Y)= 3

WANT TO CHANGE FORCE OR DIRECTION (1=YES, 0=NO) ? ? 1

WITH WHAT FORCE (IN NEWTONS) AND DIRECTION (IN DEGREES) WILL
YOU PADDLE? 199,20
T= 1.5         X= 2583        Y= 199         V(X)= 57       V(Y)= 3

WANT TO CHANGE FORCE OR DIRECTION (1=YES, 0=NO) ? ? 0
T= 2           X= 4520        Y= 269         V(X)= 76       V(Y)= 8

WANT TO CHANGE FORCE OR DIRECTION (1=YES, 0=NO) ? ? 0
T= 2.5         X= 7103        Y= 321         V(X)= 96       V(Y)= 1

WANT TO CHANGE FORCE OR DIRECTION (1=YES, 0=NO) ? ? 1

WITH WHAT FORCE (IN NEWTONS) AND DIRECTION (IN DEGREES) WILL
YOU PADDLE? 150,10
T= 3           X= 10173       Y= 246         V(X)= 109      V(Y)=-6
YOU HAVE REACHED THE OPPOSITE SHORE,
BUT ARE 246 METERS OFF COURSE.
ALL THAT WORK FOR NOTHING!
YOU'RE LOST IN THE SWAMPS FOREVER!  GOODBYE.

SEE IF YOU CAN IMPROVE YOUR ABILITY LATER.


READY
```

```
100REM A.C. CAGGIANO; PATCHOGUE H.S.; DEC.4,68
105 REM  REVISED BY C.LOSIK  8-16-70
106 REM  X AND Y COORDINATES ARE USED WITH 1-TYPE VARIABLES IN
107 REM  X DIRECTION  AND 2-TYPE VARIABLES IN THE Y DIRECTION
108 REM  IT IS BEST TO CHECK THE EQUATIONS BELOW
110REM THIS IS A PHYSICS PROGRAM WHICH ATTEMPTS TO DEVELOP A
120REM     'FEEL' FO THE F=MA RELATIONSHIP
130 PRINT"HINT: GRAPH PAPER IS HELPFUL IN RUNNING THIS PROGRAM."
140LETX=0
150 LET Y=0
160LETV1=0
170LETV2=0
180 PRINT
190LETQ=0
200PRINT" ","  ";"F    MA   SPEED,..."
210PRINT" ","  ";"----------------"
220PRINT
230PRINT"YOU'RE TRYING TO ESCAPE FROM DEVIL'S ISLAND ON A SMALL BOAT."
235 PRINT "DEVIL'S ISLAND IS LOCATED AT COORDINATES (0,0)."
240PRINT"TO SUCCEED, YOU MUST REACH A CHANNEL 50 METERS WIDE AND"
250PRINT"10000 METERS DUE EAST, AT ABOUT (10000,0)."
260PRINT
270PRINT"IN ADDITION, YOU MUST GET THERE IN FIVE MINUTES OR LESS OR"
280PRINT"SUFFER RECAPTURE ---  (HEH,HEH,HEH--)"
290PRINT
300PRINT"WHAT DO YOU WEIGH (IN POUNDS)";
310INPUTW
320 RANDOMIZE
360LETP1=RND(X)*8
380LET P=20*INT(P1+.5)
390 LET M=INT(W/2.2+100.5)
400 LET T1=0
410 LET T=30
420PRINT
430PRINT"YOUR SITUATION IS AS FOLLOWS:"
440PRINT
450PRINT"THE WIND IS BLOWING FROM THE NORTHEAST (45 DEGREES) EXERTING"
460PRINT"A FORCE OF"P"NEWTONS ON YOUR BOAT. YOU MAY PADDLE WITH"
470 PRINT "ANY FORCE IN THE EASTWARD DIRECTION (ZERO DEGREES IS EAST)"
475 PRINT "TO ACCELERATE YOUR BOAT ACROSS THE BAY AND THUS"
477 PRINT "REACH THE OPPOSITE SHORE (AND FREEDOM)."
480PRINT"(NOTE: THE MASS OF THE BOAT WITH YOU ABOARD IS"M"KILOGRAMS)."
490PRINT
500PRINT"WITH WHAT FORCE (IN NEWTONS) AND DIRECTION (IN DEGREES) WILL"
510PRINT"YOU PADDLE";
520 INPUT F,A
530 IF ABS(F-100)<100 THEN 610
540LETQ=Q+1
550 IF Q>2 THEN 970
560IFQ>1THEN590
570PRINT"YOU MUST THINK YOU'RE SUPERMAN. BE PRACTICAL."
580GOTO490
590PRINT"IF YOU'RE THAT STRONG, JUMP THE CHANNEL---"
600GOTO490
610 IF ABS(A)<90 THEN 640
620PRINT"YOU'RE HEADING THE WRONG WAY. TRY AGAIN."
630GOTO490
640 LET A0=.0175*A
650 LET A1=(F*COS(A0)-.717*P)/M
660 LET A2=(F*SIN(A0)-.717*P)/M
670 LET X=A1*T*T/2+V1*T+X
680 LET Y=A2*T*T/2+V2*T+Y
```

```
690 LET V1=A1*T+V1
700 LET V2=A2*T+V2
710 LET T1=T1+.5
720 PRINT "T="T1,"X="INT(X+.5),"Y="INT(Y+.5),"V(X)="INT(V1+.5),
721 PRINT "V(Y)="INT(V2+.5)
722 IF X>0 THEN 730
724 PRINT "NO HELP THAT WAY. YOU'RE GOING BACKWARDS."
726 GO TO 490
730 IF X>10000 THEN 800
735 IF T1>5 THEN 900
740 PRINT
750 PRINT "WANT TO CHANGE FORCE OR DIRECTION (1=YES, 0=NO) ? "
760 INPUT R
770 IF R=1 THEN 490
775 IF R=0 THEN 650
780 GO TO 750
800 PRINT "YOU HAVE REACHED THE OPPOSITE SHORE."
810 IF ABS(Y)<200 THEN 830
820 PRINT "BUT ARE"INT(Y+.5)"METERS OFF COURSE."
825 PRINT "ALL THAT WORK FOR NOTHING!"
826 PRINT "YOU'RE LOST IN THE SWAMPS FOREVER!  GOODBYE."
827 GO TO 950
830 IF ABS(Y)<100 THEN 850
840 PRINT "AND YOU MIGHT MAKE IT, THOUGH YOU ARE OFF COURSE."
843 GO TO 950
850 IF ABS(Y)<25 THEN 870
860 PRINT "BUT YOU'RE CLOSE ENOUGH TO GET AWAY.  GOOD LUCK!"
863 GO TO 950
870 PRINT "AND HAVE REACHED THE CHANNEL."
880 PRINT "HOW SWEET SUCCESS IS !!!"
890 GO TO 950
900 PRINT "YOUR TIME IS UP."
910 IF X<=10000 THEN 800
920 PRINT "YOU HAVE NOT REACHED THE CHANNEL, AND ARE ONLY"
930 PRINT INT(SQR(X*X+Y*Y)+.5)"METERS FROM WHERE YOU STARTED."
940 PRINT "YOU MUST SUFFER RECAPTURE."
945 PRINT "SORRY, CHUM, BUT THAT'S PHYSICS."
950 PRINT
960 PRINT "SEE IF YOU CAN IMPROVE YOUR ABILITY LATER."
970 END
```

DISCIPLINE PHYSICS

SUBJECT PHOTOELECTRIC EFFECT

PROGRAM NAME PHOTEL

DESCRIPTION:

An experiment involving the photoelectric effect is simulated by the computer, to enable students to develop a qualitative understanding of the phenomenon.

OBJECTIVES:

To demonstrate a "critical wavelength" for photo-electronic emission.

PRELIMINARY PREPARATION:

A. Student

1. Prior discussion of the phenomenon as an introduction to modern physics
2. Students must be previously aware of such properties of light as wavelength and intensity.

B. Materials - none

DISCUSSION:

The student is permitted to select any one of five metals, which is subsequently subjected to ultraviolet radiation. The electrons are "counted" by an ammeter incorporated in the simulated experimental set-up.

The data collected is tabulated for three trials, indicating the current measured for various wavelengths. The data will indicate that:

1. The photoelectric emission is a function of wavelength;
2. For light of wavelength less than the critical value, the number of electrons emitted is dependent upon the incident light intensity; and
3. For wavelengths greater than the critical value, light intensity has no effect on the emission of electrons.

The program is designed for individual qualitative investigation of the phenomena, but may also be utilized by small groups.

It should be noted that this program is advantageous where limited or non-existent lab equipment hinders actual experimentation.

THE PHOTOELECTRIC EFFECT

WHEN LIGHT OF SHORT WAVELENGTH FALLS ON A METAL SURFACE, ELECTRONS ARE EJECTED FROM THE METAL. ACCORDING TO THE DESCRIPTION OF THIS PHENOMENON BY EINSTEIN, THERE IS A MAXIMUM WAVELENGTH FOR EACH METAL ABOVE WHICH NO ELECTRONS ARE EMITTED. IN THIS EXPERIMENT WE WILL DETERMINE THE CRITICAL WAVELENGTH AT WHICH THIS OCCURS.

THE METAL SELECTED WILL BE PLACED IN A VACUUM WHERE IT WILL BE BOMBARDED BY SOFT X-RAYS. THE NUMBER OF ELECTRONS EJECTED WILL BE COLLECTED AND COUNTED WITH AN AMMETER. (NOTE:THE CURRENT IS RELATED TO THE NUMBER OF ELECTRONS EMITTED BY THE METAL).

SELECT ONE OF THE METALS LISTED BY TYPING ITS NUMBER.

1) SILVER
2) BISMUTH
3) CADMIUM
4) LEAD
5) PLATINUM

? 3

	MEASURED CURRENT (MICROAMPERES)		
WAVELENGTH	TRIAL 1	TRIAL 2	TRIAL 3
2380	20.7	20.3	20.1
2500	20.6	20.3	20.4
2631	20.5	20.1	20.8
2777	20.8	20.5	20.2
2941	20.1	20.8	20.8
3125	20.2	20	20.8
3333	2.2	3.6	3
3571	4	2.8	4
3846	1	3.5	3.5

DO YOU WISH TO INCREASE THE LIGHT INTENSITY?
(1=YES, 0=NO) : ? 1

BY WHAT FACTOR? (SELECT FACTOR BETWEEN 1 AND 10).
? 7

	MEASURED CURRENT (MICROAMPERES)		
WAVELENGTH	TRIAL 1	TRIAL 2	TRIAL 3
2380	140	140	140
2500	140.1	140.1	140.1
2631	140	140	140.1
2777	140	140.1	140.1
2941	140.1	140	140.1
3125	140.1	140	140
3333	4.5	4.9	3.3
3571	1.4	1.7	2
3846	4	0	3.7

```
DO YOU WISH TO INCREASE THE LIGHT INTENSITY?
(1=YES, 0=NO) : ? 0

DO YOU WISH TO TRY ANOTHER METAL (1=YES, 0=NO) : ? 1
SELECT ONE OF THE METALS LISTED BY TYPING ITS NUMBER.

               1) SILVER
               2) BISMUTH
               3) CADMIUM
               4) LEAD
               5) PLATINUM

? 2

               MEASURED CURRENT (MICROAMPERES)
WAVELENGTH     TRIAL 1        TRIAL 2        TRIAL 3
 2380           10.6           10.7           11.1
 2500           10.4           11             10.5
 2631           11             11             11.4
 2777           11.4           10.6           10
 2941           11.1           11.6           10.9
 3125           3.3            4.8            1.7
 3333           4.2            3.6            2
 3571           3.9            4              4.2
 3846           1              1.7            3.3

DO YOU WISH TO INCREASE THE LIGHT INTENSITY?
(1=YES, 0=NO) : ? 0

DO YOU WISH TO TRY ANOTHER METAL (1=YES, 0=NO) : ? 0

NOW BY PLOTTING THE WAVELENGTH VS. THE MEASURED CURRENT,
(AVERAGE OF THREE TRIALS), THE PHOTOELECTRIC EFFECT AS
DESCRIBED BY EINSTEIN WILL BECOME APPARENT.

THANK YOU.


READY
```

```
100 REM A.C. CAGGIANO;PATCHOGUE H.S. PHYSICS; 7-16-68
105 REM  REVISED BY C.LOSIK  8-21-70
106 REM  A IS WHICH METAL, L IS THE PRINTED WAVELENGTH,
107 REM  N ARE THE CURRENTS, K IS INTENSITY, F IS INCREASING INTENSITY
110 PRINT " ","THE PHOTOELECTRIC EFFECT"
120PRINT
130PRINT
140PRINT"WHEN LIGHT OF SHORT WAVELENGTH FALLS ON A METAL SURFACE,"
150PRINT"ELECTRONS ARE EJECTED FROM THE METAL. ACCORDING TO THE"
160PRINT"DESCRIPTION OF THIS PHENOMENON BY EINSTEIN, THERE IS A"
170PRINT"MAXIMUM WAVELENGTH FOR EACH METAL ABOVE WHICH NO ELECTRONS"
180 PRINT"ARE EMITTED. IN THIS EXPERIMENT WE WILL DETERMINE THE"
190PRINT"CRITICAL WAVELENGTH AT WHICH THIS OCCURS."
200PRINT
210PRINT"THE METAL SELECTED WILL BE PLACED IN A VACUUM WHERE IT"
220PRINT"WILL BE BOMBARDED BY SOFT X-RAYS. THE NUMBER OF ELECTRONS"
230PRINT"EJECTED WILL BE COLLECTED AND COUNTED WITH AN AMMETER."
240PRINT"(NOTE:THE CURRENT IS RELATED TO THE NUMBER OF ELECTRONS"
250PRINT"EMITTED BY THE METAL)."
260PRINT
270PRINT"SELECT ONE OF THE METALS LISTED BY TYPING ITS NUMBER."
280PRINT
290PRINT" ","1) SILVER"
300PRINT" ","2) BISMUTH"
310PRINT" ","3) CADMIUM"
320PRINT" ","4) LEAD"
330PRINT" ","5) PLATINUM"
340PRINT
345 RANDOMIZE
350INPUT A
360 IF A>1 THEN 380
370 LET V0=.308
380 IF A<>2 THEN 400
390 LET V0=.338
400 IF A<>3 THEN 420
410 LET V0=.318
420 IF A<>4 THEN 440
430LET V0=.340
440 IF A<5 THEN 460
450 LET V0=.385
460 LET K=INT(1+2*RND(X))
470 PRINT
480 PRINT" ","MEASURED CURRENT (MICROAMPERES)"
490PRINT "WAVELENGTH","TRIAL 1","TRIAL 2","TRIAL 3"
500 FOR L=.420 TO .250 STEP -.02
510 LET M=INT(1000/L)
520 PRINT M,
530 FOR J=1 TO 3
540 IF L> V0 THEN 570
550 LET I=SQR(INT(25*RND(X)))
560 GO TO 580
570 LET I=SQR(K*K*100+INT(35*RND(X)))
580 LET N=INT(10*I+.5)/10
590 PRINT N,
```

```
600 NEXT J
610 PRINT
620 NEXT L
630PRINT
640PRINT"DO YOU WISH TO INCREASE THE LIGHT INTENSITY?"
650 PRINT "(1=YES, 0=NO) : ";
660 INPUT G
670 IF G=0 THEN 730
675 IF G<>1 THEN 650
680 PRINT
690 PRINT"BY WHAT FACTOR? (SELECT FACTOR BETWEEN 1 AND 10)."
700 INPUT F
705 IF ABS(F-5.5)>4.5 THEN 690
710 LET K=K*F
720 GO TO 470
730 PRINT
740PRINT"DO YOU WISH TO TRY ANOTHER METAL (1=YES, 0=NO) : ";
750 INPUT H
760 IF H=1 THEN 270
765 IF H<>0 THEN 740
770 PRINT
780PRINT"NOW BY PLOTTING THE WAVELENGTH VS. THE MEASURED CURRENT,"
790PRINT"(AVERAGE OF THREE TRIALS), THE PHOTOELECTRIC EFFECT AS"
800PRINT"DESCRIBED BY EINSTEIN WILL BECOME APPARENT."
810PRINT
820PRINT"THANK YOU."
830 END
```

DISCIPLINE PHYSICS

SUBJECT ENERGY LEVELS

PROGRAM NAME PHOTON

DESCRIPTION:

The student fires 15 shots, from a photon gun, at a mythical gaseous element with 4 randomly-selected energy levels. After each shot, the computer prints out the energies of photons, if any, emitted by the gas. The student is to construct an energy level diagram for the element from a knowledge of the energies of the photons emitted.

OBJECTIVES:

To promote a better understanding of how energy levels are determined from a knowledge of the emissions of excited atoms.

PRELIMINARY PREPARATION:

A. Student - It is desirable that he have run BOHR, but it is not a necessity.

B. Materials - none

DISCUSSION:

The computer randomly selects 4 energy levels for the element. The energies range between 1×10^{-19} and 15×10^{-19} joules.

The energies of the students' 15 shots are picked at random, but cover the range from 1 to 15. Whenever one of the photons shot by the student is capable of exciting the atom all of the possible photon emissions from that excited state are printed.

By examining the photons emitted as a result of the 15 shots the student can construct an energy-level diagram of the element and account for each photon.

IMAGINE THAT YOU HAVE A PHOTON GUN THAT FIRES PHOTONS WITH RANDOMLY SELECTED ENERGIES.

YOU WANT TO FIND SOME OF THE ENERGY LEVELS OF A GAS THAT YOU HAVE ISOLATED FROM A SAMPLE OF MOON ROCK. YOU WILL DO IT BY FIRING PHOTONS INTO THE GAS AND MEASURING THE ENERGIES OF PHOTONS EMITTED BY THE GAS. THE GAS WILL EMIT ONLY IF THE PHOTON YOU FIRED IS CAPABLE OF EXCITING ITS ATOMS TO HIGHER ENERGY STATES.

TO FIRE A BURST OF SINGLE ENERGY PHOTONS INTO THE GAS TYPE 1
TO CEASE FIRING PHOTONS TYPE 0
YOU HAVE 15 SHOTS TO DETERMINE THE ENERGY LEVELS.

	SHOT NUMBER	ENERGY OF EMITTED PHOTONS (E-19 JOULES)
FIRE!!? 1	1	0
FIRE!!? 1	2	0
FIRE!!? 1	3	0
FIRE!!? 1	4	0
FIRE!!? 1	5	14 7 5 3 11 4 8 9
FIRE!!? 1	6	5 3 8 9
FIRE!!? 1	7	3 9
FIRE!!? 1	8	0
FIRE!!? 1	9	0
FIRE!!? 1	10	0
FIRE!!? 1	11	0
FIRE!!? 1	12	7 5 3 4 8
FIRE!!? 1	13	0
FIRE!!? 1	14	0
FIRE!!? 1	15	0

FIND THE ENERGY LEVELS OF OUR ELEMENT - MYSTERIUM AND ACCOUNT FOR EACH OF THE EMITTED PHOTONS BY DRAWING AN ENERGY LEVEL DIAGRAM AND SHOWING WHICH TRANSITIONS GIVE RISE TO THE PHOTONS.

READY

```
110 REM   JOHN W. HOSIE - NORTHPORT HIGH - PHYSICS   8/1/69
113 REM  REVISED BY L.BRAUN  8-20-70
115 REM  R(I) ARE THE ENERGY LEVELS
120 RANDOMIZE
125 DIM R(25),K(15),E(6)
130 FOR J=0 TO 6
140 LET E(J)=0
150 NEXT J
170 PRINT"IMAGINE THAT YOU HAVE A PHOTON GUN THAT FIRES PHOTONS WITH"
180 PRINT"RANDOMLY SELECTED ENERGIES."
190 PRINT
200 PRINT"YOU WANT TO FIND SOME OF THE ENERGY LEVELS OF A GAS THAT"
210 PRINT"YOU HAVE ISOLATED FROM A SAMPLE OF MOON ROCK.  YOU WILL"
220 PRINT"DO IT BY FIRING PHOTONS INTO THE GAS AND MEASURING THE"
230 PRINT"ENERGIES OF PHOTONS EMITTED BY THE GAS.  THE GAS WILL EMIT"
240 PRINT"ONLY IF THE PHOTON YOU FIRED IS CAPABLE OF EXCITING ITS"
250 PRINT"ATOMS TO HIGHER ENERGY STATES."
260 PRINT
270 PRINT
300 REM  THIS GENERATES A RANDOM NO. IN RANGE 0-15
310 LET K=INT(15*RND(X)+.5)
320 FOR I=2 TO 5
330 IF K=E(I) THEN 310
340 NEXT I
350 FOR J=2 TO 5
360 IF E(J)=0 THEN 390
370 NEXT J
380 GO TO 410
390 LET E(J)=K
400 GO TO 310
410 FOR J=1 TO 4
420 FOR I=J+1 TO 5
430 IF E(J)<E(I) THEN 470
440 LET K=E(J)
450 LET E(J)=E(I)
460 LET E(I)=K
470 NEXT I
480 NEXT J
490 PRINT"TO FIRE A BURST OF SINGLE ENERGY PHOTONS INTO THE GAS TYPE 1"
500 PRINT"TO CEASE FIRING PHOTONS TYPE 0"
510 PRINT"YOU HAVE 15 SHOTS TO DETERMINE THE ENERGY LEVELS."
530 PRINT
540 PRINT " ","SHOT NUMBER","ENERGY OF EMITTED PHOTONS (E-19 JOULES)"
550 PRINT
560 IF D=15 THEN 950
570 PRINT "FIRE!!";
580 INPUT F
583 IF F=0 THEN 950
586 IF F<>1 THEN 570
590 LET D=D+1
600 LET N=0
620 LET P=INT(15*RND(X)+.5)
630 FOR I=1 TO 15
640 IF P=K(I) THEN 620
650 NEXT I
660 LET K(D)=P
670 FOR I=1 TO 5
680 IF P=E(I) THEN 720
690 NEXT I
700 PRINT " ",D,"0"
710 GO TO 560
720 FOR H=1 TO I
```

```
730 FOR J=1 TO I
740 LET N=N+1
750 LET R(N)=E(I+1-J)-E(H)
760 NEXT J
770 NEXT H
780 PRINT" ",D,
790 FOR I=1 TO 25
800 IF R(I)>0 THEN 830
810 LET R(I)=0
820 GO TO 870
830 FOR J=1 TO 25-I
840 IF R(I)<>R(I+J) THEN 860
850 LET R(I+J)=0
860 NEXT J
870 NEXT I
880 FOR N=1 TO 25
890 IF R(N)=0 THEN 920
900 PRINT R(N);
920 NEXT N
930 PRINT" "
940 GO TO 560
950 PRINT
960 PRINT
970 PRINT"FIND THE ENERGY LEVELS OF OUR ELEMENT - MYSTERIUM"
980 PRINT"AND ACCOUNT FOR EACH OF THE EMITTED PHOTONS BY DRAWING"
990 PRINT"AN ENERGY LEVEL DIAGRAM AND SHOWING WHICH TRANSITIONS"
1000 PRINT"GIVE RISE TO THE PHOTONS."
1010 END
```

DISCIPLINE PHYSICS

SUBJECT PHOTOELECTRIC EFFECT

PROGRAM NAME PLANK

DESCRIPTION:

This program simulates an experiment to determine Planck's constant, threshold frequency, and work function of a metal.

OBJECTIVES:

A. To enable the student to do an experiment on the computer that he is not likely to be able to do in a high-school laboratory.

B. A better understanding of the photoelectric effect.

PRELIMINARY PREPARATION:

A. Student

1. He should have read and studied about threshold frequency, cut-off potential, and know (schematically) how the experimental apparatus used in such an experiment works.
2. It is desirable that he have run PHOTEL, though not a necessity.

B. Materials - Graph paper

DISCUSSION:

The student may choose one of the five metals in the program, the intensity of the x-rays used, and the number of different x-ray frequencies he would like to use. The computer then randomly chooses an x-ray frequency, and prints it for the student to see. The student enters voltages to be used as retarding potentials in the simulated tube and the computer prints a current for each potential entered until the current is zero when the cut-off potential is reached. A new frequency x-ray is then used and the student again tries to find the correct potential for cut-off.

Finally, a table of frequencies and cut-off potentials are printed and an assignment given (plot a graph and answer questions).

The student may then run the program again with a different intensity and the same metal, or he may change the metal and intensity.

```
  IN THIS EXPERIMENT YOU WILL BE GIVEN THE FREQUENCY OF THE
X-RAYS BEING USED AND YOU ARE TO DETERMINE THE VOLTAGE SET-
TING (RETARDING POTENTIAL) NECESSARY TO CAUSE THE COLLECTOR
CURRENT TO DECREASE TO ZERO.

  FIRST CHOOSE THE METAL YOU WISH TO USE FOR YOUR PHOTO-
SENSITIVE SURFACE.

  1 SILVER,  2 BISMUTH,  3 CADMIUM,  4 LEAD,  5 PLATINUM

WHICH METAL DO YOU CHOOSE? 4

WHAT INTENSITY OF X-RAYS WILL YOU USE (FROM 1 TO 5)? 3

HOW MANY DIFFERENT X-RAY FREQUENCIES WOULD YOU LIKE TO
USE TO RADIATE YOUR SAMPLE (FROM 5 TO 8)? 7

THE X-RAY FREQUENCY IS 14.59   E15

VOLTAGES HIGHER THAN CUT OFF WILL GIVE CURRENT READINGS OF
ZERO SO TRY LOWER ONES.  I'LL NOTIFY YOU OF CUT OFF.
  FIND THE CUT OFF (STOPPING) VOLTAGE.

V=? 24
          I=0
V=? 20
          I= 9.780632   E-6
V=? 23
          I= 2.232173   E-6
V=? 23.5
          I= .9988859   E-6
V=? 23.6
          I= .747686    E-6
V=? 23.8
           CUT OFF   I=0

THE X-RAY FREQUENCY IS 18.24   E15

  FIND THE CUT OFF (STOPPING) VOLTAGE.

V=? 24
          I= 83.08785   E-6
V=? 28
          I= 16.97117   E-6
V=? 35
          I= 6.16513    E-6
V=? 40
          I=0
V=? 38
          I= 1.580819   E-6
V=? 39.5
          I=0
V=? 39
           CUT OFF   I=0

THE X-RAY FREQUENCY IS 9.06   E15

  FIND THE CUT OFF (STOPPING) VOLTAGE.

V=? 10
          I=0
V=? 5
          I=0
V=? 1
           CUT OFF   I=0
```

```
THE X-RAY FREQUENCY IS 13.2   E15

  FIND THE CUT OFF (STOPPING) VOLTAGE.

V=? 20
          I=0
V=? 17
          I= 3.755741   E-6
V=? 19
          I=0
V=? 18
          I= .4778505   E-6
V=? 18.5
          I=0
V=? 18.2
           CUT OFF   I=0

THE X-RAY FREQUENCY IS 12.44   E15

  FIND THE CUT OFF (STOPPING) VOLTAGE.

V=? 15
           CUT OFF   I=0

THE X-RAY FREQUENCY IS 9.43   E15

  FIND THE CUT OFF (STOPPING) VOLTAGE.

V=? 10
          I=0
V=? 5
          I=0
V=? 2
          I= 12.50934   E-6
V=? 3
          I=0
V=? 2.7
          I=0
V=? 2.5
           CUT OFF   I=0

THE X-RAY FREQUENCY IS 8.65   E15

  FIND THE CUT OFF (STOPPING) VOLTAGE.

V=? 1
          I=0
V=? .4
          I=0
V=? .2
THIS FREQUENCY IS TOO LOW TO CAUSE PHOTOELECTRIC EMISSION
I'LL GIVE YOU A NEW FREQUENCY.

THE X-RAY FREQUENCY IS 10.55   E15

  FIND THE CUT OFF (STOPPING) VOLTAGE.

V=? 10
          I=0
V=? 8
          I=0
V=? 5
          I= 18.15371   E-6
V=? 6
          I= 9.785334   E-6
V=? 7
          I= 1.385315   E-6
V=? 7.5
          I=0
V=? 7.3
          I=0
V=? 7.9
           CUT OFF   I=0
```

X-RAY FREQUENCY E15 FPS	CUT OFF VOLTAGE VOLTS
14.59	23.8
18.84	39
9.06	1
13.2	18.2
12.44	15
9.43	8.5
10.55	7.8

PLOT A GRAPH OF CUT OFF VOLTAGES (Y AXIS) VS. FREQUENCY

WHAT IS THE MEANING OF THE POINT AT WHICH THE EXTRAPOLATED GRAPH INTERCEPTS THE VOLTAGE AXIS?

WHAT IS THE LOWEST FREQUENCY THAT WILL CAUSE EMISSION OF PHOTOELECTRONS FROM THIS METAL?

REMEMBER THAT THE RETARDING POTENTIAL APPLIED BETWEEN THE EMITTER AND THE COLLECTOR AT CUT OFF, EXPRESSED IN ELECTRON VOLTS, IS EQUAL TO THE KINETIC ENERGY OF THE FASTEST ELECTRONS ESCAPING FROM THE EMITTER. FIND THE SLOPE OF THE GRAPH BUT EXPRESS THE STOPPING POTENTIAL IN JOULES.

WHAT IS THE VALUE OF THE SLOPE OF THE GRAPH AND WHAT SPECIAL NAME IS GIVEN TO THIS CONSTANT?

THE SAME METAL WITH A DIFFERENT INTENSITY IS WORTH INVESTIGATING. WHEN YOU DO THIS EXPLAIN THE MEANING OF ITS GRAPH WHEN COMPARED TO THE PREVIOUS ONE.

YOU MAY ALSO WISH TO TRY A DIFFERENT METAL AND EXPLAIN THE MEANING OF ITS GRAPH WHEN COMPARED TO YOUR OTHER ONES, OR COMPARED WITH THOSE OF ANOTHER STUDENT.

DO YOU WISH TO TRY A DIFFERENT INTENSITY OR A DIFFERENT METAL (1=YES, 0=NO) ? 0

READY

```
100 REM JOHN W. HOSIE  - NORTHPORT HIGH -  7/29/69
105 REM  REVISED BY C.LOSIK  8-21-70
106 REM  M IS WHICH METAL, C IS X-RAY INTENSITY, K IS NO. OF FREQS.
107 REM  F(S) ARE THE FREQS. USED, V(S) ARE THE CUT OFF POINTS
110 LET P=0
120 PRINT
130 PRINT "  IN THIS EXPERIMENT YOU WILL BE GIVEN THE FREQUENCY OF THE"
140PRINT "X-RAYS BEING USED AND YOU ARE TO DETERMINE THE VOLTAGE SET-"
150 PRINT "TING (RETARDING POTENTIAL) NECESSARY TO CAUSE THE COLLECTOR"
160 PRINT "CURRENT TO DECREASE TO ZERO."
170 PRINT
180 PRINT "  FIRST CHOOSE THE METAL YOU WISH TO USE FOR YOUR PHOTO-"
190 PRINT "SENSITIVE SURFACE."
200 PRINT
210 PRINT "  1 SILVER,  2 BISMUTH,  3 CADMIUM,  4 LEAD,  5 PLATINUM"
220 PRINT
230 LET K=0
240 PRINT "WHICH METAL DO YOU CHOOSE";
250 INPUT M
260 IF M>=1 THEN 280
270 GO TO 430
280 IF M>1 THEN 310
290 LET F0=9.74
300 GO TO 450
310 IF M>2 THEN 340
320 LET F0=8.88
330 GO TO 450
340 IF M>3 THEN 370
350 LET F0=9.43
360 GO TO 450
370 IF M>4 THEN 400
380 LET F0=8.82
390 GO TO 450
400 IF M>5 THEN 430
410 LET F0=7.79
420 GO TO 450
430 PRINT "SORRY - THE METALS HAVE NUMBERS FROM 1 TO 5"
440 GO TO 240
450 DIM F(10),V(10)
460 PRINT
480 PRINT "WHAT INTENSITY OF X-RAYS WILL YOU USE (FROM 1 TO 5)";
490 INPUT C
500 IF C>5 THEN 480
510 IF C<1 THEN 480
520 LET S=0
530 PRINT
540 PRINT"HOW MANY DIFFERENT X-RAY FREQUENCIES WOULD YOU LIKE TO"
550 PRINT"USE TO RADIATE YOUR SAMPLE (FROM 5 TO 8)";
560 INPUT K
570 PRINT
580 IF K>=5 THEN 610
590 PRINT "I SAID BETWEEN 5 AND 8 FREQUENCIES."
600 GO TO 530
610 IF K<=8 THEN 680
620 PRINT "TOO MANY TRIALS FOR THE AVAILABLE TIME."
630 GO TO 530
650 RANDOMIZE
670 LET R=0
680 FOR I=1 TO 100
690 LET P=RND(X)
```

```
700 LET F=INT(8000*F+.5)/100
710 IF F>7 THEN 730
720 NEXT I
730 LET V=4.14*(F-F0)
740 LET J=0
750 PRINT "THE X-RAY FREQUENCY IS"F"  E15"
760 PRINT
770 IF R>0 THEN 800
780 PRINT "VOLTAGES HIGHER THAN CUT OFF WILL GIVE CURRENT READINGS OF"
790 PRINT "ZERO SO TRY LOWER ONES.  I'LL NOTIFY YOU OF CUT OFF."
800 PRINT "  FIND THE CUT OFF (STOPPING) VOLTAGE."
810 PRINT
820 PRINT "V=";
830 INPUT V1
840 LET R=R+1
850 LET J=J+1
860 LET I=(20*C*(V-V1))/V+.05*RND(X)
870 IF I<C*20.5 THEN 890
880 LET I=20*C+.1*RND(X)
890 IF V-V1>0 THEN 910
900 LET I=0
910 IF ABS(V-V1)<.1 THEN 1000
920 IF V>0 THEN 950
930 LET I=0
940 IF J=3 THEN 1020
950 IF I=0 THEN 980
960 PRINT "           I="I"  E-6"
970 GO TO 820
980 PRINT "           I=0"
990 GO TO 820
1000 PRINT "            CUT OFF    I=0"
1010 GO TO 1060
1020 PRINT "THIS FREQUENCY IS TOO LOW TO CAUSE PHOTOELECTRIC EMISSION"
1030 PRINT "I'LL GIVE YOU A NEW FREQUENCY."
1040 PRINT
1050 GO TO 680
1060 LET S=S+1
1070 LET F(S)=F
1080 LET V(S)=V1
1090 PRINT
1100 IF S<>K THEN 680
1110 PRINT "  X-RAY","CUT OFF"
1120 PRINT "FREQUENCY","VOLTAGE"
1130 PRINT " E15 FPS","VOLTS"
1140 PRINT "---------","-------"
1150 PRINT
1160 FOR S=1 TO K
1170 PRINT F(S),V(S)
1180 NEXT S
1190 LET P=P+1
1210 PRINT
1220 PRINT"  PLOT A GRAPH OF CUT OFF VOLTAGES (Y AXIS) VS. FREQUENCY"
1230 PRINT
1235 IF P>1 THEN 1460
1240 PRINT"WHAT IS THE MEANING OF THE POINT AT WHICH THE EXTRAPOLATED"
1250 PRINT"GRAPH INTERCEPTS THE VOLTAGE AXIS?"
1260 PRINT
1270 PRINT"WHAT IS THE LOWEST FREQUENCY THAT WILL CAUSE EMISSION OF"
1280 PRINT"PHOTOELECTRONS FROM THIS METAL?"
1290 PRINT
1300 PRINT"REMEMBER THAT THE RETARDING POTENTIAL APPLIED BETWEEN THE"
1310 PRINT"EMITTER AND THE COLLECTOR AT CUT OFF, EXPRESSED IN ELECTRON"
1320 PRINT"VOLTS, IS EQUAL TO THE KINETIC ENERGY OF THE FASTEST"
1330 PRINT"ELECTRONS ESCAPING FROM THE EMITTER.  FIND THE SLOPE OF THE"
```

```
1340 PRINT"GRAPH BUT EXPRESS THE STOPPING POTENTIAL IN JOULES."
1350 PRINT
1360 PRINT"WHAT IS THE VALUE OF THE SLOPE OF THE GRAPH AND WHAT SPECIAL"
1370 PRINT"NAME IS GIVEN TO THIS CONSTANT?"
1380 PRINT
1390 PRINT"THE SAME METAL WITH A DIFFERENT INTENSITY IS WORTH"
1400 PRINT"INVESTIGATING.  WHEN YOU DO THIS EXPLAIN THE MEANING"
1410 PRINT"OF ITS GRAPH WHEN COMPARED TO THE PREVIOUS ONE."
1420 PRINT
1430 PRINT"YOU MAY ALSO WISH TO TRY A DIFFERENT METAL AND EXPLAIN"
1440 PRINT"THE MEANING OF ITS GRAPH WHEN COMPARED TO YOUR OTHER ONES,"
1450 PRINT "OR COMPARED WITH THOSE OF ANOTHER STUDENT."
1460 PRINT
1470 PRINT "DO YOU WISH TO TRY A DIFFERENT INTENSITY OR A"
1480 PRINT "DIFFERENT METAL (1=YES, 0=NO) : ";
1490 INPUT Q
1500 IF Q=1 THEN 170
1510 IF Q<>0 THEN 1460
1520 END
```

DISCIPLINE PHYSICS

SUBJECT PROJECTILE MOTION

PROGRAM NAME PRJTL

DESCRIPTION:

By entering the firing angle and initial speed, the computer calculates the coordinates, vertical and horizontal velocities, and speed of a projectile for equal time intervals.

OBJECTIVES:

To show the independence of the horizontal and vertical velocities of a projectile, and to facilitate the plotting of its path by eliminating tedious calculations.

PRELIMINARY PREPARATION:

A. Student - Knowledge of motion at constant velocity and at constant acceleration; and the vector nature of velocity and acceleration.

B. Materials - graph paper

DISCUSSION:

The student enters an angle and an initial speed of a projectile. A table of time, X and Y coordinates, horizontal and vertical velocities, and speed of the projectile is printed.

The student may then plot a graph of the position of the projectile, and draw vectors at each coordinate point to show the vertical and horizontal components of its velocity.

SUPPOSE YOU ARE GOING TO FIRE A PROJECTILE INTO THE AIR.
IF YOU ENTER A VALUE FOR ANGLE OF ELEVATION AND INITIAL
VELOCITY, THE RANGE AND HEIGHT WILL BE EVALUATED. ENTER
YOUR INFORMATION IN THE FORM A,V AFTER THE QUESTION MARK.
(REMEMBER, THE ANGLE IS IN DEGREES AND THE INITIAL
VELOCITY IS IN METERS/SECOND.)

WHAT ARE YOUR VALUES? 30, 200

THE TOTAL FLIGHT TIME WAS 20.39431 SECONDS
THE RANGE WAS 3532.399 METERS
THE MAXIMUM HEIGHT WAS 509.8573 METERS

BECAUSE THERE IS NO FRICTION, THE HORIZONTAL VELOCITY IS
CONSTANT. HORIZONTAL VELOCITY = 173.2051

THE FOLLOWING ARE POINTS ON THE CURVE AT VARIOUS TIME INTERVALS:

TIME	X-COORD	Y-COORD	VERTICAL VELOCITY	SPEED
0	0	0	99.99998	200
1.854028	321.1272	168.5479	81.81812	191.5574
3.708056	642.2543	303.3861	63.63632	184.5253
5.562084	963.3815	404.5148	45.45451	179.0702
7.416112	1284.509	471.9338	27.27271	175.3391
9.27014	1605.636	505.6432	9.090902	173.4435
11.12417	1926.763	505.643	-9.090901	173.4435
12.9782	2247.89	471.9332	-27.27271	175.3391
14.83222	2569.017	404.5136	-45.45451	179.0702
16.68625	2890.144	303.3848	-63.63631	184.5253
18.54028	3211.272	168.5468	-81.81818	191.5574
20.39431	3532.399	0	-99.99998	200

THE ANGLE AT WHICH YOU FIRED THE PROJECTILE DOES NOT YIELD THE MAX
IMUM RANGE. WHAT ANGLE DOES? 45

45 DEGREES GIVES THE MAXIMUM RANGE OF 4077.654

WOULD YOU LIKE ANOTHER RUN WITH DIFFERENT A AND V?
(1=YES, 0=NO) : ? 0

READY

```
100 REM J.CARACCIOLO, LONGWOOD H.S., 10-26-68,   BASIC
101 REM REVISED 8/25/70 (C. LOSIK)
110 REM PHYSICS       PROJECTILE MOTION
120 PRINT "SUPPOSE YOU ARE GOING TO FIRE A PROJECTILE INTO THE AIR."
130 PRINT "IF YOU ENTER A VALUE FOR ANGLE OF ELEVATION AND INITIAL"
140 PRINT "VELOCITY, THE RANGE AND HEIGHT WILL BE EVALUATED.  ENTER"
150 PRINT "YOUR INFORMATION IN THE FORM A,V AFTER THE QUESTION MARK."
160 PRINT "(REMEMBER, THE ANGLE IS IN DEGREES AND THE INITIAL"
170 PRINT "VELOCITY IS IN METERS/SECOND.)"
180 PRINT
190 PRINT
200 PRINT "WHAT ARE YOUR VALUES";
220 INPUT A,V0
230 IF V0=0 THEN 690
240 IF A=0 THEN 690
250 IF V0<0 THEN 720
260 IF A<0 THEN 720
270 IF A>=90 THEN 750
280 LET A=A*3.14159/180
290 LET K=V0*SIN(A)
300 LET L=V0*COS(A)
310 LET T=2*K/9.80665
320 LET R=2*K*L/9.80665
330 LET H=(K↑2)/19.6133
340 PRINT
350 PRINT
360 PRINT"THE TOTAL FLIGHT TIME WAS";T;"SECONDS"
370 PRINT"THE RANGE WAS";R;"METERS"
380 PRINT"THE MAXIMUM HEIGHT WAS";H;"METERS"
390 PRINT
393 PRINT "BECAUSE THERE IS NO FRICTION, THE HORIZONTAL VELOCITY IS"
396 PRINT "CONSTANT.  HORIZONTAL VELOCITY ="L
400 PRINT
410 PRINT"   THE FOLLOWING ARE POINTS ON THE CURVE AT VARIOUS ";
420 PRINT "TIME INTERVALS:"
430 PRINT
440 PRINT
445 PRINT " "," "," "," VERTICAL "
450 PRINT" TIME "," X-COORD "," Y-COORD "," VELOCITY "," SPEED "
460 PRINT" ---- "," ------- "," ------- "," -------- "," ----- "
470 LET N=T/11
480 LET T1=T
490 FOR T=0 TO T1 STEP N
491 LET Q=K*T-4.90333*T↑2
492 IF Q>0 THEN 495
493 LET Q=0
495 LET V1=K-9.80665*T
500 PRINT T,L*T,Q,V1,SQR(V1*V1+L*L)
510 NEXT T
520 IF ABS(A-.785398)<.00001 THEN 610
530 PRINT
540 PRINT
550 PRINT"THE ANGLE AT WHICH YOU FIRED THE PROJECTILE DOES NOT ";
560 PRINT "YIELD THE MAXIMUM RANGE.  WHAT ANGLE DOES";
580 INPUT A
590 LET A=A*3.14159/180
600 GO TO 520
610 PRINT
620PRINT"45 DEGREES GIVES THE MAXIMUM RANGE OF"2*(V0↑2)*(.707↑2)/9.8066
640 PRINT
642 PRINT "WOULD YOU LIKE ANOTHER RUN WITH DIFFERENT A AND V?"
```

```
644 PRINT "(1=YES, 0=NO) : ";
645 INPUT V1
646 IF V1=0 THEN 78
648 IF V1<>1 THEN 644
649 PRINT
650 PRINT"ENTER NEW VALUES FOR A,V AFTER THE QUESTION MARK."
660 GO TO 180
690 PRINT "  DON'T ENTER VALUES OF ZERO.";
700 GO TO 180
720 PRINT "  NO NEGATIVE VALUES.  PLEASE ENTER THEM CORRECTLY";
730 GO TO 180
750 PRINT "  GREAT SHOT.  YOU COULD KILL YOURSELF THAT WAY, YOU KNOW."
760 PRINT "  TRY AGAIN (NOT TO KILL YOURSELF, THAT IS) ";
770 GO TO 180
780 END
```

DISCIPLINE PHYSICS

SUBJECT PRINCIPLE OF LEAST TIME

PROGRAM NAME REFLCT

DESCRIPTION:

An analogy is given for a light-ray reflected from a plane surface to demonstrate the "least-time" principle and its relationship to the reflection laws of light.

OBJECTIVES:

To demonstrate the consequences of the "least-time" principle.

PRELIMINARY PREPARATION:

A. Student - Should be familiar with the reflection laws of light.

B. Materials - graph paper

DISCUSSION:

Given points P_1 and P_2 and the line l, the student can vary the point P_3 to note the effects on angles P and Q and their relationship to the time required to traverse the path $\overline{P_1P_3P_2}$.



The program is presented as a game in which a horse (lightray) must complete a journey within a specified time. The student is limited to seven choices of P_3 to complete the task. After a successful journey, the student may vary the point P_2 to further establish the principle of least time.

This program has been extremely helpful in developing the least-time concept and its relationship to the laws of reflection.

It is applicable to a classroom situation as well as small study groups.

```
YOU ARE CAMPING OUT WEST IN COORDINATE NATIONAL PARK ON
ORDINATE MOUNTAIN, LOCATED 10 MILES NORTH OF THE
DESERTED TOWN OF ORIGIN, WHICH IS CONVENIENTLY LOCATED
AT (0,0) ON THE LOCAL MAP.

A CALAMITY STRIKES!  THE NEAREST HELP IS AT THE
BAR 30:30 RANCH, LOCATED AT COORDINATES (30,30).
TO GET THERE, YOU MUST RIDE AN OLD HORSE
(NAMED LIGHTRAY)  WHO :
          A) WILL ONLY WALK 5 MILES PER HOUR
          B) WILL CEASE TO WALK (AND EXIST) AFTER 10 HOURS
          C) MUST HAVE A DRINK OF WATER SOMEWHERE ALONG THE
          ABSCISSA RIVER, WHICH (IF YOU HAVEN'T GUESSED) RUNS
          ALONG THE ABSCISSA IN COORDINATE PARK

HERE IS YOUR PROBLEM:  YOU MUST PICK A SPOT
(FROM 0 TO 30) ALONG THE ABSCISSA RIVER DURING THE
TRIP TO GIVE LIGHTRAY A DRINK, AND STILL MAKE IT TO
THE BAR 30:30 WITHIN THE TIME ALLOWED.  LIGHTRAY, USING
HORSE SENSE, KNOWS ALL THE ANGLES, SO WE WILL GIVE
THEM TO YOU, TOO.

WHERE WILL LIGHTRAY STOP FOR A DRINK? 7

ANGLE APPROACHING RIVER IS 55 DEGREES.
ANGLE LEAVING RIVER IS 53 DEGREES.

WHERE WILL LIGHTRAY STOP FOR A DRINK? 7.25

ANGLE APPROACHING RIVER IS 54 DEGREES.
ANGLE LEAVING RIVER IS 53 DEGREES.
WELL, YOU ARE CLOSER THAN LAST TIME.
KEEP AN EYE ON THOSE ANGLES, THOUGH.
LET'S GO BACK ... OR ANOTHER HORSE.

WHERE WILL LIGHTRAY STOP FOR A DRINK? 7.3

ANGLE APPROACHING RIVER IS 54 DEGREES.
ANGLE LEAVING RIVER IS 53 DEGREES.
C'MON -- YOU TRIED THAT LAST TIME.

WHERE WILL LIGHTRAY STOP FOR A DRINK? 7.35

ANGLE APPROACHING RIVER IS 54 DEGREES.
ANGLE LEAVING RIVER IS 53 DEGREES.
C'MON -- YOU TRIED THAT LAST TIME.

WHERE WILL LIGHTRAY STOP FOR A DRINK? 7.4

ANGLE APPROACHING RIVER IS 53 DEGREES.
ANGLE LEAVING RIVER IS 53 DEGREES.
NICE WORK.  YOU MADE IT.
THE TRIP TOOK ABOUT 10.00007  HOURS.
YOU CAN SEE THAT USING HORSE SENSE, LIGHTRAY KNOWS THAT
THE ANGLES HAVE TO BE EQUAL OF REFLECTION FOR A
MINIMUM TIME TRIP.

IF YOU WANT TO MOVE THE RANCH, TYPE 1
IF YOU WANT TO SEE SOMETHING ELSE, TYPE 2
IF YOU WANT TO QUIT, TYPE 3
? 3

                THANK YOU FOR PLAYING.


READY
```

```
100 REM  PROGRAM BY GERARD M. DAMM, WYANDANCH HS,  8/68
101 REM  REVISED BY C.LOSIK  8-16-70
110 DIM A(7),B(7),C(7)
115 REM  INTRODUCTION
120 PRINT "YOU ARE CAMPING OUT WEST IN COORDINATE NATIONAL PARK ON"
130 PRINT "ORDINATE MOUNTAIN, LOCATED 10 MILES NORTH OF THE"
140 PRINT "DESERTED TOWN OF ORIGIN, WHICH IS CONVENIENTLY LOCATED"
150 PRINT "AT (0,0) ON THE LOCAL MAP."
155 PRINT
160 PRINT "A CALAMITY STRIKES!  THE NEAREST HELP IS AT THE"
170 PRINT "BAR 30:30 RANCH, LOCATED AT COORDINATES (30,30)."
180 PRINT "TO GET THERE, YOU MUST RIDE AN OLD HORSE"
190 PRINT "(NAMED LIGHTRAY)  WHO :"
200 PRINT " ","A) WILL ONLY WALK 5 MILES PER HOUR"
210 PRINT " ","B) WILL CEASE TO WALK (AND EXIST) AFTER 10 HOURS"
220 PRINT " ","C) MUST HAVE A DRINK OF WATER SOMEWHERE ALONG THE"
230 PRINT " ","ABSCISSA RIVER, WHICH (IF YOU HAVEN'T GUESSED) RUNS"
240 PRINT " ","ALONG THE ABSCISSA IN COORDINATE PARK"
245 PRINT
250 PRINT "HERE IS YOUR PROBLEM:  YOU MUST PICK A SPOT"
260 PRINT "(FROM 0 TO 30) ALONG THE ABSCISSA RIVER DURING THE"
270 PRINT "TRIP TO GIVE LIGHTRAY A DRINK, AND STILL MAKE IT TO"
280 PRINT "THE BAR 30:30 WITHIN THE TIME ALLOWED.  LIGHTRAY, USING"
290 PRINT "HORSE SENSE, KNOWS ALL THE ANGLES, SO WE WILL GIVE"
295 PRINT "THEM TO YOU, TOO."
300 REM  FIRST RUN, INITIALIZE RANCH
310 LET A=30
320 LET B=30
330 LET T9=10
335 LET R=180/3.14159
340 GO TO 500
345 REM  SECOND RUN (OPTIONAL)
350 PRINT
360 PRINT "ENTER TWO NUMBERS, SEPARATED BY A COMMA, FOR THE NEW"
370 PRINT "RANCH COORDINATES  ";
380 INPUT A,B
390 LET X=10*A/(B+10)
400 LET T9=(SQR(X*X+100)+SQR((A-X)*(A-X)+B*B))/5
410 PRINT "O.K., LIGHTRAY WILL LAST ABOUT"T9+.001"HOURS THIS TIME."
499 REM  LOOP FOR SEVEN TRYS - EXIT LOOP IF CORRECT
500 FOR I=1 TO 7
505 PRINT
510 PRINT "WHERE WILL LIGHTRAY STOP FOR A DRINK";
520 INPUT X
530 LET A(I)=INT(R*ATN(10/X)+.5)
535 PRINT
540 LET B(I)=INT(R*ATN(B/(A-X))+.5)
550 PRINT "ANGLE APPROACHING RIVER IS"A(I)"DEGREES."
560 PRINT "ANGLE LEAVING RIVER IS"B(I)"DEGREES."
570 IF A(I)=B(I) THEN 660
580 LET C(I)=ABS(A(I)-B(I))
585 IF I=1 THEN 700
590 IF C(I)>C(I-1) THEN 630
592 IF C(I)<C(I-1) THEN 600
594 PRINT "C'MON -- YOU TRIED THAT LAST TIME."
596 GO TO 700
600 PRINT "WELL, YOU ARE CLOSER THAN LAST TIME."
610 PRINT "KEEP AN EYE ON THOSE ANGLES, THOUGH."
615 PRINT "LET'S GO BACK FOR ANOTHER HORSE."
620 GO TO 700
630 PRINT "HEY - THIS TRIP TAKES LONGER!"
```

```
640 PRINT "YOU HAVE A DEAD HORSE ON YOUR HANDS.  TRY AGAIN."
650 GO TO 700
660 PRINT "NICE WORK.  YOU MADE IT."
670 PRINT "THE TRIP TOOK ABOUT"(SQR(X*X+100)+SQR((A-X)*(A-X)+B*B))/5;
675 PRINT " HOURS."
680 PRINT "YOU CAN SEE THAT USING HORSE SENSE, LIGHTRAY KNOWS THAT"
690 PRINT "THE ANGLES HAVE TO BE EQUAL OF REFLECTION FOR A"
692 PRINT "MINIMUM TIME TRIP."
695 GO TO 800
700 NEXT I
710 PRINT
720 PRINT "TOO MANY GUESSES, WE ARE OUT OF HORSES."
800 PRINT
810 PRINT "IF YOU WANT TO MOVE THE RANCH, TYPE 1"
820 PRINT "IF YOU WANT TO SEE SOMETHING ELSE, TYPE 2"
830 PRINT "IF YOU WANT TO QUIT, TYPE 3"
840 INPUT X
850 IF X=1 THEN 350
860 IF X=2 THEN 900
870 IF X=3 THEN 920
880 GO TO 840
900 PRINT "TIME FOR AN ACTUAL LIGHTRAY TO COMPLETE THE TRIP IS:"
910 PRINT (SQR(X*X+100)+SQR((A-X)*(A-X)+B*B))/1.86E5;" SECONDS."
920 PRINT
930 PRINT " ","THANK YOU FOR PLAYING."
940 END
```

DISCIPLINE PHYSICS

SUBJECT YOUNG'S DOUBLE SLIT EXP

PROGRAM NAME SLITS

DESCRIPTION:

Young's Double Slit Experiment is simulated by the computer to permit greater exploration of the influence of wavelength and slit-separation on the interference pattern. (This is a plotting program).

OBJECTIVES:

To determine, qualitatively, the effects of slit-separation, inter-screen spacing distance (d) and wavelength (w), in altering the location of the maxima and minima of the intensity bands of light.

PRELIMINARY PREPARATION:

A. Student - An instruction sheet is helpful in leading the student through a logical approach. It is also recommended that students understand the superposition of waves before executing this program.

B. Materials - none

DISCUSSION:

A. Operational Suggestions

1. The objectives of this program are best accomplished with small groups (3 to 4 students) to permit discussion and development of ideas concerning the relationships involved.
2. The program has worked well with highly-motivated students and has often led into detailed discussions of related topics. However, it has been found to be relatively ineffectual with poorly-motivated students.

B. Suggested Follow-up

This program permits the exploration of the parameters involved in double-slit interference patterns without the requirement of extensive equipment and/or set-ups. It is recommended that this simulated experiment be employed after the student has familiarized himself with the normal lab experiment.

Follow-up (con' t)

To enhance the operation of this program, it is further recommended that an instruction sheet (see attachment) be constructed to enable efficient exploration of this phenomenon. By varying the slit-separation (d), the student can observe the effects by noting the relative separations between adjacent maxima. In a similar manner, changes effectuated by the various wavelengths can also be noted.

Computer Instruction Sheet for Young's Double-Slit Experiment

The crucial experiment for the establishment of the wave nature of light was Young's double-slit experiment. The experiment clearly demonstrated diffraction and interference of light: a phenomenon characteristic of a wave-like nature. Realizing this wave property of light, we can now use the double-slit set-up to further study light sources.

Young's double-slit experiment is illustrated in figure 1, showing a symmetrical layout about line AB. The slits are located on an opaque screen a distance L from the observation screen. The slits are separated a distance d from center to center.

A wave front from the coherent light source reaches the opaque screen as a train of plane waves. Each slit then acts as a new light source (in phase with each other) which interfere with each other creating rays of high-intensity light (constructive interference); and rays of low-intensity light (destructive interference). These rays are most easily observed on the screen.

In this program we will attempt to determine the effects of the slit-separation distance (d) and the wavelength of the light on the interference (intensity) pattern.

ADDRESS COMPUTER PROGRAM SLITS



Figure 1

After addressing the program, type RUN. The teletype will then print out the intensity pattern observed when;

L = 2 meters; d = .5 millimeters; and W = 6000 Angstroms.

The left-hand margin shows the distance measured above (positive) and below (negative) the center line. This measurement (x) is in centimeters. This measurement is used to determine the position of the maxima (points of high-intensity light) and/or minima (points of low-intensity light).

The teletype will now ask you to specify a new value of d.

STUDY: How is the intensity pattern affected by changing the slit separation distance? (try several values, if necessary, to determine its effect).

DETERMINE: What happens to the distance x between maxima and minima as d is halved or doubled? Can you determine this relationship?

When you have varied d to your satisfaction, type 100 when asked to specify a new value of d. The teletype will then ask you to specify a new wavelength.

STUDY: How is the intensity pattern affected by changes in the wavelength? (Try several values if necessary).

DETERMINE: What happens to the distance between maxima and minima as W is halved or doubled? Can you determine this relationship?

You can test your ideas by typing 100 when asked to specify a new wavelength. You will then be asked to pick one of four light sources whose wavelength is unknown to you. You will also be asked to specify the value of d you will use in determining the unknown wavelength. You will have the opportunity of changing the d if you so desire.

If you are successful, or if time permits, you may try all four tests.

NOTE: To terminate the program during operation, type STOP after any of the question marks that appear; then return carriage.

To sign off the air, type BYE.

YOUNG'S DOUBLE SLIT EXPERIMENT

L = 2 METERS W = 6000 ANGSTROMS D = .5 MILLIMETERS

DISTANCE (MM'S FROM CENTER)



ABOVE IS AN ILLUSTRATIVE RUN WITH PRE-DETERMINED VALUES FOR WAVELENGTH (W), DISTANCE BETWEEN SLITS AND SCREEN (L), AND SLIT SEPARATION - CENTER TO CENTER (D). NOW YOU MAY VARY THESE PARAMETERS, ONE AT A TIME.

*****

```
WHAT IS THE NEW SLIT SEPARATION (D) IN MILLIMETERS? 1

L = 2 METERS     W = 6000 ANGSTROMS     D = 1 MILLIMETERS

DISTANCE (MM'S FROM CENTER)
```




```
WOULD YOU LIKE TO TRY ANOTHER VALUE OF D (1-YES, 0-NO)? 1
WHAT IS THE NEW SLIT SEPARATION (D) IN MILLIMETERS? .25

L = 2 METERS     W = 6000 ANGSTROMS     D = .25 MILLIMETERS
```



WOULD YOU LIKE TO TRY ANOTHER VALUE OF D (1-YES, 0-NO)? 0

*****

WHAT IS THE NEW WAVELENGTH (W) IN ANGSTROMS? 3000

L = 2 METERS W = 3000 ANGSTROMS D = .5 MILLIMETERS

DISTANCE (MM'S FROM CENTER)



WOULD YOU LIKE TO TRY ANOTHER VALUE OF W (1-YES, 0-NO)? 1
WHAT IS THE NEW WAVELENGTH (W) IN ANGSTROMS? 15000
A WAVELENGTH OF 15000 IS INFRARED LIGHT AND NOT VISIBLE.
THE INTERFERENCE PATTERN WILL BE VISIBLE USING DETECTORS ONLY. TRY ANOTHER WAVELENGTH.
WHAT IS THE NEW WAVELENGTH (W) IN ANGSTROMS? 6900

```
L = 2 METERS     W = 6900 ANGSTROMS     D = .5 MILLIMETERS

DISTANCE (MM'S FROM CENTER)
          -.26    .
          -.24    .
          -.22    .
          -.2     .
          -.18    .
          -.16    .*
          -.14    .
          -.12    .*
          -.1     .
          -.08    .
          -.06    .
          -.04    .
          -.02    .
..............0....INTENSITY.....
           .02    .
           .04    .
           .06    .
           .08    .
           .1     .
           .12    .*
           .14    .
           .16    .*
           .18    .
           .2     .
           .22    .
           .24    .
           .26    .

WOULD YOU LIKE TO TRY ANOTHER VALUE OF W (1-YES, 0-NO)? 0

*****

WHAT IS THE NEW DISTANCE FROM SLITS TO SCREEN (L) IN METERS? 5
```

```
L = 5 METERS     W = 6000 ANGSTROMS     D = .5 MILLIMETERS

DISTANCE (MM'S FROM CENTER)
          -.26    .*
          -.24    .  *
          -.22    .    *
          -.2     .      *
          -.18    .        *
          -.16    .          *
          -.14    .            *
          -.12    .              *
          -.1     .                *
          -.08    .                  *
          -.06    .                    *
          -.04    .                      *
          -.02    .                        *
..............0....INTENSITY.....*
           .02    .                        *
           .04    .                      *
           .06    .                    *
           .08    .                  *
           .1     .                *
           .12    .              *
           .14    .            *
           .16    .          *
           .18    .        *
           .2     .      *
           .22    .    *
           .24    .  *
           .26    .*
```

WOULD YOU LIKE TO TRY ANOTHER VALUE OF L (1-YES, 0-NO)? 0

*****

YOU WILL NOW BE GIVEN A LIGHT SOURCE OF UNKNOWN
WAVELENGTH. YOU WILL SPECIFY THE SLIT SEPARATION (D),
AND THE DISTANCE FROM SLITS TO SCREEN (L).
WHAT IS THE NEW SLIT SEPARATION (D) IN MILLIMETERS? .5
WHAT IS THE NEW DISTANCE FROM SLITS TO SCREEN (L) IN METERS? 3000
ALTHOUGH ANY DISTANCE LARGER THAN 5.000000E-3 METERS
IS VALID, ABOVE 5 METERS BECOMES HARD TO SEE. TRY ANOTHER VALUE.
WHAT IS THE NEW DISTANCE FROM SLITS TO SCREEN (L) IN METERS? 4

```
L = 4 METERS     W = ? ANGSTROMS     D = .5 MILLIMETERS

DISTANCE (MM'S FROM CENTER)
          -.26    *
          -.24    *
          -.22    *
          -.2     .*
          -.18    .  *
          -.16    .    *
          -.14    .      *
          -.12    .        *
          -.1     .          *
          -.08    .            *
          -.06    .              *
          -.04    .                *
          -.02    .                 *
..............0.....INTENSITY.......*
           .02    .                 *
           .04    .                *
           .06    .              *
           .08    .            *
           .1     .          *
           .12    .        *
           .14    .      *
           .16    .    *
           .18    .  *
           .2     .*
           .22    *
           .24    *
           .26    *
```

```
WOULD YOU LIKE A PLOT FOR OTHER VALUES OF D AND L (1-YES, 0-NO)? 0
WHAT DO YOU THINK THE UNKNOWN WAVELENGTH (W) IS? 6000
PRETTY GOOD!  THE WAVELENGTH WAS  6000  ANGSTROMS.
WOULD YOU LIKE TO TRY ANOTHER UNKNOWN WAVELENGTH(1-YES, 0-NO)? 0

*****

WOULD YOU LIKE A PLOT WITH YOUR OWN VALUES FOR WAVELENGTH
 (W), SLIT SEPARATION (D), AND DISTANCE FROM SLITS TO
SCREEN (L) (1-YES, 0-NO)? 1
WHAT IS THE NEW WAVELENGTH (W) IN ANGSTROMS? 5500
WHAT IS THE NEW SLIT SEPARATION (D) IN MILLIMETERS? .75
WHAT IS THE NEW DISTANCE FROM SLITS TO SCREEN (L) IN METERS? 3
```

```
L = 3 METERS     W = 5500 ANGSTROMS     D = .75 MILLIMETERS

DISTANCE (MM'S FROM CENTER)
```




```
ANOTHER ONE (1-YES, 0-NO)
? 0

**********

HOPE YOU HAD FUN!

READY
```

```
100 REM YOUNG'S DOUBLE SLIT EXPERIMENT
101 REM A.C. CAGGIANO
102 REM REVISED 7/28/70 (L. BRAUN, D. PESSEL)
103 REM IMPORTANT VARIABLES:  L-DISTANCE BETWEEN SLITS+SCREEN;
104 REM W-WAVELENGTH; D-SLIT SEPARATION(CENTER TO CENTER)
105 REM
106 REM U:  PRINT 'ARAMETER FOR UNKNOWN WAVELENGTH
107 LET U=0
110 PRINT " ", "YOUNG'S DOUBLE SLIT EXPERIMENT"
111 PRINT
120 REM ILLUSTRATIVE RUN
130 LET L=2
140 LET W=6000
150 LET D=.5
160 REM PLOT ROUTINE
170 GOSUB 850
171 PRINT
180 PRINT "ABOVE IS AN ILLUSTRATIVE RUN WITH PRE-DETERMINED"
181 PRINT "VALUES FOR WAVELENGTH (W), DISTANCE BETWEEN SLITS"
182 PRINT "AND SCREEN (L), AND SLIT SEPARATION - CENTER TO"
183 PRINT "CENTER (D).  NOW YOU MAY VARY THESE PARAMETERS,"
184 PRINT "ONE AT A TIME."
186 PRINT
187 PRINT "*****"
188 PRINT
190 REM D INPUT SUBROUTINE
200 GOSUB 920
210 REM PLOT ROUTINE
220 GOSUB 850
221 PRINT
230 PRINT "WOULD YOU LIKE TO TRY ANOTHER VALUE OF D (1-YES, 0-NO)";
240 INPUT Q1
250 IF Q1>0 THEN 190
260 PRINT
261 PRINT "*****"
262 PRINT
270 REM RESET D
280 LET D=.5
290 REM W INPUT SUBROUTINE
300 GOSUB 942
310 REM PLOT SUBROUTINE
320 GOSUB 850
321 PRINT
330 PRINT "WOULD YOU LIKE TO TRY ANOTHER VALUE OF W (1-YES, 0-NO)";
340 INPUT Q2
350 IF Q2>0 THEN 290
360 PRINT
```

```
361 PRINT "*****"
362 PRINT
370 REM RESET W
380 LET W=6000
390 REM L INPUT SUBROUTINE
400 GOSUB 900
410 REM PLOT SUBROUTINE
420 GOSUB 850
421 PRINT
430 PRINT "WOULD YOU LIKE TO TRY ANOTHER VALUE OF L (1-YES, 0-NO)";
440 INPUT Q3
450 IF Q3>0 THEN 390
460 PRINT
461 PRINT "*****"
462 PRINT
470 REM RESET L
480 LET L=2
490 PRINT "YOU WILL NOW BE GIVEN A LIGHT SOURCE OF UNKNOWN"
491 PRINT "WAVELENGTH.  YOU WILL SPECIFY THE SLIT SEPARATION (D),"
492 PRINT "AND THE DISTANCE FROM SLITS TO SCREEN (L)."
507 REM Q5 DETERMINES IF W IS TO BE CHANGED
508 LET Q5=0
520 REM D INPUT SUBROUTINE
530 GOSUB 920
550 REM L INPUT SUBROUTINE
560 GOSUB 900
565 REM CHANGE W?
566 IF Q5>0 THEN 600
570 REM RANDOMLY DETERMINE WAVELENGTH
580 RANDOMIZE
590 LET W=1000*INT(3*RND(X)+4.5)
600 REM PLOT SUBROUTINE (UNKNOWN W)
601 LET U=1
605 GOSUB 850
606 PRINT
610 PRINT "WOULD YOU LIKE A PLOT FOR OTHER VALUES OF D AND L ";
611 PRINT "(1-YES, 0-NO)";
620 INPUT Q5
630 IF Q5>0 THEN 520
640 PRINT "WHAT DO YOU THINK THE UNKNOWN WAVELENGTH (W) IS";
650 INPUT W1
660 IF ABS(W1-W)<.1*W THEN 700
670 PRINT "YOU ARE MORE THEN 10% OFF.  TO HELP YOU, YOU MAY ";
680 PRINT "OBTAIN MORE PLOTS."
690 GO TO 610
700 PRINT "PRETTY GOOD!  THE WAVELENGTH WAS "W" ANGSTROMS."
701 PRINT "WOULD YOU LIKE TO TRY ANOTHER UNKNOWN WAVELENGTH";
702 PRINT "(1-YES, 0-NO)";
703 INPUT Q6
704 IF Q6<1 THEN 967
705 PRINT "YOU MAY SPECIFY A NEW SLIT SEPARATION (D) AND DISTANCE"
706 PRINT "FROM SLITS TO SCREEN (L)."
707 GO TO 508
```

```
849 REM
850 REM PLOT ROUTINE
855 PRINT
856 PRINT
857 REM U>0 DO NOT PRINT WAVELENGTH
858 IF U>0 THEN 870
860 PRINT "L ="L"METERS     W ="W"ANGSTROMS     D ="D"MILLIMETERS"
861 PRINT
865 GO TO 875
870 PRINT "L ="L"METERS     W = ? ANGSTROMS     D ="D"MILLIMETERS"
871 PRINT
875 PRINT "DISTANCE (MM'S FROM CENTER)"
880 REM A:PLOT LOWER LIMIT (MM'S); B:UPPER LIMIT (MM'S)
881 LET A=-.26
882 LET B=.26
883 REM R:PRELIMINARY CALC. FOR INTENSITY; 10E4:CONVERSION FACTOR
884 LET R=(3.1416*D*10E4)/(W*L)
885 REM LOOP TO CALCULATE PATTERN AND PLOT IT
886 FOR X=A TO B STEP .02
887 REM Y:INTENSITY
888 REM 20:SCALE FACTOR FOR PLOT; X:DISTANCE (MM'S)
889 LET Y=20*COS(R*X)*COS(R*X)
890 IF ABS(X)<.0001 THEN 893
891 PRINT TAB(8);INT(X*100+.5)/100;TAB(15); "."; TAB(INT(Y+15.5)); "*"
892 GO TO 895
893 PRINT ".............0....INTENSITY......*"
895 NEXT X
896 LET U=0
897 PRINT
898 RETURN
899 REM
900 REM L INPUT SUBROUTINE
902 PRINT "WHAT IS THE NEW DISTANCE FROM SLITS TO SCREEN (L) ";
903 PRINT "IN METERS";
904 INPUT L
905 REM 1000:  CONVERT L(METERS) TO L(MILLIMETERS)
906 IF 1000*L>= 10*D THEN 912
907 PRINT "THIS DISTANCE IS TOO SMALL FOR GOOD INTERFERENCE PATTERNS."
908 PRINT "TRY ANOTHER VALUE."
910 GO TO 902
912 IF L<=5 THEN 918
913 PRINT "ALTHOUGH ANY DISTANCE LARGER THAN "10*D/1000" METERS"
914 PRINT "IS VALID, ABOVE 5 METERS BECOMES HARD TO SEE.";
915 PRINT "  TRY ANOTHER VALUE."
916 GO TO 902
918 RETURN
919 REM
920 REM D INPUT SUBROUTINE
922 PRINT "WHAT IS THE NEW SLIT SEPARATION (D) IN MILLIMETERS";
924 INPUT D
926 IF D>=.1 THEN 932
928 PRINT "SLITS ARE SO CLOSE THEY APPROXIMATE A SINGLE SLIT."
```

```
929 PRINT "TRY ANOTHER VALUE."
930 GO TO 920
932 IF D<=.1*1000*L THEN 940
933 PRINT "FOR A VALID INTERFERENCE PATTERN, THE SLIT SEPARATION"
934 PRINT "SHOULD BE LESS THAN ".1*1000*L" MILLIMETERS.  TRY";
935 PRINT " ANOTHER VALUE."
938 GO TO 920
940 RETURN
941 REM
942 REM W INPUT SUBROUTINE
944 PRINT "WHAT IS THE NEW WAVELENGTH (W) IN ANGSTROMS";
946 INPUT W
947 IF W>=300 THEN 954
948 IF W<1000 THEN 959
949 PRINT "A WAVELENGTH OF "W" IS ULTRAVIOLET LIGHT AND NOT VISIBLE."
950 GO TO 956
952 GO TO 944
954 IF W<=8000 THEN 965
955 PRINT "A WAVELENGTH OF "W" IS INFRARED LIGHT AND NOT VISIBLE."
956 PRINT "THE INTERFERENCE PATTERN WILL BE VISIBLE USING DETECTORS"
957 PRINT "ONLY.  TRY ANOTHER WAVELENGTH."
958 GO TO 944
959 PRINT "A WAVELENGTH OF "W" IS X-RAYS AND NOT VISIBLE."
960 GO TO 956
965 RETURN
966 REM
967 PRINT
968 PRINT "*****"
969 PRINT
970 REM MISCELLANEOUS RUNS
972 PRINT "WOULD YOU LIKE A PLOT WITH YOUR OWN VALUES FOR WAVELENGTH"
973 PRINT " (W), SLIT SEPARATION (D), AND DISTANCE FROM SLITS TO"
974 PRINT "SCREEN (L) (1-YES, 0-NO)";
976 INPUT Q9
980 IF Q9<1 THEN 995
982 GOSUB 942
984 GOSUB 920
986 GOSUB 900
988 GOSUB 850
990 PRINT "ANOTHER ONE (1-YES, 0-NO)"
992 INPUT Q8
993 IF Q8>0 THEN 982
994 REM
995 PRINT
996 PRINT "**********"
997 PRINT
998 PRINT "HOPE YOU HAD FUN!"
999 END

READY
```

DISCIPLINE PHYSICS

SUBJECT SNELL'S LAW

PROGRAM NAME SNELL

DESCRIPTION:

Snell's law is presented pictorially by plotting the path of a light ray as it crosses a boundary separating two different media.

OBJECTIVES:

To permit students to "see" the refraction of light, including the case when the critical angle is exceeded and reflection occurs.

PRELIMINARY PREPARATION:

A. Student - The terms associated with Snell's law, such as refraction, media, normals, etc.: must be presented prior to the running of this program.

B. Materials - No additional supplies or materials are necessary.

DISCUSSION:

Snell's law can be investigated independently by students by altering the angle of incidence, and/or the indices of refraction. The pictorial presentation is especially beneficial to students with reading problems, since the concepts implied by the mathematical relationships are presented hueristically.

In addition, the critical angle may be approached and exceeded, in the special case where n_2 (second medium) is less than n_1 (initial medium).

Queries are included as part of the program to reinforce the concepts.

The program is well suited for small groups or individuals, but may be utilized for large group presentation without program modification.

---REFRACTION OF LIGHT---

THIS PROGRAM WILL HELP YOU VISUALIZE THE REFRACTION OF LIGHT AS IT CROSSES A BOUNDARY SEPARATING TWO DIFFERENT MEDIA.

THE DIAGRAM BELOW SHOWS LIGHT INCIDENT TO THE BOUNDARY AT 45 DEGREES. THE INDICES OF REFRACTION ARE N1=1.0 AND N2=1.5 RESPECTIVELY.



WHAT DO YOU THINK THE ANGLE OF REFRACTION IS? 30
YOU ARE WITHIN 10 PERCENT.
THE ANGLE OF REFRACTION, A2= 28.126

DO YOU WANT TO CONTINUE (1=YES, 0=NO) ? ? 1

NOW YOU CAN CHANGE THE INCIDENT ANGLE. THE REFRACTIVE INDICES WILL REMAIN AS N1=1.0 AND N2=1.5 .

REMEMBER, ONLY POSITIVE ANGLES BETWEEN 0 AND 90 DEGREES ARE PERMISSIBLE ENTRIES.
SO, WHAT ANGLE DO YOU WANT? 60



WHAT DO YOU THINK THE ANGLE OF REFRACTION IS? 35
YOU ARE WITHIN 10 PERCENT.
THE ANGLE OF REFRACTION, A2= 35.264

```
DO YOU WANT TO CONTINUE (1=YES, 0=NO) ? ? 1

     NOW SPECIFY NEW VALUES FOR  N1, N2, AND ANGLE 1.
SEPARATE WITH COMMAS.  OKAY, WHAT VALUES? 2,3.5,15
VALUE OF N2 IS UNREASONABLE.
YOU MUST RE-TYPE ALL THREE NUMBERS.

 OKAY, WHAT VALUES? 1.5,2.5,15
```




```
WHAT DO YOU THINK THE ANGLE OF REFRACTION IS? 10
YOU ARE MORE THAN 10 PERCENT OFF.
THE ANGLE OF REFRACTION, A2= 8.934

DO YOU WANT TO CONTINUE (1=YES, 0=NO) ? ? 1
SPECIFY NEW VALUES FOR  N1, N2, AND ANGLE 1.
SEPARATE WITH COMMAS.  OKAY, WHAT VALUES? .05,1,45
VALUE OF N1 IS UNREASONABLE.
YOU MUST RE-TYPE ALL THREE NUMBERS.

 OKAY, WHAT VALUES? .25,1,45
VALUE OF N1 IS UNREASONABLE.
YOU MUST RE-TYPE ALL THREE NUMBERS.

 OKAY, WHAT VALUES? .5,1,45
VALUE OF N1 IS UNREASONABLE.
I SUGGEST YOU LEARN MORE ABOUT REFRACTION SO YOU CAN
ENTER MORE MEANINGFUL INDICES AND ANGLES.


READY
```

```
                ---REFRACTION OF LIGHT---

    THIS PROGRAM WILL HELP YOU VISUALIZE THE REFRACTION
OF LIGHT AS IT CROSSES A BOUNDARY SEPARATING TWO DIFFERENT
MEDIA.

THE DIAGRAM BELOW SHOWS LIGHT INCIDENT TO THE BOUNDARY AT
45 DEGREES.  THE INDICES OF REFRACTION ARE  N1=1.0 AND
N2=1.5  RESPECTIVELY.
```




```
WHAT DO YOU THINK THE ANGLE OF REFRACTION IS? 30
YOU ARE WITHIN 10 PERCENT.
THE ANGLE OF REFRACTION, A2= 28.126

DO YOU WANT TO CONTINUE (1=YES, 0=NO) : ? 1

    NOW YOU CAN CHANGE THE INCIDENT ANGLE. THE REFRACTIVE
INDICES WILL REMAIN AS  N1=1.0  AND  N2=1.5 .
```

```
REMEMBER, ONLY POSITIVE ANGLES BETWEEN 0 AND 90 DEGREES ARE
PERMISSIBLE  ENTRIES.
SO, WHAT ANGLE DO YOU WANT? 60
```




```
WHAT DO YOU THINK THE ANGLE OF REFRACTION IS? 35
YOU ARE WITHIN 10 PERCENT.
THE ANGLE OF REFRACTION, A2= 35.264

DO YOU WANT TO CONTINUE (1=YES, 0=NO) : ? 1

     NOW SPECIFY NEW VALUES FOR  N1, N2, AND ANGLE 1.
SEPARATE WITH COMMAS.  OKAY, WHAT VALUES? 2,3.5,15
VALUE OF N2 IS UNREASONABLE.
YOU MUST RE-TYPE ALL THREE NUMBERS.

 OKAY, WHAT VALUES? 1.5,2.5,15
```



```
WHAT DO YOU THINK THE ANGLE OF REFRACTION IS? 9
YOU ARE WITHIN 10 PERCENT.
THE ANGLE OF REFRACTION, A2= 8.934

DO YOU WANT TO CONTINUE (1=YES, 0=NO) : ? 1
SPECIFY NEW VALUES FOR  N1, N2, AND ANGLE 1.
SEPARATE WITH COMMAS.  OKAY, WHAT VALUES? 2,1,75
```




```
YOU WENT PAST THE CRITICAL ANGLE.

WHAT DO YOU THINK THE ANGLE OF REFLECTION IS? 75
THAT'S RIGHT, THE ANGLE OF REFLECTION IS 75 DEGREES.

DO YOU WANT TO CONTINUE (1=YES, 0=NO) : ? 0


READY
```

```
100 REM  RICHARD F. PAV    PATCHOGUE H.S.    OCT.'68
110 REM  THIS PROGRAM IS DESIGNED TO HELP A STUDENT VISUALIZE
120 REM  SNELL'S LAW.
130 REM  REVISED BY C.LOSIK  8-25-70
140 REM  A AND A1 ARE ANGLES, N1 AND N2 INDICES OF REFRACTION
160 PRINT " ","---REFRACTION OF LIGHT---"
170 PRINT
180 PRINT "     THIS PROGRAM WILL HELP YOU VISUALIZE THE REFRACTION"
190 PRINT "OF LIGHT AS IT CROSSES A BOUNDARY SEPARATING TWO DIFFERENT"
200 PRINT "MEDIA."
210 PRINT
220 PRINT "THE DIAGRAM BELOW SHOWS LIGHT INCIDENT TO THE BOUNDARY AT"
230 PRINT "45 DEGREES.  THE INDICES OF REFRACTION ARE  N1=1.0 AND"
240 PRINT "N2=1.5  RESPECTIVELY."
250 PRINT
260 LET N1=1
270 LET N2=1.5
280 LET A=45
290 GOSUB 820
300 PRINT
310 PRINT "     NOW YOU CAN CHANGE THE INCIDENT ANGLE. THE REFRACTIVE"
320 PRINT "INDICES WILL REMAIN AS  N1=1.0  AND  N2=1.5 ."
330 PRINT
340 PRINT "REMEMBER, ONLY POSITIVE ANGLES BETWEEN 0 AND 90 DEGREES ARE"
350 PRINT "PERMISSIBLE  ENTRIES."
360 PRINT"SO, WHAT ANGLE DO YOU WANT";
370 INPUT A1
380 LET A=A1
390 IF A<90 THEN 490
400 LET A=60
410 GOSUB 820
420 GO TO 520
430 PRINT
440PRINT"YOUR VALUE FOR THE INCIDENT ANGLE ("3A1;"DEGREES) DIDN'T MAKE"
450 PRINT "SENSE SO I AUTOMATICALLY MADE IT 60 DEGREES."
460 PRINT
470 LET A1=60
480 GO TO 1580
490 IF A<0 THEN 400
510 GO SUB 820
520 PRINT
530 PRINT "     NOW ";
540 PRINT "SPECIFY NEW VALUES FOR  N1, N2, AND ANGLE 1."
550 PRINT "SEPARATE WITH COMMAS. ";
570 PRINT " OKAY, WHAT VALUES";
580 INPUT N1,N2,A
590 IF N1<=3 THEN 630
610 PRINT "VALUE OF N1 IS UNREASONABLE."
620 GOTO 640
630 IF N1<1 THEN 610
640 IF N2<=3 THEN 670
650 PRINT "VALUE OF N2 IS UNREASONABLE."
660 GOTO 680
670 IF N2<1 THEN 650
680 IF A<90 THEN 710
690 PRINT "VALUE OF ANGLE 1 IS UNREASONABLE."
700 GOTO 780
710 IF A<0 THEN 690
720 IF N1>3 THEN 780
730 IF N1<1 THEN 780
740 IF N2>3 THEN 780
750 IF N2<1 THEN 780
760 GOSUB 820
770 GOTO 540
780 LET B=B+1
785 IF B>=3 THEN 1730
790 PRINT "YOU MUST RE-TYPE ALL THREE NUMBERS."
800 PRINT
810 GO TO 570
820 LET B=0
```

```
830 PRINT
840 PRINT "                    N"
850 LET C=(N1/N2)*SIN(A*1.74533E-2)
860 IF C>=1 THEN 1300
870 GOSUB 1520
880 FOR Y=8 TO 1 STEP -1
890 LET X=2*Y*((SIN(A*1.74533E-2))/(COS(A*1.74533E-2)))
900 IF X>16 THEN 960
910 PRINT TAB(16-X);"*";
950 GOTO 970
960 LET X=16
970 PRINT TAB(16);"."
1010 NEXT Y
1020 PRINT "                *.          MEDIUM 1, N1=";N1
1030 PRINT "----------------*-------------------------------"
1040 PRINT "                 .*         MEDIUM 2, N2=";N2
1050 FOR Y=1 TO 8 STEP 1
1060 LET X=2*Y*C/SQR(1-C*C)
1070 PRINT TAB(16);"."
1110 IF X>40 THEN 1160
1120 PRINT TAB(17+X);"*";
1160 PRINT " "
1170 NEXT Y
1180 PRINT
1190 GOSUB 1560
1200 PRINT "THE ANGLE OF REFRACTION, A2=";A2
1210 PRINT
1211 PRINT "DO YOU WANT TO CONTINUE (1=YES, 0=NO) ? ";
1212 INPUT J
1213 IF J=0 THEN 1750
1214 IF J<>1 THEN 1210
1220 RETURN
1230 PRINT "                *.*         MEDIUM 1, N1="N1
1240 PRINT "----------------*-------------------------------"
1250 PRINT "                            MEDIUM 2, N2=";N2
1260 PRINT "YOU WENT PAST THE CRITICAL ANGLE."
1270 GOSUB 1650
1280 PRINT "THE ANGLE OF REFLECTION IS";A;"DEGREES."
1290 GOTO 1210
1300 FOR Y=8 TO 1 STEP -1
1310 LET X=2*Y*((SIN(A*1.74533E-2))/(COS(A*1.74533E-2)))
1320 LET X2=X
1330 IF X>16 THEN 1390
1340 PRINT TAB(16-X);"*";
1380 GOTO 1400
1390 LET X=16
1400 PRINT TAB(16);".";
1440 IF X2>40 THEN 1490
1450 PRINT TAB(17+X2);"*";
1490 PRINT " "
1500 NEXT Y
1510 GOTO 1230
1520 LET F=C/SQR(1-C*C)
1530 LET G=ATN(F)
1540 LET A2=INT(1000*(G/1.74533E-2)+.5)/1000
1550 RETURN
1560 IF A1>=90 THEN 430
1570 IF A1<0 THEN 430
1580 PRINT "WHAT DO YOU THINK THE ANGLE OF REFRACTION IS";
1590 INPUT A3
1600 IF ABS(A2-A3)>.1*A2 THEN 1630
1610 PRINT "YOU ARE WITHIN 10 PERCENT."
1620 GOTO 1640
1630 PRINT "YOU ARE MORE THAN 10 PERCENT OFF."
1640 RETURN
1650 PRINT
1660 PRINT "WHAT DO YOU THINK THE ANGLE OF REFLECTION IS";
1670 INPUT A4
1680 IF A4<>A THEN 1710
1690 PRINT "THAT'S RIGHT, ";
1700 GOTO 1720
1710 PRINT "YOU HAD BETTER STUDY THE LAWS OF REFLECTION."
1720 RETURN
1730 PRINT "I SUGGEST YOU LEARN MORE ABOUT REFRACTION SO YOU CAN"
1740 PRINT "ENTER MORE MEANINGFUL INDICES AND ANGLES."
1750 END
```

DISCIPLINE PHYSICS

SUBJECT ORBITAL MOTION

PROGRAM NAME SPACE

AVG. EXECUTION TIME 3 min.

DESCRIPTION:

The effects of speed on orbital motion can be demonstrated by incrementally altering the tangential velocity of an orbitting spacecraft. Limiting cases are included, i. e. exceeding the escape velocity and/or crashing into the earth.

OBJECTIVES:

To demonstrate the effects of speed on orbital motion.

PRELIMINARY PREPARATION:

A. Student - Student should be familiar with circular motion, central forces, and have some knowledge of conic sections.

B. Materials - None

DISCUSSION:

Orbital motion is described in terms of the eccentricity (E) of the orbit. the period (T), and the maximum and minimum tangential velocities. The student selects the initial apogee and perigee (in miles) to define the orbit.

After describing the initial orbit the limiting changes required to produce circular and parabolic orbits are given, as well as the changes required to produce an orbit that will be tangent to the earth' s surface.

The speed at the apogee and perigee is given and the student may alter either of these values (+ or -) incrementally. The new orbit will again be described in terms of E, T, and the velocities.

SPACECRAFT ORBITS

```
WHAT IS THE MINIMUM AND MAXIMUM ALTITUDE OF THE SPACECRAFT
ABOVE THE SURFACE OF THE EARTH IN MILES? 150,230

THE ECCENTRICITY OF THE ORBIT IS 9.640877E-3

THE VELOCITY AT THE PERIGEE IS  25593.36  FEET/SECOND.

THE VELOCITY AT THE APOGEE IS  25404.58  FEET/SECOND.

THE PERIOD OF THE ORBIT IS  90.50426  MINUTES.

ADDING A VELOCITY INCREMENT TO THE PERIGEE
OF  10427.89  FT/SEC WOULD RESULT IN A
PARABOLIC ORBIT-- CAUSING THE SPACECRAFT TO FLY OFF INTO SPACE.

A CHANGE OF -128.4863  FT/SEC WOULD PRODUCE A
CIRCULAR ORBIT. HOWEVER, A VELOCITY INCREMENT
OF -360.3738  FT/SEC WOULD PRODUCE AN ORBIT
THAT WOULD BE TANGENT TO THE EARTH'S SURFACE.

ADDING A VELOCITY INCREMENT TO THE APOGEE
OF  10571.05  FT/SEC WOULD RESULT IN A
PARABOLIC ORBIT-- CAUSING THE SPACECRAFT TO FLY OFF INTO SPACE.

A CHANGE OF  121.8972  FT/SEC WOULD PRODUCE A
CIRCULAR ORBIT. HOWEVER, A VELOCITY INCREMENT
OF -836.6951  FT/SEC WOULD PRODUCE AN ORBIT
THAT WOULD BE TANGENT TO THE EARTH'S SURFACE.


DO YOU WANT TO ADD A VELOCITY INCREMENT AT THE
PERIGEE(TYPE 1) OR AT THE APOGEE(TYPE 2)  ? 1

WHAT VELOCITY INCREMENT IS TO BE ADDED? -850

THE NEW ORBIT IS ELLIPTICAL
 WITH AN ECCENTRICITY OF 9.987444E-3

THE POINT WHERE THE VELOCITY INCREMENT WAS ADDED
CORRESPONDS TO THE APOGEE OF THE NEW ORBIT
THE PERIGEE OF THE NEW ORBIT IN MILES IS 68.73486

 THE VELOCITY AT THE PERIGEE IS  85854.69  FT/SEC.

THE PERIOD OF THE NEW ORBIT IS  87.8788  MINUTES.

*****

BASED ON YOUR ORIGINAL ALTITUDES OF  150  AND  230  MILES
WOULD YOU LIKE TO TRY DIFFERENT VELOCITY INCREMENTS
(1-YES, 0-NO)? 0

WOULD YOU LIKE TO RUN THE PROGRAM AGAIN (1-YES, 0-NO)? 0

*****


READY
```

```
100 REM SPACECRAFT ORBITS, M. VISICH, JR., 12/09/68
101 REM REVISED 8/25/70  (D. PESSEL)
120 DIM V(2),R(2)
130 REM THIS PROGRAM CAN BE USED TO DETERMINE THE EFFECT OF
140 REM ADDING A VELOCITY INCREMENT TO A SPACECRAFT INITIALLY IN
150 REM AN ELLIPTIC ORBIT AROUND THE EARTH. VELOCITY INCREMENTS
160 REM CAN ONLY BE ADDED AT THE APOGEE OR PERIGEE OF THE INITIAL ORBIT
170 REM AND ONLY IN A DIRECTION TANGENT TO THE INITIAL ORBIT.
171 LET Q5=0
173 LET Y=1.40753E16
175 PRINT TAB(30);"SPACECRAFT ORBITS"
176 PRINT
180 PRINT "WHAT IS THE MINIMUM AND MAXIMUM ALTITUDE OF THE SPACECRAFT"
190 PRINT"ABOVE THE SURFACE OF THE EARTH IN MILES";
200 INPUT H1,H2
205 IF H1>0 THEN 208
206 PRINT "BOTH ALTITUDES MUST BE POSITIVE!"
207 GO TO 180
208 IF H2>0 THEN 210
209 GO TO 206
210 PRINT
220 GOSUB 960
225 IF Q5>0 THEN 307
230 PRINT"THE ECCENTRICITY OF THE ORBIT IS" E
240 PRINT
250 PRINT"THE VELOCITY AT THE PERIGEE IS "V1" FEET/SECOND."
270 PRINT
280 PRINT"THE VELOCITY AT THE APOGEE IS "V2" FEET/SECOND."
300 PRINT
304 PRINT"THE PERIOD OF THE ORBIT IS "T" MINUTES."
305 PRINT
307 LET V(1)=V1
308 LET V(2)=V2
309 LET R(1)=R1
310 LET R(2)=R2
311 IF Q5>0 THEN 334
312 LET V(2)=V2
313 LET R(1)=R1
314 LET R(2)=R2
315 FOR J=1 TO 2
316 PRINT"ADDING A VELOCITY INCREMENT TO THE ";
317 IF J=1 THEN 320
318 PRINT"APOGEE"
319 GOTO321
320 PRINT"PERIGEE"
321 PRINT"OF "SQR(2*Y/R(J))-V(J)" FT/SEC WOULD RESULT IN A"
322 PRINT"PARABOLIC ORBIT-- CAUSING THE SPACECRAFT TO FLY OFF INTO SPACE
323 PRINT
324 PRINT"A CHANGE OF "SQR(Y/R(J))-V(J)" FT/SEC WOULD PRODUCE A"
325 PRINT"CIRCULAR ORBIT. HOWEVER, A VELOCITY INCREMENT "
327 LET R=5280*3959
329 LET E2=(R-R(J))/(R+R(J))
330 PRINT"OF "SQR(Y*(1+E2)/R(J))-V(J)" FT/SEC WOULD PRODUCE AN ORBIT".
331 PRINT"THAT WOULD BE TANGENT TO THE EARTH'S SURFACE."
332 PRINT
333 NEXT J
334 PRINT
340 PRINT"DO YOU WANT TO ADD A VELOCITY INCREMENT AT THE"
350 PRINT"PERIGEE(TYPE 1) OR AT THE APOGEE(TYPE 2)".
360 INPUT N
370 PRINT
380 IF (N-1)*(N-2)=0 THEN 420
```

```
400 PRINT"YOU WERE TO PICK EITHER 1 OR 2-TRY AGAIN"
410 GO TO 340
420 PRINT"WHAT VELOCITY INCREMENT IS TO BE ADDED";
440 INPUT V3
450 PRINT
460 GOSUB 1050
470 PRINT"THE NEW ORBIT IS ";
480 IF E1=0 THEN680
490 IF E1=1 THEN 900
500 IF E1>1 THEN 930
305 PRINT"ELLIPTICAL "
510 PRINT" WITH AN ECCENTRICITY OF"E1
520 PRINT
530 IF V9>1 THEN 680
540 IF H3<0 THEN 660
550 PRINT"THE POINT WHERE THE VELOCITY INCREMENT WAS ADDED"
560 PRINT"CORRESPONDS TO THE APOGEE OF THE NEW ORBIT"
570 PRINT "THE PERIGEE OF THE NEW ORBIT IN MILES IS"R3
580 PRINT
590 PRINT" THE VELOCITY AT THE PERIGEE IS "V1" FT/SEC."
600 PRINT
610 LET R8=R3
680 GOSUB 1010
640 PRINT"THE PERIOD OF THE NEW ORBIT IS "T" MINUTES."
650 GO TO 1230
660 PRINT"YOU CRASHED INTO THE EARTH"
670 GO TO 1230
680 PRINT"THE POINT WHERE THE VELOCITY INCREMENT WAS ADDED"
690 PRINT"CORRESPONDS TO THE PERIGEE OF THE NEW ORBIT"
700 PRINT
710 PRINT"THE APOGEE OF THE NEW ORBIT IS "H4" MILES."
720 PRINT
730 PRINT"THE VELOCITY AT THE APOGEE IS "V8" FT/SEC."
740 PRINT
750 LET R8=R4
760 GOSUB 1010
780 PRINT"THE PERIOD OF THE NEW ORBIT IS "T" MINUTES."
790 PRINT
800 GOTO 1230
880 PRINT"CIRCULAR."
890 GOTO 1230
900 PRINT"PARABOLIC. "
910 PRINT "WITH AN ECCENTRICITY OF"E1
980 GO TO 1230
930 PRINT"HYPERBOLIC. "
940 PRINT "WITH AN ECCENTRICITY OF"E1
950 GO TO 1230
960 LET R1=(H1+3959)*5280
970 LET R8=(H2+3959)*5280
980 LET E=(R2-R1)/(R1+R2)
990 LET V1=SQR(Y*(1+E)/R1)
1000 LET V2=V1*R1/R2
1010 LET A=(R1+R2)/2
1020 LET P=39.479*A*A*A/Y
1030 LET T=SQR(P)/60
1040 RETURN
1050 IF N=1 THEN 1080
1060 LET V1=V2
1070 LET R1=R2
1080 LET V5=V1+V3
1090 LET V7=SQR(Y/R1)
1100 LET V9=V5/V7
1110 LET E1=ABS(V9*V9-1)
1120 IF V9>1 THEN 1160
1140 LET R3=(1-E1)*R1/(1+E1)
```

```
1150 LET H3=R3/5280-3959
1160 LET V1=V5*R1/R3
1170 GOTO 1220
1180 LET R4=(1+E1)*R1/(1-E1)
1200 LET H4=R4/5280-3959
1210 LET V2=V5*R1/R4
1220 RETURN
1230 PRINT
1231 PRINT "*****"
1232 PRINT
1234 PRINT "BASED ON YOUR ORIGINAL ALTITUDES OF "H1" AND "H2" MILES"
1235 PRINT "WOULD YOU LIKE TO TRY DIFFERENT VELOCITY INCREMENTS"
1236 PRINT "(1-YES, 0-NO)";
1237 INPUT Q5
1238 PRINT
1241 IF Q5>0 THEN 220
1250 PRINT "WOULD YOU LIKE TO RUN THE PROGRAM AGAIN (1-YES, 0-NO)";
1251 INPUT Q6
1252 PRINT
1253 PRINT "*****"
1254 PRINT
1255 IF Q6>0 THEN 180
1260 END
```

DISCIPLINE PHYSICS

SUBJECT ELECTRICAL POTENTIAL ENERGY

PROGRAM NAME VFIELD

DESCRIPTION:

This program plots a picture of the relative potential field strengths of regions surrounding two point charges.

OBJECTIVES:

To give the student a feel for how the electric potential field is altered by changing the positions of two point charges.

PRELIMINARY PREPARATION:

A. Student - The concept of electric potential for a point charge should be understood, as well as equipotential lines and potential hills or wells.

B. Materials - none needed

DISCUSSION:

One run of this program requires much time, so it is not advised for use with a whole class. Individual students or small groups will derive the greatest benefit, or, alternatively, the teacher may make several runs before class and display the resulting plots during a class discussion.

The coordinate plane occupied by the charges is 30 x 30.

NOTE: The numbers in the plots indicate relative field strengths.

```
THIS PROGRAM PLOTS A PICTURE OF THE RELATIVE ELECTRICAL
POTENTIAL FIELD STRENGTHS IN THE REGION SURROUNDING TWO
POINT CHARGES.  THE CHARGES ARE IN A COORDINATE PLANE
30 BY 30.  THE CHARGES MAY HAVE ANY VALUE WHOSE
MAGNITUDE IS LESS THAN 10 , AND MAY BE ANYWHERE BETWEEN
0 AND 30 ON THE X AND Y AXES.
THE MAGNITUDE OF THE FIELD DECREASES FROM 9 TO 1, 0, A TO J
THAT IS, 9 TO 1 IS A POSITIVE POTENTIAL, 0 IS ROUGHLY 0,
AND A=-1, B=-2, ... J=-9. (THERE IS NO 'I'.)

WHAT VALUES OF CHARGES DO YOU WISH TO STUDY?
TO STUDY ONLY ONE CHARGE, MAKE THE SECOND CHARGE 0.
ENTER TWO VALUES OF CHARGE : ? 10,-3.5
WHERE SHALL THE FIRST CHARGE BE LOCATED? 15,15
WHERE SHALL THE SECOND CHARGE BE LOCATED? 15,25
```




```
DO YOU WISH TO VIEW ANOTHER PLOT (1=YES, 0=NO) : ? 0


READY
```

```
100 REM  JOHN HOSIE - NORTHPORT HS - 7-8-69
105 REM  REVISED BY C.LOSIK  8-21-70
110 REM  YOU HAVE MY BLESSING TO USE COMPUTED GO-TO'S AND
111 REM  STRINGS IF YOU HAVE THEM (WE DIDN'T)
113 REM  V IS THE FIELD STRENGTH, R1,R2,Q1,Q2 ARE STANDARD NOT.
116 REM  THIS MAY BE CHANGED TO ALLOW HIGHER VALUED CHARGES
117 LET Q9=10
120 PRINT "THIS PROGRAM PLOTS A PICTURE OF THE RELATIVE ELECTRICAL"
130 PRINT "POTENTIAL FIELD STRENGTHS IN THE REGION SURROUNDING TWO"
140 PRINT "POINT CHARGES.  THE CHARGES ARE IN A COORDINATE PLANE"
150 PRINT "30 BY 30.  THE CHARGES MAY HAVE ANY VALUE WHOSE"
160 PRINT "MAGNITUDE IS LESS THAN"Q9", AND MAY BE ANYWHERE BETWEEN"
170 PRINT "0 AND 30 ON THE X AND Y AXES."
171 PRINT "THE MAGNITUDE OF THE FIELD DECREASES FROM 9 TO 1. 0, A TO J"
172 PRINT "THAT IS, 9 TO 1 IS A POSITIVE POTENTIAL, 0 IS ROUGHLY 0,"
173 PRINT "AND A=-1, B=-2, ... J=-9. (THERE IS NO 'I'.)"
178 PRINT
179 PRINT "WHAT VALUES OF CHARGES DO YOU WISH TO STUDY?"
180 PRINT "TO STUDY ONLY ONE CHARGE, MAKE THE SECOND CHARGE 0."
185 PRINT "ENTER TWO VALUES OF CHARGE : ";
190 INPUT Q1,Q2
195 IF Q1=0 THEN 180
200 IF ABS(Q1)<=Q9 THEN 230
210 PRINT "VALUES MUST BE IN THE RANGE ("-Q9","Q9")."
220 GO TO 185
230 IF ABS(Q2)>Q9 THEN 210
232 REM  FOR EFFICIENCY, WE 'HIDE' THE ELECTROSTATIC CONSTANT HERE
233 LET Q1=2*Q1
236 LET Q2=2*Q2
240 PRINT "WHERE SHALL THE FIRST CHARGE BE LOCATED";
250 INPUT X1,Y1
260 LET X1=INT(X1+.5)
270 LET Y1=INT(Y1+.5)
280 IF ABS(X1-15)<=15 THEN 310
290 PRINT "VALUES MUST BE IN THE RANGE (0,30)."
300 GO TO 240
310 IF ABS(Y1-15)>15 THEN 290
320 IF Q2=0 THEN 410
330 PRINT "WHERE SHALL THE SECOND CHARGE BE LOCATED";
340 INPUT X2,Y2
350 LET X2=INT(X2+.5)
360 LET Y2=INT(Y2+.5)
370 IF ABS(X2-15)<=15 THEN 400
380 PRINT "VALUES MUST BE IN THE RANGE (0,30)."
390 GO TO 330
400 IF ABS(Y2-15)>15 THEN 380
410 PRINT
420 PRINT
430 PRINT " ","*0         6         12        18        24        30*"
440 PRINT " ","*I---------I---------I---------I---------I---------I*"
450 FOR Y=30 TO 0 STEP -1
453 PRINT "       "INT(Y+.5),"*";
456 LET Y6=(Y-Y1)*(Y-Y1)
457 LET Y7=(Y-Y2)*(Y-Y2)
459 REM  THIS IS FOR 50 ITERATIONS
460 FOR X=0 TO 30 STEP .6
465 LET X6=X-X1
470 LET R1=SQR(X6*X6+Y6)
475 LET X7=X-X2
480 LET R2=SQR(X7*X7+Y7)
483 IF R1<.5 THEN 800
486 IF R2<.5 THEN 850
```

```
488 REM  THE ELECTROSTATIC CONSTANT IS 2
490 LET V=Q1/R1+Q2/R2
498 FOR J=-9 TO 9
500 IF ABS(V-J)<.35 THEN 508
502 NEXT J
504 PRINT " ";
506 GO TO 700
508 IF J>0 THEN 610
510 IF J<>-9 THEN 520
513 PRINT "J";
516 GO TO 700
520 IF J<>-8 THEN 530
523 PRINT "H";
526 GO TO 700
530 IF J<>-7 THEN 540
533 PRINT "G";
536 GO TO 700
540 IF J<>-6 THEN 550
543 PRINT "F";
546 GO TO 700
550 IF J<>-5 THEN 560
553 PRINT "E";
556 GO TO 700
560 IF J<>-4 THEN 570
563 PRINT "D";
566 GO TO 700
570 IF J<>-3 THEN 580
573 PRINT "C";
576 GO TO 700
580 IF J<>-2 THEN 590
583 PRINT "B";
586 GO TO 700
590 IF J<>-1 THEN 600
593 PRINT "A";
596 GO TO 700
600 IF J<>0 THEN 610
603 PRINT "0";
606 GO TO 700
610 IF J<>1 THEN 620
613 PRINT "1";
616 GO TO 700
620 IF J<>2 THEN 630
623 PRINT "2";
626 GO TO 700
630 IF J<>3 THEN 640
633 PRINT "3";
636 GO TO 700
640 IF J<>4 THEN 650
643 PRINT "4";
646 GO TO 700
650 IF J<>5 THEN 660
653 PRINT "5";
656 GO TO 700
660 IF J<>6 THEN 670
663 PRINT "6";
666 GO TO 700
670 IF J<>7 THEN 680
673 PRINT "7";
676 GO TO 700
```

```
680 IF J<>8 THEN 690
683 PRINT "8";
686 GO TO 700
690 IF J<>9 THEN 504
693 PRINT "9";
700 NEXT X
710 PRINT "*"
780 NEXT Y
730 PRINT " ","*I---------I---------I---------I---------I---------I*"
740 PRINT
750 PRINT
760 PRINT "DO YOU WISH TO VIEW ANOTHER PLOT (1=YES, 0=NO) : ";
770 INPUT Q1
775 PRINT
776 PRINT
780 IF Q1=1 THEN 178
790 IF Q1=0 THEN 999
795 GO TO 750
800 IF Q1>0 THEN 830
810 PRINT "-";
820 GO TO 700
830 PRINT "+";
840 GO TO 700
850 IF Q2=0 THEN 490
860 IF Q2>0 THEN 830
870 GO TO 810
999 END
```

DISCIPLINE PHYSICS

SUBJECT INSTANTANEOUS VELOCITY

PROGRAM NAME VLOCTY

DESCRIPTION:

A graph of distance vs. time is plotted for a body accelerating at 1m/sec/sec. The average velocity is found for a point on the graph several times using V average $=(d_2 - d_1)/(T_2 - T_1)$ as $(T_2 - T_1)$ gets smaller and smaller.

The program prints the instantaneous velocity at the points and allows the student to change some of the parameters involved.

OBJECTIVES:

To aid the student in understanding the meaning of instantaneous velocity and taking a limit.

PRELIMINARY PREPARATION:

A. Student - should know the definitions of average and instantaneous velocity

B. Materials - none

DISCUSSION:

A good tutorial program or teaching aid. Student should realize that the slope of the line drawn between the points d1, T1, and d2, T2 is the average velocity. As the second point is made to approach the first, the slope of this line approaches the value of the slope of the tangent line drawn to the first point - which is called the instantaneous velocity.

The student may then change the acceleration, time at which he wants to know the average speed, and the time interval, delta T.

A more theoretical view of this same problem will be obtained by running the program SLOPE.

AVERAGE AND INSTANTANEOUS VELOCITY

THIS PROGRAM CONSIDERS DISTANCE AS A FUNCTION OF TIME,
D=F(T). IT WILL CALCULATE THE AVERAGE VELOCITY DURING
THE TIME INTERVAL T1,T2 BY EVALUATING D AT THOSE TIMES
GIVING D1 AND D2. THE RESULT OF (D2-D1)/(T2-T1) YIELDS
THE AVERAGE VELOCITY. AS T2 IS BROUGHT CLOSER AND CLOSER
TO T1 THE RESULTANT AVERAGE VELOCITY WILL APROACH THE
INSTANTANEOUS VELOCITY AT T1.

AFTER THE PROGRAM STOPS, TYPE IN THE FOLLOWING:
(END EACH LINE, INCLUDING 'RUN', WITH THE 'RETURN' KEY)

```
1 GO TO 300
300 DEF FND(T)=....(YOUR FUNCTION OF TIME)....
RUN
```

FOR EXAMPLE, TO USE THE EQUATION D=A*T*T WITH A=1
YOU WOULD TYPE AS FOLLOWS:

```
1 GO TO 300
300 DEF FND(T)=1*T*T
RUN
```

YOU MIGHT TRY THAT AS YOUR FIRST RUN.
FOR SUBSEQUENT RUNS, YOU NEED ONLY CHANGE LINE 300 FOR
A NEW FUNCTION, FOLLOWED BY 'RUN'.

READY

```
1 GO TO 300
300 DEF FND(T)=1*T*T
RUN
```

*****

WHAT ARE YOUR VALUES OF T1 AND T2 (SMALLER FIRST: T1,T2)? 5,50

THE DISTANCE TRAVELED DURING THE INTERVAL IS 2475
THE AVERAGE VELOCITY IS 55

WOULD YOU LIKE TO CHANGE T2 (1-YES, 0-NO)? 1
WHAT IS YOU NEW VALUE FOR T2 (T2 MUST BE GREATER THAN T1)? 105

THE DISTANCE TRAVELED DURING THE INTERVAL IS 11000
THE AVERAGE VELOCITY IS 110

WOULD YOU LIKE TO CHANGE T2 (1-YES, 0-NO)? 0

NOW WATCH THE AVERAGE VELOCITY AS T2 APPROACHES T1.

T1 = 5 D1 = 25

T2	T2-T1	D2	D2-D1	(D2-D1)/(T2-T1)
105	100	11025	11000	110
55	50	3025	3000	60
30	25	900	875	35
17.5	12.5	306.25	281.25	22.5
11.25	6.25	126.5625	101.5625	16.25
8.125	3.125	66.01563	41.01563	13.125
6.5625	1.5625	43.06641	18.06641	11.5625
5.78125	.78125	33.42285	8.422852	10.78125
5.390625	.390625	29.05884	4.058838	10.39063
5.195313	.1953125	26.99127	1.991272	10.19531
5.097656	.09765625	25.9861	.9860992	10.09766
5.048828	.04882813	25.49067	.4906654	10.04883
5.024414	.02441406	25.24474	.2447367	10.02441
5.012207	.01220703	25.12222	.1222193	10.01221
5.006104	6.103516E-3	25.06107	.06107235	10.00609
5.003052	3.051758E-3	25.03053	.03052688	10.00305
5.001526	1.525879E-3	25.01526	.01526117	10.00156
5.000763	7.629395E-4	25.00763	7.629871E-3	10.00063

```
NOTE THAT THE AVERAGE VELOCITY CHANGES VERY LITTLE
AS T2 APPROACHES T1.  T2 CAN NEVER EQUAL T1 SINCE
(D2-D1)/(T2-T1) WOULD THEN RESULT IN A DIVISION BY ZERO.

*****

WOULD YOU LIKE TO TRY DIFFERENT VALUES OF T1 AND T2
(1-YES, 0-NO)? 0
TO CHANGE YOUR FUNCTION SEE THE INSTRUCTIONS.
IF YOU ARE FINISHED, TYPE '1', AND THE 'RETURN' KEY
AFTER THE PROGRAM STOPS.


READY

1
```

```
100 REM VELOCITY, J. HOSIE, Q. J. O'CONNOR, 8/12/68
101 REM REVISED 8/26/70 (D. PESSEL)
102 REM IMPORTANT VARIABLES:  S-SECANT SLOPE; P-PERCENT CHANGE;
103 REM D-INVERSE OF CHANGE IN X; Y-CHANGE IN Y
104 REM SEE SLOPE FOR A MORE THEORETICAL APPROACH TO THE SAME PROBLEM
105 LET S1=0
110 PRINT TAB(10);"AVERAGE AND INSTANTANEOUS VELOCITY"
120 PRINT
130 PRINT "THIS PROGRAM CONSIDERS DISTANCE AS A FUNCTION OF TIME,"
131 PRINT "D=F(T).  IT WILL CALCULATE THE AVERAGE VELOCITY DURING"
132 PRINT "THE TIME INTERVAL T1,T2 BY EVALUATING D AT THOSE TIMES"
133 PRINT "GIVING D1 AND D2.  THE RESULT OF (D2-D1)/(T2-T1) YIELDS"
134 PRINT "THE AVERAGE VELOCITY.  AS T2 IS BROUGHT CLOSER AND CLOSER"
135 PRINT "TO T1 THE RESULTANT AVERAGE VELOCITY WILL APROACH THE"
136 PRINT "INSTANTANEOUS VELOCITY AT T1."
138 PRINT
139 PRINT "AFTER THE PROGRAM STOPS, TYPE IN THE FOLLOWING:"
140 PRINT "(END EACH LINE, INCLUDING 'RUN', WITH THE 'RETURN' KEY)"
141 PRINT
142 PRINT "          1 GO TO 300"
143 PRINT "          300 DEF FND(T)=....(YOUR FUNCTION OF TIME)...."
145 PRINT "          RUN"
146 PRINT
147 PRINT "FOR EXAMPLE, TO USE THE EQUATION D=A*T*T WITH A=1"
148 PRINT "YOU WOULD TYPE AS FOLLOWS:"
149 PRINT
150 PRINT "          1 GO TO 300"
151 PRINT "          300 DEF FND(T)=1*T*T"
153 PRINT "          RUN"
154 PRINT
155 PRINT "YOU MIGHT TRY THAT AS YOUR FIRST RUN."
156 PRINT "FOR SUBSEQUENT RUNS, YOU NEED ONLY CHANGE LINE 300 FOR"
157 PRINT "A NEW FUNCTION, FOLLOWED BY 'RUN'."
160 STOP
290 REM CALCULATION OF SLOPE AND PRINTOUT
300 DEF FND(T)=1*T*T
301 PRINT
302 PRINT "*****"
303 PRINT
305 PRINT "WHAT ARE YOUR VALUES OF T1 AND T2 (SMALLER FIRST: T1,T2)";
306 INPUT T1,T2
307 IF T2>T1 THEN 310
308 PRINT "T2 MUST BE GREATER THAN T1!"
309 GO TO 305
310 PRINT
311 PRINT "THE DISTANCE TRAVELED DURING THE INTERVAL IS "FND(T2)-FND(T1)
312 PRINT "THE AVERAGE VELOCITY IS "(FND(T2)-FND(T1))/(T2-T1)
313 PRINT
314 PRINT "WOULD YOU LIKE TO CHANGE T2 (1-YES, 0-NO)";
315 INPUT Q1
316 IF Q1<1 THEN 330
317 PRINT "WHAT IS YOU NEW VALUE FOR T2 (T2 MUST BE GREATER THAN T1)";
318 INPUT T2
319 IF T2>T1 THEN 310
320 PRINT "T2 MUST BE GREATER THAN T1!"
321 GO TO 317
330 PRINT
331 PRINT "NOW WATCH THE AVERAGE VELOCITY AS T2 APPROACHES T1."
335 PRINT
344 LET D1=FND(T1)
345 PRINT " T1 = "T1," "," D1 = "FND(T1)
346 PRINT
```

```
350 PRINT " T2","  T2-T1","  D2","  D2-D1","  (D2-D1)/(T2-T1)"
352 PRINT " --","  -----","  --","  -----","  ---------------"
361 LET D2=FND(T2)
370 PRINT T2,T2-T1,D2,D2-D1,(D2-D1)/(T2-T1)
380 IF ABS(T2-T1)<.001 THEN 390
382 LET T2=T2-.5*(T2-T1)
384 GO TO 361
390 PRINT
400 PRINT "NOTE THAT THE AVERAGE VELOCITY CHANGES VERY LITTLE"
401 PRINT "AS T2 APPROACHES T1.  T2 CAN NEVER EQUAL T1 SINCE"
402 PRINT "(D2-D1)/(T2-T1) WOULD THEN RESULT IN A DIVISION BY ZERO."
470 PRINT
480 PRINT "*****"
490 PRINT
495 PRINT "WOULD YOU LIKE TO TRY DIFFERENT VALUES OF T1 AND T2"
496 PRINT "(1-YES, 0-NO)";
498 INPUT Q5
500 IF Q5>0 THEN 300
510 PRINT "TO CHANGE YOUR FUNCTION SEE THE INSTRUCTIONS."
520 PRINT "IF YOU ARE FINISHED, TYPE '1', AND THE 'RETURN' KEY"
530 PRINT "AFTER THE PROGRAM STOPS."
540 END
```

DISCIPLINE PHYSICS

SUBJECT WAVES

PROGRAM NAME WAVES

DESCRIPTION:

This program finds the sum of two waves: one predetermined by the program, and the other determined by the student. There are options of either displaying both waves and their sum, or just their sum.

OBJECTIVES:

To enable the student to study, independently, the effect of changes in wavelength, amplitude,and phase on the superposition pattern formed by two waves.

PRELIMINARY PREPARATION:

A. Student - Some experience with "SLINKY" wave superposition: knowledge of phase, amplitude, and wave length.

B. Materials - none

DISCUSSION:

The student controlled wave ("B") may have wavelengths ranging from 2 to 8, though only a wave length of 4 may be fully displayed. Its amplitude can be varied between 5 and 11, and its phase can be any decimal part of a wavelength.

The fixed wave ("A") has a wavelength of 4, and an amplitude of 10.

The display consists of both waves, side by side, and their superposition, or just their superposition.

```
                WAVES AND THEIR SUPERPOSITION
DO YOU NEED INSTRUCTIONS (1=YES, 0=NO) : ? 1

IN THIS PROGRAM YOU MAY OBSERVE THE EFFECTS OF
CHANGING WAVELENGTH, AMPLITUDE, AND PHASE ON TWO
WAVES AND ON THEIR SUM (OR SUPERPOSITION).

WAVE 'A' IS FIXED.  ITS WAVELENGTH IS 4, ITS AMPLITUDE
IS 10, AND ITS PHASE IS 0.

WAVE 'B' MAY BE CHANGED BY YOU.  FOR BEST RESULTS :
               WAVELENGTH (L) BETWEEN 2 AND 4
               AMPLITUDE (A) BETWEEN 5 AND 10
               PHASE (P) BETWEEN 0 AND 1
               (FOR EXAMPLE, .5 PHASE = 1/2 WAVELENGTH

IT IS EASIEST TO SEE THE EFFECTS OF CHANGES IN EACH
PARAMETER IF YOU HOLD TWO CONSTANT AND VARY THE OTHER,
ALTHOUGH ALL THREE MAY BE VARIED AT ONCE.

WHAT IS YOUR CHOICE OF WAVELENGTH, AMPLITUDE, AND PHASE? 2,10,0

NOTATION:      A = 'A' WAVE
               B = 'B' WAVE
               S = SUPERPOSITION WAVE
```



```
                WAVES AND THEIR SUPERPOSITION
DO YOU NEED INSTRUCTIONS (1=YES, 0=NO) : ? 0

WHAT IS YOUR CHOICE OF WAVELENGTH, AMPLITUDE, AND PHASE? 3,5,0.25

NOTATION:      A = 'A' WAVE
               B = 'B' WAVE
               S = SUPERPOSITION WAVE
```




```
WANT TO TRY ANOTHER SET OF VALUES (1=YES, 0=NO) : ? 0

READY
```

```
                WAVES AND THEIR SUPERPOSITION
DO YOU NEED INSTRUCTIONS (1=YES, 0=NO) : ? 0

WHAT IS YOUR CHOICE OF WAVELENGTH, AMPLITUDE, AND PHASE? 4,10,.5

NOTATION:       A = 'A' WAVE
                B = 'B' WAVE
                S = SUPERPOSITION WAVE
```




```
WANT TO TRY ANOTHER SET OF VALUES (1=YES, 0=NO) : ? 0

READY
```

WANT TO TRY ANOTHER SET OF VALUES (1=YES, 0=NO) : ? 1

WHAT IS YOUR CHOICE OF WAVELENGTH, AMPLITUDE, AND PHASE? 6,10,0

NOTATION: A = 'A' WAVE
B = 'B' WAVE
S = SUPERPOSITION WAVE



WANT TO TRY ANOTHER SET OF VALUES (1=YES, 0=NO) : ? 1

WHAT IS YOUR CHOICE OF WAVELENGTH, AMPLITUDE, AND PHASE? 6,10,0.25

NOTATION: A = 'A' WAVE
B = 'B' WAVE
S = SUPERPOSITION WAVE



WANT TO TRY ANOTHER SET OF VALUES (1=YES, 0=NO) : ? 0

READY

```
100 REM  JOHN W. HOSIE, NORTHPORT HS, PHYSICS, 8/9/68
105 REM  REVISED BY C.LOSIK  8-17-70
110 REM  WE SORT THE W(I) TO PLOT THE WAVE VALUES.
115 REM  L,A,P ARE WAVELENGTH, AMPLITUDE, AND PHASE
120 DIM W(3)
124 REM  TWO PI !
125 LET P2=2*3.14159
130 PRINT " ","WAVES AND THEIR SUPERPOSITION"
140 PRINT "DO YOU NEED INSTRUCTIONS (1=YES, 0=NO) : ";
150 INPUT A
160 IF A=0 THEN 350
170 IF A<>1 THEN 140
180 PRINT
190 PRINT "IN THIS PROGRAM YOU MAY OBSERVE THE EFFECTS OF"
200 PRINT "CHANGING WAVELENGTH, AMPLITUDE, AND PHASE ON TWO"
210 PRINT "WAVES AND ON THEIR SUM (OR SUPERPOSITION)."
220 PRINT
230 PRINT "WAVE 'A' IS FIXED.  ITS WAVELENGTH IS 4, ITS AMPLITUDE"
240 PRINT "IS 10, AND ITS PHASE IS 0."
250 PRINT
260 PRINT "WAVE 'B' MAY BE CHANGED BY YOU.  FOR BEST RESULTS :"
270 PRINT " ","WAVELENGTH (L) BETWEEN 2 AND 4"
280 PRINT " ","AMPLITUDE (A) BETWEEN 5 AND 10"
290 PRINT " ","PHASE (P) BETWEEN 0 AND 1"
300 PRINT " ","(FOR EXAMPLE, .5 PHASE = 1/2 WAVELENGTH"
310 PRINT
320 PRINT "IT IS EASIEST TO SEE THE EFFECTS OF CHANGES IN EACH"
330 PRINT "PARAMETER IF YOU HOLD TWO CONSTANT AND VARY THE OTHER."
340 PRINT "ALTHOUGH ALL THREE MAY BE VARIED AT ONCE."
350 PRINT
360 PRINT "WHAT IS YOUR CHOICE OF WAVELENGTH, AMPLITUDE, AND PHASE";
370 INPUT L,A,P
373 IF L>0 THEN 380
375 PRINT "WAVELENGTH IS ALWAYS A POSITIVE QUANTITY."
377 GO TO 350
380 IF L<=6 THEN 410
390 PRINT "YOUR WAVELENGTH IS TOO LONG FOR GOOD DISPLAY."
400 GO TO 350
410 IF L>=1 THEN 415
412 PRINT "YOUR WAVELENGTH IS TOO SHORT FOR GOOD DISPLAY."
413 GO TO 350
415 IF ABS(A)<=15 THEN 440
420 PRINT "YOUR AMPLITUDE IS TOO LARGE FOR DISPLAY."
430 GO TO 350
440 IF ABS(P-.5)<=.5 THEN 470
450 PRINT "YOUR PHASE CAN ONLY BE BETWEEN 0 AND 1."
460 GO TO 350
470 PRINT
```

```
480 PRINT "NOTATION:","A = 'A' WAVE"
490 PRINT " ","B = 'B' WAVE"
500 PRINT " ","S = SUPERPOSITION WAVE"
510 PRINT
520 PRINT "      X (CM.)","     -20        -10         0         +10      "
530 PRINT " ","-I----I---------I---------I---------I---------I----I"
540 FOR X=0 TO 4 STEP .2
545 PRINT "       "; INT(10*X+.5)/10;"-";
550 LET W(1)=INT(10*SIN(P2*X/4)+.5)
560 LET W(2)=INT(A*SIN(P2*(X/L+P))+.5)
570 LET W(3)=INT(W(1)+W(2)+.5)
580 REM  FIND WHICH IS SMALLEST, THEN PRINT IT AND MAXIMIZE IT
600 FOR Q=1 TO 3
605 LET K=1E20
610 FOR I=1 TO 3
620 IF W(I)>K THEN 640
630 LET K=W(I)
640 NEXT I
650 PRINT TAB(K+40);
660 FOR I=1 TO 3
670 IF ABS(W(I)-K)<.0001 THEN 700
680 NEXT I
690 STOP
700 IF I<>1 THEN 730
710 PRINT "A";
720 GO TO 780
730 IF I<>2 THEN 760
740 PRINT "B";
750 GO TO 780
760 IF I<>3 THEN 690
770 PRINT "S";
780 LET W(I)=1E25
790 NEXT Q
795 PRINT " "
800 NEXT X
810 PRINT " ","-I----I---------I---------I---------I---------I----I"
820 PRINT
830 PRINT "WANT TO TRY ANOTHER SET OF VALUES (1=YES, 0=NO) : ";
840 INPUT A
850 IF A=1 THEN 350
860 IF A<>0 THEN 820
870 END

READY
```